
  .به هر نحوي غيرمجاز است نويسنده  طرف ز ابدون مجوز كتبينشر غير الكترونيكي اين مطالب چاپ و يا 
  . آن به صورت مستقيم و يا غير مستقيم در نشريات الكترونيكي بلا مانع استانتشار اين مطالب بر روي اينترنت و يا استفاده از

 ٢

  :مقدمه 
     

 بـه همـين جهـت در    . اسـت " مطالعـه  -برنامه نويسي   -مطالعه ":     بهترين روش فراگيري يك زبان برنامه نويسي، سيستم  
 ارائه شـده و سـپس نكـات مهـم     اجراامل و قابل ، در ابتدا يك مثال ك    قسمت دوم آشنايي با برنامه نويسي اسمبلي ويندوز         

  .باشند تمامي كدهاي ذكر شده در متن ، ضميمه اين مجموعه مي.گيرند كدنويسي آن مورد بررسي قرار مي
  
  
  

  ها در ويندوز  پنجره-۱مثال 
  

  :   برنامه زير را در نظر بگيريد
.486 
.model flat, stdcall 
option casemap:none 
 
includelib \masm32\lib\kernel32.lib 
includelib \masm32\lib\user32.lib 
includelib \masm32\lib\gdi32.lib 
 
include \masm32\include\windows.inc 
include \masm32\include\kernel32.inc 
include \masm32\include\user32.inc 
include \masm32\include\gdi32.inc 
WinMain PROTO STDCALL   :DWORD, :DWORD, :DWORD, :DWORD 
WndProc PROTO STDCALL   :DWORD, :DWORD, :DWORD, :DWORD 
 
.data? 
hInstance       dd      ? 
 
 
.data 
ClassName       db      "FirstWindowClass",0 
AppName         db      "FirstWindow",0 
 
.code 
start: 
 
    invoke  GetModuleHandle, NULL 
    mov     hInstance, eax 
    invoke  WinMain, hInstance, NULL, NULL, SW_SHOWNORMAL 
    invoke  ExitProcess, NULL 
 
WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:DWORD 
    LOCAL wc:WNDCLASSEX 
    LOCAL hwnd:DWORD 
    LOCAL msg:MSG 
    mov     wc.cbSize,SIZEOF WNDCLASSEX 
    mov     wc.style,   CS_HREDRAW or CS_VREDRAW 
    mov     wc.lpfnWndProc, OFFSET WndProc 

http://mohandesyar.com


  قسمت دوم- آشنايي با برنامه نويسي اسمبلي ويندوز

 ٣

    mov     wc.cbClsExtra,NULL 
    mov     wc.cbWndExtra,NULL 
    push    hInst 
    pop     wc.hInstance 
    mov     wc.hbrBackground,COLOR_WINDOW 
    mov     wc.lpszMenuName,NULL 
    mov     wc.lpszClassName,OFFSET ClassName 
    invoke  LoadIcon,NULL,IDI_APPLICATION 
    mov     wc.hIcon,   eax 
    mov     wc.hIconSm, eax 
    invoke  LoadCursor,NULL,IDC_ARROW 
    mov     wc.hCursor,eax 
    invoke  RegisterClassEx,    addr wc 
     
    INVOKE CreateWindowEx,NULL,ADDR ClassName,ADDR AppName,\ 
           WS_OVERLAPPEDWINDOW-WS_SIZEBOX-WS_MAXIMIZEBOX,CW_USEDEFAULT,\ 
           CW_USEDEFAULT,400,300,NULL,NULL,\ 
           hInst,NULL 
    mov   hwnd,eax 
    invoke ShowWindow, hwnd,SW_SHOWNORMAL 
    invoke UpdateWindow, hwnd 
    .WHILE TRUE 
        invoke GetMessage, ADDR msg,NULL,0,0 
        .BREAK .IF (!eax) 
                invoke TranslateMessage, ADDR msg 
                invoke DispatchMessage, ADDR msg 
    .ENDW 
    mov     eax,msg.wParam 
    ret 
WinMain endp 
 
WndProc proc hWnd:DWORD, uMsg:DWORD, wParam:DWORD, lParam:DWORD 
mov eax, uMsg 
.IF eax==WM_CREATE 
    invoke  MessageBox, NULL, ADDR AppName, ADDR AppName, NULL 
.ELSEIF eax==WM_DESTROY 
    invoke PostQuitMessage, NULL 
.ELSE 
    invoke DefWindowProc, hWnd, uMsg, wParam, lParam 
.ENDIF 
ret 
WndProc endp 
end start 

  
  .دهد ، داريم اي را نمايش مي كه پنجره،    در ادامه قصد بررسي كامل كد فوق 

  
) هـاي روميـزي   در حالـت برنامـه  (تحت ويندوز   !گويند  ، ويندوز مي   windowsبه سادگي مي توان حدس زد كه چرا به             

هاي داس در يـك    شبيه به برنامه consoleهاي حالت    برنامه.قابل تصور هستند   GUI و   Consoleهاي    دو نوع برنامه در حالت    
 graphical user (GUI داراي هاي مـورد اسـتفاده   اكثر برنامهاما ).  بيتي هستند۳۲هر چند (شوند  پنجره داس مانند اجرا مي

interface (  هـا   ايـن رابـط گرافيكـي توسـط پنجـره         . هـستند كنش با كـاربر      جهت برهم ،   "رابط گرافيكي كاربر  "و يا بقولي
)windows (استتشكيل شده يك پنجره از كنيد  تقريبا هر چيزي را كه در ويندوز مشاهده مي.شوند ايجاد مي.  

  :احل زير بايد صورت گيردمر،    براي ايجاد يك پنجره 
  

 بيتـي ايـن عـدد در حقيقـت     ۳۲در وينـدوز هـاي   . )instance handle (دريافت دستگيره وهله برنامه ايجـاد شـده   •
 .آدرس خطي برنامه در حافظه است

تنها در صورتيكه برنامه شـما ايـن مـورد را نيـاز داشـته باشـد بايـد          . اجباري نيست (دريافت دستورات خط فرمان      •
 ).است قابل انجام  GetCommandLineاينكار با استفاده از تابع . ودپردازش ش

 رجيستر كردن كلاس پنجره •

 ايجاد پنجره •

 نمايش پنجره بر روي صفحه ميزكاري •

 به روز رساني پنجره •

 ايجاد حلقه اي بي پايان جهت پردازش پيغام هاي رسيده به برنامه •

 پردازش پيغام هاي رسيده توسط رويه پنجره •

  .درصورتيكه كاربر پنجره را ببندد  ،شيدن به اجراي برنامه خاتمه بخ •
  
  

   :هاي ايجاد پنجره كلاس
  

كـه  ي يتوان كلاسي دلخواه را ايجاد كرد اما بـراي كنتـرل هـا     والد مي ي   براي پنجره  .   هر پنجره داراي يك كلاس است     
  .نمودتوان استفاده   و غيره ميEDIT ، STATICاز كلاسهاي استانداردي مانند ، د نمي شووالد ايجاد روي فرم بر 
  

  :ساختارها
  :گردد اين تابع به صورت زير تعريف مي.شود  رجيستر ميRegisterClassEx   كلاس پنجره با كمك تابع 

 
ATOM RegisterClassEx( 
CONST WNDCLASSEX *lpwcx // address of structure with class data 
); 
lpwcx: Points to a WNDCLASSEX structure. You must fill the structure with the appropriate class attributes 
before passing it to the function. 

  .باشد مي WNDCLASSEXگر به ساختار  آن يك اشارهدر   تنها پارامتر مورد استفاده 
  

  :تارهاساخو استفاده از تعريف نحوه    در ادامه مرور مجددي خواهيم داشت بر 
  : به شكل زير قابل تعريف استSTRUCTبا رهنمود . است)ها داده (   يك ساختار مجموعه اي از متغيرها

SOMESTRUCTURE   STRUCT 
    dword1 dd ? 
    dword2 dd ? 
    some_word dw ? 
    abyte db ? 
    anotherbyte db ? 
SOMESTRUCTURE   ENDS 

  : داريمر قسمت داده هاي كد   براي مقدار دهي به آن د
Initializedstructure  SOMESTRUCTURE <100,200,10,'A',90h> 

  
  :هاي مقدار دهي اوليه نشده تعريف كنيم    و يا اگر بخواهيم آنرا در قسمت داده

UnInitializedstructure  SOMESTRUCTURE  <> 

  .شوند غيرهاي ساختار با صفر مقدار دهي مي   در اين حالت صرفا حافظه ي مورد نياز پيش بيني شده و مت
  

  :شود    پس از تعريف ساختار براي استفاده از آن به صورت زير عمل مي
Mov  eax, Initializedstructure.some_word 
; eax will hold 10 now 
Inc  UnInitializedstructure.dword1 
; the dword1 of the structure is increased by one 

  
  :  نحوه ذخيره شدن اعضاي ساختار فوق در حافظه به شكل زير است

  
 محتوا موقعيت در حافظه

offset of Initializedstructure 100 (dword, 4 bytes) 
offset of Initializedstructure + 4 200 (dword, 4 bytes) 
offset of Initializedstructure + 8 10 (word, 2 bytes) 
offset of Initializedstructure + 10 65 or 'A' (1 byte) 
offset of Initializedstructure + 11 90h (1 byte) 

  
  

   : WNDCLASSEXساختار 
  
  :تعريف اين ساختار استاندارد در ويندوز به شكل زير است،  MSDNبا توجه به    

typedef struct _WNDCLASSEX { // wc 
   UINT cbSize; 
   UINT style; 
   WNDPROC lpfnWndProc; 
   int cbClsExtra; 
   int cbWndExtra; 
   HANDLE hInstance; HICON hIcon; 
   HCURSOR hCursor; HBRUSH hbrBackground; 
   LPCTSTR lpszMenuName; LPCTSTR lpszClassName; 
   HICON hIconSm; 
} WNDCLASSEX; 

  : در مورد اعضاي آن يتوضيحات
  

 عضو  يحتوض
  :به صورت زير محاسبه مي شود. WNDCLASSEXاندازه ساختار 

mov  ws.cbSize, SIZEOF WNDCLASSEX 

  

cbSize 

 style ).هاي لغزنده و غيره براي مثال داشتن ميله(سبك پنجره را تعيين مي كند 

  )در ادامه بيشتر توضيح داده خواهد شد. (گري به تابع پنجره  اشاره
lpfn = long pointer to function  

بدليل مدل معمـاري   ( وجود ندارند"near" و نزديك   "far"گرهاي دور  تحت ويندوز اشاره  
  .گر وجود دارد  در اينجا تنها اشاره.)تخت

lpfnWndProc 

 cbClsExtra )براي ما اهميتي ندارد. (هاي اضافي كه نياز است تخصيص داده شود بايت

. تخـصيص داده شـود  ، ي نمونه ايجاد شده كلاس پنجره هاي اضافي كه نياز است برا     بايت
 )براي ما اهميتي ندارد(

cbWndExtra 

 بدسـت  GetModuleHandleبـا اسـتفاده از تـابع    . اي به وهله ايجـاد شـده از برنامـه          دستگيره
  .آيد مي

hInstance 

 hIcon .اي به منبع آيكون مورد استفاده براي پنجره دستگيره

 hCursor  . مكان نما مورد استفاده براي پنجرهاي به منبع دستگيره

  :و يا يكي از ثوابت زير! رود اي به برسي كه براي نقاشي پنجره بكار مي دستگيره
COLOR_WINDOW, COLOR_BTNFACE , COLOR_BACKGROUND 

hbrBackground 

را اي مختـوم بـه نـال كـه نـام منبـع منـوي مـورد اسـتفاده بـراي پنجـره                      گري به رشته    اشاره
  .تواند باشد  يك منبع نيز ميIDهمچنين .كند مشخص مي

lpszMenuName 

 lpszClassName .كند اي مختوم به نال كه نام كلاس پنجره را مشخص مي گري به رشته اشاره

 hIconSm .اي به منبع آيكوني كوچك كه مرتبط است به كلاس پنجره دستگيره

  
  
  

  :رجيستر كردن يك كلاس
  
نوبـت    هاي مورد نياز كه پيشتر در مورد آنها كاملا بحث گرديـد ،  مانند اضافه كردن كتابخانه  ،ف اوليه كد    پس از تعاري     

  . استWinMainبه رجيستر كردن كلاس پنجره در رويه 

ر  دExitProcess و تـابع  user32.dll در كتابخانه CreateWindowEx, RegisterWindowClassEx   توابع مورد استفاده مانند 
 و فايـل  user32.libهايي مانند  توان تشخيص داد كه كتابخانه      بنابراين به سادگي مي    . واقع شده است   kernel32.dllكتابخانه  

incمربوطه بايد به برنامه ضميمه شوند .  
  
  .شود مياوليه ايجاد و مقدار دهي پنجره كلاس ،در اين رويه    
  

WinMain proc hInst:DWORD, hPrevInst:DWORD, CmdLine:DWORD, CmdShow:DWORD 
 
ret 
WinMain endp 

  
 ++VC.هـاي وينـدوز   استفاده نخواهد شد و صرفا روشي اسـت متـداول جهـت آغـاز برنامـه     مستقيما    از اين رويه هيچگاه  

  .هيمكند، اما در اينجا بايد اينكار را به صورت دستي انجام د پارامترهاي اين تابع را به صورت خودكار مقدار دهي مي
   hInst  و يـا Instance handle (= module handle)        دسـتگيره اي بـه وهلـه ايجـاد شـده از برنامـه اسـت ،. CmdShow ، 

 ShowWindow دربـاره  MSDNاطلاعات بيشتر در اين مورد را در      (دهد  نمايش پنجره را نمايش مي    پرچمي است كه نحوه     
  .)توان يافت مي

  
  : داريم)WinMainز تابع پيش ا (   در ابتداي كد برنامه

    invoke  GetModuleHandle, NULL 
    mov     hInstance, eax 
    invoke  WinMain, hInstance, NULL, NULL, SW_SHOWNORMAL 
    invoke  ExitProcess, NULL 

  
 كـه پيـشتر نيـز ذكـر شـد رويـه         همـانطور . گردانـد   ، دستگيره وهله ايجاد شده از برنامه را بر مي          getmodulehandleتابع     

  ).خط دوم در كد فوق(دهند    قرار ميeaxاين است كه خروجي خود را در رجيستر ،  ويندوز APIاستاندارد توابع 
  .يابد  با مقادير پيش فرض و همچنين مورد نياز آن فراخواني شده و در آخر كد خاتمه ميWinMain   سپس تابع 

 hPrevInstپـارامتر دوم مـساوي   .نـد مقـدار دهـي شـده ا    Nullترهاي دوم و سوم اين تابع با    همانطور كه دقت كرديد پارام  
 هـر   وبنابراين مقدار اين آرگومان همواره مـساوي صـفر اسـت        .ي پيشيني وجود ندارد     بيتي وهله  ۳۲ در ويندوزهاي    .است

زمـاني كـه تمـام    .  بيتـي اسـت  ۱۶اي  اين آرگومان بـه ارث رسـيده از وينـدوزه   .باشد برنامه در فضاي آدرس خود تنها مي     
در . پارامتر سوم هم مربوط به دريافت دستورات خط فرمـان اسـت  .شدند  در يك فضاي آدرس اجرا مي ها  هاي برنامه   وهله

  :حالت كلي تعريف اين تابع به صورت زير است
  

WinMain proc Inst:HINSTANCE, hPrevInst:HINSTANCE, CmdLine:LPSTR, CmdShow:DWORD  

ــه  ــه شــده WNDCLASSEXاعــضاي ســاختار ابتــدا  ، WinMain   در بدنــه روي  و كــلاس پنجــره رجيــستر  مقــدار دهــي اولي
  :تنها يادآوري يك نكته در مورد آن حائز اهميت است.شود مي

Push  hInst 
pop    wc.hInstance 

     چرا ما در اينجا از دستور زير استفاده نكرديم؟
Mov  wc.hInstance, hInst 

  
بنـابراين  . كنـد  به صورت مستقيم مهيـا نمـي  ، از يك مكان به مكاني ديگر    را   امكان انتقال حافظه     Mov   زيرا دستورالعمل   

  .شود ابتدا بر روي پشته قرار داده شده و سپس در مقصد بازيابي ميداده 
  
ريـف ايـن متغيرهـا بايـد      تع.رود جهـت متغيرهـاي محلـي بكـار مـي         ،  جهت تخصيص حافظه از پـشته        ،   LOCALرهنمود     

از پـشته بـه انـدازه    گويـد كـه     مـي MASMبـه   ، LOCAL wc:WNDCLASSEXبنـابراين  .باشد ، PROCبلافاصله پس از تعريف 
 در كـد بـه سـادگي امكـان پـذير      wc سـپس اسـتفاده از   . ، حافظه اختصاص دهدwc، جهت متغير     WNDCLASSEXساختار  

  .پشته در اينجا نيستاعمال  در مورد نگرانيبوده و نيازي به 
شـوند و خـارج از رويـه قابـل دسترسـي        فقط بايد دقت داشت كه متغيرهاي محلي پس از اتمام كار رويـه تخريـب مـي           

  .نيستند
  
پـارامتر اول ،  . پـذيرد   دو پـارامتر مـي  LoadIconدر اينجـا تـابع   . داردنمـا   و مكـان پنجره نيـاز بـه يـك آيكـون     در ادامه ،      

module handle ، و يـا يـك    (رشته اي كه مساوي نام منبع آيكون اسـت گري است به   اشاره بوده و پارامتر دومID(.  اگـر 
  .توان از آيكون هاي استاندارد انتخاب كرد  در نظر بگيريم ، پارامتر دوم را ميNullپارامتر اول را مساوي 

invoke LoadIcon, NULL, IDI_APPLICATION 
mov  wc.hIcon, eax 
mov  wc.hIconSm, eax 

invoke LoadCursor,NULL,IDC_ARROW 
mov  wc.hCursor,eax 

  
  :  و در پايان كدهاي رجيستر كردن كلاس پنجره داريم

Invoke  RegisterClassEx, ADDR wc 

  :ايجاد پنجره 
  

  .توان از آن استفاده كرد    پس از ثبت كلاس پنجره مي
HWND CreateWindowEx( 
   DWORD dwExStyle, // extended window style 
   LPCTSTR lpClassName, // pointer to registered class name 
   LPCTSTR lpWindowName, // pointer to window name 
   DWORD dwStyle, // window style 
   int x, // horizontal position of window 
   int y, // vertical position of window 
   int nWidth, // window width 
   int nHeight, // window height 
   HWND hWndParent, // handle to parent or owner window 
   HMENU hMenu, // handle to menu, or child-window identifier 
   HINSTANCE hInstance, // handle to application instance 
   LPVOID lpParam // pointer to window-creation data 
); 

  
  

  :   نحوه فراخواني آن در برنامه 
.data 
AppName "FirstWindow",0 
.code 
INVOKE CreateWindowEx,NULL,ADDR ClassName,ADDR AppName,\ 
WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\ 
CW_USEDEFAULT,400,300,NULL,NULL,\ 
hInst,NULL 
mov hwnd, eax 
invoke ShowWindow, hwnd, SW_SHOWNORMAL 
invoke UpdateWindow, hwnd 

  
  .استفاده خواهد شددر انتهاي سطر  \" "يك سطر از كد در دو يا چند سطر ادامه يابد ازنياز باشد تا    اگر 

  . استفاده خواهيم كرد جهت نمايش آن ،) دستگيره اي به پنجره ايجاد شده (CreateWindowEx   از خروجي تابع 
  
  

  :ها حلقه پيغام
  

هـر  .گيـرد  هـا صـورت مـي    هاي ويندوز با اسـتفاده از پيغـام       نحوه ردوبدل كردن اطلاعات بين پنجره، برنامه و ساير برنامه         
يـن  در صورتيكه پيغامي به پنجره برسـد ، در ا .باشد  آن ميبههاي رسيده  اي بي پايان جهت بررسي پيغام  پنجره داراي حلقه  

  .كند رويه پنجره شما را فراخواني مي اين تابع ، .شود  ارسال ميDispatchMessageحلقه بررسي شده و به تابع 

، )كه به معناي درخواسـت خاتمـه بخـشيدن بـه برنامـه اسـت      ( را دريافت كند    WM_QUIT پيغام     ، GetMessageاگر تابع      
 ترجمـه  هـا   ، فشرده شدن كليـدها را بـه پيغـام   TranslateMessageابع  ت.يابد مقدار صفر را بازگشت داده و حلقه خاتمه مي   

  .كند مي
  .گيرد ها مورد استفاده قرار مي    اين حلقه در تمامي برنامه

  
  

  :رويه پنجره
  

  : رويه پنجره بايد همواره به اين شكل باشد.شود به رويه پنجره فرستاده مي،    پيغام 
 
WndProc PROTO STDCALL :DWORD, :DWORD, :DWORD, :DWORD 
 
.code 
 
WndProc proc hWnd:DWORD, uMsg:DWORD, wParam:DWORD, lParam:DWORD 
 
   mov eax, uMsg 
   .IF eax==XXXX 
   .ELSEIF eax==XXXX 
   .ELSE 
       invoke DefWindowProc, hWnd, uMsg, wParam, lParam 
   .ENDIF 
 
ret 
WndProc endp 

  
  .فرستاده شوند  DefWindowProcكه پنجره آنها را پردازش نخواهد كرد بايد به تابع    پيغامهايي 

  
  
بـسيار جالـب بـوده و در مـورد درك     )  OllyDbgبراي مثال توسط ( ايجاد شده    exeبررسي ديس اسمبلي فايل     ،  در پايان     

  .)باشد و نيازي به تكرار آن در اينجا نمي (نحوه ترجمه و مقدار دهي رجيسترها و پشته و غيره در برنامه بسيار مؤثر است
 

  
بايـد دقـت   .    تا اينجا شايد اينطور به نظر آيد كه صرفا ايجاد و نمايش يـك پنجـره سـاده بـسيار پيچيـده و طـولاني اسـت         

هاي مشابه تكـرار خواهـد    باشد كه در تمامي برنامه مي) template(اي از كد فوق صرفا يك قالب    داشت كه قسمت عمده   
  .يدگرد

   يك طرفهمقايسه كد زبان سي و كد اسمبلي يك شرط ساده
  

  
  

 آشنا مي شـوند ، دانـستن   C زبانهاي سطح بالاتري مانند اب،    از آنجائيكه اكثر برنامه نويسان قبل از آشنايي با زبان اسمبلي      
يك نمونه از ايـن مـوارد    از هر مثال  بنابراين در ادامه پس   .نمايد  معادلهاي دستورات اين زبانها به زبان اسمبلي ضروري مي        

  . د گرديدنبررسي خواهنيز 
  

  :   در شكل فوق ، شرط ساده زير به زبان اسمبلي نمايش داده شده است
if (SomeVariable == 0) 
            CallAFunction(); 

  
اينكـار  .شود استفاده مي  EAXفر  با صتساويمقايسه  براي TEST از دستورالعمل ،   همانطور كه در قسمت قبل نيز ذكر شد       

مـساوي صـفر     eaxدر ادامـه اگـر   .گيـرد   صـورت مـي    )ZF ( و تنظيم پـرچم نهـايي      eax بر روي    ANDبا انجام عمليات بيتي     
  . پرش صورت خواهد گرفت نباشد،

  ترسيم متن -۲مثال 
  

بـه همـين جهـت از    .تـشكيل شـده اسـت   ) pixel(هر كاراكتر از تعداد زيادي نقطه . استGUIنوعي شيء    متن در ويندوز    
هـر چنـد   ( برنامه انجام مي شـود  جاري ترسيم متن در ناحيه ، در حالت معمول  .ي ترسيم بجاي نوشتن استفاده گرديد      واژه

  ). برنامه نيز ميسر استترسيم متن خارج از ناحيه 
  
 ۸۰ايش را در حالت معمـول مـي تـوان    صفحه نمابعاد  تحت داس .ترسيم متن در ويندوز با داس به شدت متفاوت است      

هـاي مختلـف بـه اشـتراك گذاشـته شـده و بايـد قـوانيني را          صفحه نمايش بين برنامـه    اما در ويندوز    . در نظر گرفت   ۲۵در  
  . متون مربوط به خود را بر روي يكديگر جاي نويسي نكنند،ها درنظر گرفت تا برنامه

  
ي   انـدازه .كنـد  هـاي در حـال نمـايش آن محـدود مـي      ه را بـه ابعـاد پنجـره      به همين جهت ويندوز ناحيه ترسيمي هر برنام 

  .تغيير اندازه عكس العمل نشان دادبنابراين بايد به صورت پويا به اين . ها توسط كاربران قابل تغيير است پنجره
 بـر روي   كامل و مطلقي  شما همانند داس كنترلبله،!    قبل از ترسيم متن بر روي پنجره بايد از ويندوز كسب اجازه شود          

 در اين حالت شما از ويندوز در مورد ترسيم بر روي محدوده مجاز پنجره خود درخواسـت مجـوز             .ديصفحه نمايش ندار  
 را مشخص كرده و دستگيره اي به آنـرا بـه برنامـه    GDIسپس ويندوز اندازه پنجره ، فونت ، رنگ و ساير خواص   . كنيد مي

بعنـوان مجـوزي بـراي ترسـيم بـر      ) بافت ابزار( device contextسپس از  .)handle to device context (گرداند شما بر مي
  .توان استفاده كرد روي پنجره مي

  
   Device context صرفا سـاختاري داده اي نگهـداري شـده توسـط وينـدوز اسـت       چيست؟. device context   وابـسته بـه 

گرافيكي هـستند ماننـد    تعدادي از خواص موجود در بافت افزار ،. ست اابزاري خاص مانند چاپگر و يا نمايشگر ويديويي      
با داشـتن  .توان تغييرشان داد هستند كه در صورت نياز مينيز مقادير پيش فرضي همچنين داراي   اينها  . رنگ، فونت و غيره   

  .ا نيز مقدار دهي كرد ، تمام خواص آنها رGDIفراخواني توابع ديگر نيازي نيست تا به ازاي هر فرض اين مقادير پيش 
  

هنگاميكه برنامـه اي نيـاز دارد   . توان بعنوان محيطي آماده شما براي استفاده شما توسط ويندوز ياد كرد             از بافت ابزار مي   
هـاي   به صورت معمـول از روش . به بافت ابزار دريافت نمايد)handle (تا ترسيمي را انجام دهد ، بايد در ابتدا دستگيره اي      

  :توان سود جست  اين مقصود ميزير براي
 WM_PAINTدر پاسخ به پيغام   BeginPaintفراخواني  §

 ها در پاسخ به ساير پيغام  GetDCفراخواني  §

  بافت ابزاري دلخواه براي ايجاد CreateDCفراخواني  §

سيستم عامل بازگـشت داد   آنرا به بايد، زار بپس از استفاده از دستگيره بافت ا    . يك مورد را نبايد فراموش كرد     در اينجا      
دسـتگيره اي را كـه بـه سـبب يـك درخواسـت       . و عموما اينكار پس از پايان پردازش يك پيغام رسيده بايـد انجـام شـود              

  . به جهت درخواستي ديگر بازگشت ندهيد،دريافت كرده ايد 
  

مجدد ناحيـه مربـوط بـه خـود     ترسيم  كه اكنون زمان كند اي خاص اعلام مي   به پنجره   WM_PAINT   ويندوز توسط پيغام    
امـا اگـر مـوقعيتي رخ دهـد كـه نيـاز بـه ترسـيم مجـدد          . كند ويندوز محتواي هيچ پنجره اي را ذخيره و بازيابي نمي      . است

بـراي مثـال پوشـانده شـدن قـسمتي از پنجـره       (اي باشد، با پيغامي مناسب اين مطلب را به پنجره اعلام خواهـد كـرد            پنجره
توسط كدي مناسـب ترسـيم مجـدد ناحيـه     ، ي پنجره است تا پس از دريافت اين پيغام       اين وظيفه  ) .اي ديگر  توسط پنجره 

  . همانند مثال قبل بايد در رويه پنجره صورت گيرد،  رسيده WM_PAINTپردازش پيغام . مربوطه را انجام دهد
  
اي در پنجـره   ، كـوچكترين ناحيـه  ز مـستطيل غيرمجـا  . اسـت invalid rectangle  مفهوم ديگري كه بايد با آن آشنا شـد ،   

 را بـه پنجـره ارسـال    WM_PAINTپيغـام  اگر ويندوز چنين ناحيه اي را تشخيص دهـد ،   . است كه نياز به ترسيم مجدد دارد      
صات مستطيل غيرمجـاز را در خـود دارد   تمخ كه paintstructساختار  ، WM_PAINT  پيغامدر جهت پاسخ دهي به .كند  مي

  . قابل دريافت است
  
اگر ايـن اعتبـاردهي صـورت نگيـرد     . را فراخواني كرد   BeginPaintتوان تابع     مي invalid كردن اين ناحيه     Validateبراي     

 WM_PAINTدر غيراينصورت ويندوز مرتبـا پيغـام   .  فراخواني شوندValidateRect و يا DefWindowProcحداقل بايد توابع  
  .را براي پنجره ارسال خواهد كرد

  
  :توان عمل كرد  به صورت زير ميWM_PAINTپاسخ دهي به پيغام    جهت 
 . بايد دريافت شودBeginPaintابتدا دستگيره بافت ابزار توسط  §

 ترسيم ناحيه كلاينت §

   .EndPaintبازگشت دادن دستگيره بافت افزار به سيستم عامل توسط فراخواني تابع  §
  

اينكار به صورت خودكـار توسـط فراخـواني    . مجاز ضرورتي ندارد   لازم به ذكر است كه اعتباردهي صريح مستطيل غير   
BeginPaint    بين دو فراخواني    .  قابل انجام استBeginPaint-Endpaint  توان از انواع توابع    ميGDI    جهـت ترسـيم بـر روي 

  .اي به بافت ابزار دارند تقريبا تمام آنها جهت كاركرد نياز به دستگيره.پنجره استفاده كرد

  :اي برنامه محتو
  

  . را در ميانه پنجره ترسيم نماييم!Win32 assembly is great and easy   در اين برنامه قصد داريم عبارت 
  

  
   ترسيم متن -۱ شكل

  
.386  
.model flat,stdcall  
option casemap:none  

WinMain proto :DWORD,:DWORD,:DWORD,:DWORD  

include \masm32\include\windows.inc  
include \masm32\include\user32.inc  
includelib \masm32\lib\user32.lib  
include \masm32\include\kernel32.inc  
includelib \masm32\lib\kernel32.lib  

.DATA  
ClassName db "SimpleWinClass",0  
AppName  db "Our First Window",0  
OurText  db "Win32 assembly is great and easy!",0  

.DATA?  
hInstance HINSTANCE ?  
CommandLine LPSTR ?  

.CODE  
start:  
    invoke GetModuleHandle, NULL  
    mov    hInstance,eax  
    invoke GetCommandLine 
    mov CommandLine,eax 
    invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT  
    invoke ExitProcess,eax  

WinMain proc hInst:HINSTANCE, hPrevInst:HINSTANCE, CmdLine:LPSTR, CmdShow:DWORD  
    LOCAL wc:WNDCLASSEX  
    LOCAL msg:MSG  
    LOCAL hwnd:HWND  
    mov   wc.cbSize,SIZEOF WNDCLASSEX  
    mov   wc.style, CS_HREDRAW or CS_VREDRAW  
    mov   wc.lpfnWndProc, OFFSET WndProc  

    mov   wc.cbClsExtra,NULL  
    mov   wc.cbWndExtra,NULL  
    push  hInst  
    pop   wc.hInstance  
    mov   wc.hbrBackground,COLOR_WINDOW+1  
    mov   wc.lpszMenuName,NULL  
    mov   wc.lpszClassName,OFFSET ClassName  
    invoke LoadIcon,NULL,IDI_APPLICATION  
    mov   wc.hIcon,eax  
    mov   wc.hIconSm,eax  
    invoke LoadCursor,NULL,IDC_ARROW  
    mov   wc.hCursor,eax  
    invoke RegisterClassEx, addr wc  
    invoke CreateWindowEx,NULL,ADDR ClassName,ADDR AppName,\  
           WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\  
           CW_USEDEFAULT,CW_USEDEFAULT,CW_USEDEFAULT,NULL,NULL,\  
           hInst,NULL  
    mov   hwnd,eax  
    invoke ShowWindow, hwnd,SW_SHOWNORMAL  
    invoke UpdateWindow, hwnd  
        .WHILE TRUE  
                invoke GetMessage, ADDR msg,NULL,0,0  
                .BREAK .IF (!eax)  
                invoke TranslateMessage, ADDR msg  
                invoke DispatchMessage, ADDR msg  
        .ENDW  
        mov     eax,msg.wParam  
        ret  
WinMain endp  

WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM  
    LOCAL hdc:HDC  
    LOCAL ps:PAINTSTRUCT  
    LOCAL rect:RECT  
    .IF uMsg==WM_DESTROY  
        invoke PostQuitMessage,NULL  
    .ELSEIF uMsg==WM_PAINT  
        invoke BeginPaint,hWnd, ADDR ps  
        mov    hdc,eax  
        invoke GetClientRect,hWnd, ADDR rect  
        invoke DrawText, hdc,ADDR OurText,-1, ADDR rect, \  
                DT_SINGLELINE or DT_CENTER or DT_VCENTER  
        invoke EndPaint,hWnd, ADDR ps  
    .ELSE  
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam  
        ret  
    .ENDIF  
    xor   eax, eax  
    ret  
WndProc endp  
end start 

  :بررسي كد فوق
  

  .گرديدتمركز خواهد و نكات مهم ها  دراينجا بيشتر بر روي تفاوت. عمده برنامه فوق همانند مثال قبلي است    
    LOCAL hdc:HDC  
    LOCAL ps:PAINTSTRUCT  
    LOCAL rect:RECT 

  
دستگيره بافـت ابـزار   .ته خواهند شد بكارگرفGDIتوسط توابع ،  WM_PAINT   اينها متغيرهاي محلي هستند كه در قسمت        

 psبطـور معمـول از   . اسـت PAINTSTRUCT ازنـوع    ps.گـردد    ذخيـره مـي     hdc در متغيـر     BeginPaintدريافت شده از تـابع      
 در .گـردد   ارسال شده و توسط ويندوز به صورت خودكار پـر مـي     BeginPaintاين متغير به تابع     . شود مستقيما استفاده نمي  

و بـه صـورت زيـر تعريـف       بـوده RECT از نوع ساختار rect .شود ارسال ميجهت خاتمه ترسيم  EndPaint به تابع psآخر 
  :شود مي

RECT Struct  
    left          LONG ?  
    top          LONG ?  
    right        LONG ?  
    bottom    LONG ?  
RECT ends 

  :يادآوري
  .گيرد قرار مي  y=0 در پايين y=10بنابراين . باشد پ بالا ميمبدا مختصات در اينجا گوشه سمت چ

  

        invoke BeginPaint,hWnd, ADDR ps  
        mov    hdc,eax  
        invoke GetClientRect,hWnd, ADDR rect  
        invoke DrawText, hdc,ADDR OurText,-1, ADDR rect, \  
                  DT_SINGLELINE or DT_CENTER or DT_VCENTER  
        invoke EndPaint,hWnd, ADDR ps  

  
پـارامتر اول دسـتگيره اي   .شـود   با دو پارامتر فراخواني مـي  BeginPaintتابع  ،   WM_PAINT   در جهت پاسخگويي به پيغام      

  پس از فراخواني موفقيت آميـز ايـن تـابع ،   .باشد  مقدار دهي اوليه نشده مي    psاست به پنجره مورد نظر و پارامتر دوم متغير          
 در ادامـه بـراي   .)اي به بافت ابزار و معادل است با دستگيره (گيرد قرار مي  eaxخروجي آن به صورت متداول در رجيستر        

 rectاين ابعاد در متغيـر  .گردد فراخواني مي  GetClientRectتابع ، دريافت ابعاد كلاينت مورد نظر براي ترسيم بر روي آن           
 بـه صـورت زيـر    DrawText تـابع  .گيرد  بعنوان يكي از پارامترها مورد استفاده قرار ميDrawTextقرار گرفته و توسط تابع  

  :شود تعريف مي
  

DrawText  proto hdc:HDC, lpString:DWORD, nCount:DWORD, lpRect:DWORD, uFormat:DWORD  

شده با خواص تعريف شده در دستگيره بافت ابـزار مـورد اسـتفاده ماننـد        تابع فوق متن دلخواهي را در مستطيل دريافت       
  : توضيحات پارامترهاي آن به شرح زير هستند .كند ترسيم مي فونت و غيره ،  رنگ،

hdc : اي به بافت ابزار  دستگيره)handle to device context(  
lpString:  تعريف شده ترسيم شوداي كه بايد توسط تابع در مستطيل  گري به رشته اشاره.  
nCount:  قرار داد  1-اگر رشته مختوم به نال باشد بايد اين مقدار را مساوي . تعداد كاراكترهاي نمايش يافته.  
lpRect:  لازم به ذكر است كـه ايـن مـستطيل    .كه قرار است ترسيم در آن انجام شوداست گري به مستطيلي    اشاره
  .ي آن ميسر نيست بدين معنا كه ترسيم در خارج از محدوده.دباش مي  clipping rectangleاز نوع 

uFormat :    اگر نياز به تركيب آنها باشد از . بيانگر نحوه نمايش متن استorسـه مقـدار بـراي    .گـردد    استفاده مي 
  :آن ممكن است

DT_SINGLELINE : باشد بيانگر ترسيم در يك خط مي.  
DT_CENTER : به صورت افقي (كند يمتن را در وسط صفحه ترسيم م(.  

DT_VCENTER :       به صورت عمودي و بايد با     (كند   متن را در وسط صفحه ترسيم ميDT_SINGLELINE 
  ).بكار رود

  
  !تمام  . منابع تخصيص يافته به سيستم عامل ، فراخواني شود جهت بازگرداندن EndPaintبايد تابع ،    پس از پايان ترسيم 

  
  

  :قمرور نكات مهم برنامه فو
  

  .WM_PAINTپيغام  در جهت پاسخگويي به BeginPaint-EndPaintفراخواني  §

 . انجام دادEndPaint و BeginPaintهر نوع ترسيمي را مي توان در بين  §

تـوان عمـل    روش زيـر مـي  يكـي از دو  ها باشـيد بـه    اگر مايل به ترسيم ناحيه كلاينت مطابق ساير درخواست    §
 :كرد

 .انجام ترسيمات بين اين دو فراخوانيو  GetDC-ReleaseDCفراخواني  §

جهت غيرمعتبركردن كل ناحيه كلاينـت و        UpdateWindow و   InvalidateRectفراخواني   §
سـپس انجـام ترسـيم در    . را به برنامه ارسـال كنـد     WM_PAINTوادار كردن ويندوز تا پيغام      

  .  قابل انجام استWM_PAINTقسمت 
  
  
  

http://mohandesyar.com


  قسمت دوم- آشنايي با برنامه نويسي اسمبلي ويندوز

 ١٨

  دو طرفهشرط مقايسه كد زبان سي و كد اسمبلي يك 
  
  

  
  

در كـد  . انتخـاب خواهـد شـد   ،    در اين نوع شرط ها درصورت برآورده نشدن شرط اصلي ، راه حـل ديگـر ارائـه شـده            
  . قابل ملاحظه هستند، پس از مقايسه انجام شدههاي شرطي و غير شرطي  پرش، )شكل فوق(اسمبلي معادل 

   به روشي ديگر ترسيم متن-۳مثال 
  

  .رنگ ، قلم انتخابي و اندازه آن ، بيشتر بحث كنيمر اين مثال قصد داريم تا در مورد خواص متن ترسيم شده مانند    د
  

  :مقدمه
  

هـر مؤلفـه ي آن بـازه اي از صـفر تـا       .)R=red, G=Green, B=Blue ( اسـت RGBسيستم رنگ ويندوز بر مبناي مقادير     
  اسـت و رنـگ سـفيد خـالص مـساوي اسـت بـا         255,0,0 خـالص معـادل      براي مثـال رنـگ قرمـز      .  )بايتيك  ( دارد   ۲۵۵

  . ساده نيستRGBفرماييد كار كردن با سيستم رنگ  همانطور كه ملاحظه مي . 255,255,255
و   hdcتمـام آنهـا نيـاز بـه     . مهيـا هـستند   SetBkColorو  SetTextColor توابـع ،    براي تنظيم رنگ متن و رنـگ زمينـه آن     

  : به صورت زير تعريف شده استRGB ساختار . داردRGB بيتي ۳۲مقدار 
RGB_value struct  
    unused   db 0  
    blue       db ?  
    green     db ?  
    red        db ?  
RGB_value ends 

  
ساختار بهتـر اسـت   جهت سهولت كار با اين . بكار نرفته و مساوي صفر بايد قرار گيرد  ،  بايت اول      لازم به ذكر است كه      

  :ماكروي زير تعريف شود
RGB macro red,green,blue  
    xor    eax,eax  
    mov  ah,blue  
    shl    eax,8  
    mov  ah,green  
    mov  al,red  
endm 

  . قرار گرفته و به برنامه ملحق شود  inc   بهتر است اين ماكرو در يك فايل 
  

 تنهـا يـك   CreateFontIndirect تـابع  . اسـتفاده كـرد   CreateFontIndirect ويـا   CreateFont وان از ت ـ    براي ايجاد قلم مي   
ايـن تـابع قابليـت انعطـاف بيـشتري داشـته و بـراي        .باشـد   ميLOGFONTگري به ساختار    پارامتر ورودي داشته و آن اشاره     

از آنجائيكه مـا در برنامـه تنهـا يـك قلـم را      . بودقلم تعويض شود مناسب خواهد ، حالتي كه نياز است در برنامه به دفعات   
پس از فراخـواني ايـن تـابع ، خروجـي آن دسـتگيره اي بـه قلـم        .گردد  استفاده مي  CreateFontايجاد خواهيم كرد از تابع      

فونـت  از ايـن  پس از آن هر تابع متني كه مورد استفاده قرار گيرد  .باشد  مورد استفاده جهت بكارگيري در زمينه ابزار ، مي        
  .جويد  بهره مي كه همراه زمينه ابزار شده است ،

  :كد برنامه
  
  

  
 

ترسيم متن  -۲شكل   
 

 
.386  
.model flat,stdcall  
option casemap:none  

WinMain proto :DWORD,:DWORD,:DWORD,:DWORD  

include \masm32\include\windows.inc  
include \masm32\include\user32.inc  
include \masm32\include\kernel32.inc  
include \masm32\include\gdi32.inc  
includelib \masm32\lib\user32.lib  
includelib \masm32\lib\kernel32.lib  
includelib \masm32\lib\gdi32.lib  

RGB macro red,green,blue  
        xor eax,eax  
        mov ah,blue  
        shl eax,8  
        mov ah,green  
        mov al,red  
endm  

.data  
ClassName db "SimpleWinClass",0  
AppName  db "Our First Window",0  
TestString  db "Win32 assembly is great and easy!",0  
FontName db "script",0  

.data?  
hInstance HINSTANCE ?  
CommandLine LPSTR ?  

.code  
 start:  
    invoke GetModuleHandle, NULL  
    mov    hInstance,eax  
    invoke GetCommandLine 
    mov CommandLine,eax 
    invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT  
    invoke ExitProcess,eax  

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:DWORD  
    LOCAL wc:WNDCLASSEX  
    LOCAL msg:MSG  
    LOCAL hwnd:HWND  
    mov   wc.cbSize,SIZEOF WNDCLASSEX  
    mov   wc.style, CS_HREDRAW or CS_VREDRAW  
    mov   wc.lpfnWndProc, OFFSET WndProc  
    mov   wc.cbClsExtra,NULL  
    mov   wc.cbWndExtra,NULL  
    push  hInst  
    pop   wc.hInstance  
    mov   wc.hbrBackground,COLOR_WINDOW+1  
    mov   wc.lpszMenuName,NULL  
    mov   wc.lpszClassName,OFFSET ClassName  
    invoke LoadIcon,NULL,IDI_APPLICATION  
    mov   wc.hIcon,eax  
    mov   wc.hIconSm,eax  
    invoke LoadCursor,NULL,IDC_ARROW  
    mov   wc.hCursor,eax  
    invoke RegisterClassEx, addr wc  
    invoke CreateWindowEx,NULL,ADDR ClassName,ADDR AppName,\  
           WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\  
           CW_USEDEFAULT,CW_USEDEFAULT,CW_USEDEFAULT,NULL,NULL,\  
           hInst,NULL  
    mov   hwnd,eax  
    invoke ShowWindow, hwnd,SW_SHOWNORMAL  
    invoke UpdateWindow, hwnd  
    .WHILE TRUE  
                invoke GetMessage, ADDR msg,NULL,0,0  
                .BREAK .IF (!eax)  
                invoke TranslateMessage, ADDR msg  
                invoke DispatchMessage, ADDR msg  
    .ENDW  
    mov     eax,msg.wParam  
    ret  
WinMain endp  

WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM  
    LOCAL hdc:HDC  
    LOCAL ps:PAINTSTRUCT  
    LOCAL hfont:HFONT  

    .IF uMsg==WM_DESTROY  
        invoke PostQuitMessage,NULL  
    .ELSEIF uMsg==WM_PAINT  
        invoke BeginPaint,hWnd, ADDR ps  
        mov    hdc,eax  
        invoke CreateFont,24,16,0,0,400,0,0,0,OEM_CHARSET,\  
                                       OUT_DEFAULT_PRECIS,CLIP_DEFAULT_PRECIS,\  
                                       DEFAULT_QUALITY,DEFAULT_PITCH or FF_SCRIPT,\  
                                       ADDR FontName  
        invoke SelectObject, hdc, eax  
        mov    hfont,eax  
        RGB    200,200,50  
        invoke SetTextColor,hdc,eax  
        RGB    0,0,255  
        invoke SetBkColor,hdc,eax  
        invoke TextOut,hdc,0,0,ADDR TestString,SIZEOF TestString  
        invoke SelectObject,hdc, hfont  
        invoke EndPaint,hWnd, ADDR ps  

    .ELSE  
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam  
        ret  
    .ENDIF  
    xor    eax,eax  
    ret  
WndProc endp  

end start 
  

  :بررسي كد فوق
  

        invoke CreateFont,24,16,0,0,400,0,0,0,OEM_CHARSET,\  
                                       OUT_DEFAULT_PRECIS,CLIP_DEFAULT_PRECIS,\  
                                       DEFAULT_QUALITY,DEFAULT_PITCH or FF_SCRIPT,\  
                                       ADDR FontName  

  
توضيحات پارامترهـاي آن  .  ، قلمي منطقي را كه مطابق پارامترهاي آن درخواست شده، ايجاد مي كند CreateFont   تابع  

  :به شرح زير هستند
  

CreateFont proto nHeight:DWORD,\  
                            nWidth:DWORD,\  
                            nEscapement:DWORD,\  
                            nOrientation:DWORD,\  
                            nWeight:DWORD,\  
                            cItalic:DWORD,\  
                            cUnderline:DWORD,\  
                            cStrikeOut:DWORD,\  
                            cCharSet:DWORD,\  
                            cOutputPrecision:DWORD,\  
                            cClipPrecision:DWORD,\  
                            cQuality:DWORD,\  
                            cPitchAndFamily:DWORD,\  
                            lpFacename:DWORD  

  
nHeight : ارتفاع پيش فرض استفاده خواهد شد ، اگر مساوي صفر قرار گيرد. بيانگر ارتفاع حروف است.  
nWidth :   ويندوز طول و عـرض حـروف را متناسـب بـا هـم            ،  اگر مساوي صفر قرار گيرد    .عرض دلخواه حروف

  .كند تنظيم مي
nEscapement :            حالـت معمـول   .كنـد  زاويه ترسيم كاراكتر بعدي را نسبت به قبلي بر حسب درجه مشخص مـي 

  .صفر است
nOrientation : كند درجه چرخش ترسيم حروف را بر حسب درجه مشخص مي.  

nWeight : مقادير زير براي اين پارامتر مجاز هستند.كند ضخامت هر كاراكتر را مشخص مي:  

 FW_DONTCARE      equ  0  
FW_THIN                  equ  100  
FW_EXTRALIGHT    equ  200  
FW_ULTRALIGHT    equ  200  
FW_LIGHT                equ  300  
FW_NORMAL           equ  400  
FW_REGULAR         equ  400  
FW_MEDIUM            equ  500  
FW_SEMIBOLD        equ  600  
FW_DEMIBOLD        equ  600  
FW_BOLD                 equ  700  
FW_EXTRABOLD     equ  800  
FW_ULTRABOLD     equ  800  
FW_HEAVY              equ  900  
FW_BLACK               equ  900 

  
cItalic :  براي حالت مايل هر عددي غير از صفر بايد مشخص شود. صفر بايد وارد شودمعمولبراي حالت.  

cUnderline : براي حالت معمول صفر بايد وارد شود .  
cStrikeOut : براي حالت معمول صفر بايد وارد شود .  
cCharSet : اگــر مــساويOEM_CHARSET ــه سيــستم عامــل   قــرار گيــرد ــسته ب را انتخــاب وينــدوز حالــت واب
  .خواهدكرد

cOutputPrecision :             دقت ترسيم فونت را مشخص مـي كنـد مقـدارOUT_DEFAULT_PRECIS     حالـت پـيش 
  .فرض را انتخاب خواهد كرد

cClipPrecision :        براي حالتي كه قسمتي از فونت خارج مستطيل ترسـيمي قـرار     (دقت برش را مشخص مي كند
  . كند ش فرض را مشخص مي حالت پيCLIP_DEFAULT_PRECIS.  )گيرد

cQuality :   سـه مقـدار   . كنـد  كيفيت خروجي را مـشخص مـيDEFAULT_QUALITY, PROOF_QUALITY ,  

DRAFT_QUALITY    ميزان دقتي كه بر اين مبنا .توان مشخص كرد     را براي آن ميGDI كيفيـت   بايد بخرج دهد تا
  .شود مشخص مي  ،متن ترسيمي مطابق قلم واقعي باشد

cPitchAndFamily :   تـوان از   براي تركيـب حالتهـاي مختلـف آنهـا مـي     . تراكم حروف و خانواده ترسيمي آنor 
  . استفاده كرد

lpFacename : رشته اي مختوم به نال معرف نام قلم مورد استفاده.  
  

  :ه كد داريم در ادام).MSDNمانند ( مراجعه كرد Win32 API reference   بديهي است كه براي توضيحات بيشتر بايد به 

        invoke SelectObject, hdc, eax  
        mov    hfont,eax  

 GDIجهت استفاده ساير توابـع    ، آنرا به بافت ابزارSelectObjectتابع فونت منطقي ايجاد شده،         پس از دريافت دستگيره     
  . دهد انتساب مي

        RGB     200,200,50  
        invoke  SetTextColor,hdc,eax  
        RGB     0,0,255  
        invoke  SetBkColor,hdc,eax 

  
  : در ادامه . كه پيشتر آنرا ايجاد كرديم جهت تنظيم رنگ دو تابع فوق استفاده شده استRGB   از ماكرو 

  
invoke TextOut,hdc,0,0,ADDR TestString,SIZEOF TestString  

  
  :پس از آن.  استفاده گرديده است) آن رنگ وفونت (ترسيم متني با مشخصاتي تنظيم شده براي TextOutاز تابع    

invoke  SelectObject,hdc, hfont  

  .انجام عمليات ترسيم متن ، بايد اشياء تخصيص داده شده مجددا به سيستم عامل بازگشت داده شوند   پس از 

  هاي چندگانه لي شرطمقايسه كد زبان سي و كد اسمب
  
  

  
  
  

هـاي شـرطي و    بـه كمـك پـرش   بـه زبـان اسـمبلي را       ح بـالاتر  و  زبانهاي سـط    else-ifنحوه پياده سازي    ،     در تصوير فوق    
  .توان مشاهده كرد مي، غيرشرطي 

  دريافت ورودي از صفحه كليد -۴مثال 
  

هـاي وينـدوز بايـد آنـرا بـين       تمام برنامـه   هر رايانه وجود دارد ،     از آنجائيكه بطور معمول تنها يك صفحه كليد به همراه           
بـه   ، اطلاعات كليدهاي فشرده شده بر روي صفحه كليـد  مسؤوليت ارسال عهده دار ويندوز . خويش به اشتراك بگذارند 

رنگي متفـاوت بـا    ، اي كه تمركز را در اختيار دارد  نوار عنوان پنجره  .  ورودي را در اختيار دارد     تمركزكه  است  اي   پنجره
  . ها دارد ساير پنجره

براي مثال اگر بـه صـفحه كليـد بعنـوان مجموعـه اي از كليـدها نگـاه         . هاي صفحه كليد وجود دارند        دو نوع اصلي پيغام   
اي كه تمركـز را در اختيـار دارد    به پنجره را WM_KEYDOWNويندوز پيغام   ، در اين حالت با فشرده شدن كليدي   شود ، 

 بـه پنجـره   WM_KEYUPپيغـام    ،بـا رهـا شـدن كليـد     همچنـين   .، تا آنرا از فشرده شدن كليد مطلع كند ل خواهد كرد  ارسا
 پيغـام  ، بـا فـشردن شـدن هـر كليـدي      ، و يا اگر به صفحه كليد بعنـوان دسـتگاه ورود حـروف نگـاه شـود      .گردد  ارسال مي 

WM_CHARي مثال از فشرده شدن كليد  تا آنرا براشود  به پنجره داراي تمركز فرستاده مي"a "مطلع سازد .  
 WM_CHAR پيغــام  بــهTranslateMessageتــابع   توســط WM_KEYUP و WM_KEYDOWNهــاي  در حقيقــت پيغــام    

در اغلـب حـالات از   .  رويه پنجره مختار است كه هر سه پيغام را پردازش كند و يـا تنهـا يكـي از آنهـا را    .شوند ترجمه مي 
 ترجمـه  WM_CHARتوان صرفنظر كرد زيرا همانطور كه عنوان شد به          مي WM_KEYUP و   WM_KEYDOWNهاي    پيغام

  . متمركز خواهيم شدWM_CHARما در اين مثال بر روي . شوند مي
  
  

  :كد برنامه 
  

.386  

.model flat,stdcall  
option casemap:none  

WinMain proto :DWORD,:DWORD,:DWORD,:DWORD  

include \masm32\include\windows.inc  
include \masm32\include\user32.inc  
include \masm32\include\kernel32.inc  
include \masm32\include\gdi32.inc  
includelib \masm32\lib\user32.lib  
includelib \masm32\lib\kernel32.lib  
includelib \masm32\lib\gdi32.lib  

.data  
ClassName db "SimpleWinClass",0  
AppName  db "Our First Window",0  
char WPARAM 20h                         ; the character the program receives from keyboard  

.data?  
hInstance HINSTANCE ?  
CommandLine LPSTR ?  

.code  
start:  
    invoke GetModuleHandle, NULL  
    mov    hInstance,eax  
    invoke GetCommandLine 
    mov CommandLine,eax 
    invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT  
    invoke ExitProcess,eax  

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:DWORD  
    LOCAL wc:WNDCLASSEX  
    LOCAL msg:MSG  
    LOCAL hwnd:HWND  
    mov   wc.cbSize,SIZEOF WNDCLASSEX  
    mov   wc.style, CS_HREDRAW or CS_VREDRAW  
    mov   wc.lpfnWndProc, OFFSET WndProc  
    mov   wc.cbClsExtra,NULL  
    mov   wc.cbWndExtra,NULL  
    push  hInst  
    pop   wc.hInstance  
    mov   wc.hbrBackground,COLOR_WINDOW+1  
    mov   wc.lpszMenuName,NULL  
    mov   wc.lpszClassName,OFFSET ClassName  
    invoke LoadIcon,NULL,IDI_APPLICATION  
    mov   wc.hIcon,eax  
    mov   wc.hIconSm,eax  
    invoke LoadCursor,NULL,IDC_ARROW  
    mov   wc.hCursor,eax  
    invoke RegisterClassEx, addr wc  
    invoke CreateWindowEx,NULL,ADDR ClassName,ADDR AppName,\  
           WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\  
           CW_USEDEFAULT,CW_USEDEFAULT,CW_USEDEFAULT,NULL,NULL,\  
           hInst,NULL  
    mov   hwnd,eax  
    invoke ShowWindow, hwnd,SW_SHOWNORMAL  
    invoke UpdateWindow, hwnd  
    .WHILE TRUE  
                invoke GetMessage, ADDR msg,NULL,0,0  
                .BREAK .IF (!eax)  
                invoke TranslateMessage, ADDR msg  
                invoke DispatchMessage, ADDR msg  
        .ENDW  
    mov     eax,msg.wParam  
    ret  
WinMain endp  

WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM  
    LOCAL hdc:HDC  
    LOCAL ps:PAINTSTRUCT  

    .IF uMsg==WM_DESTROY  
        invoke PostQuitMessage,NULL  
    .ELSEIF uMsg==WM_CHAR  
        push wParam  
        pop  char  
        invoke InvalidateRect, hWnd,NULL,TRUE  

    .ELSEIF uMsg==WM_PAINT  
        invoke BeginPaint,hWnd, ADDR ps  
        mov    hdc,eax  
        invoke TextOut,hdc,0,0,ADDR char,1  
        invoke EndPaint,hWnd, ADDR ps  
    .ELSE  
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam  
        ret  
    .ENDIF  
    xor    eax,eax  
    ret  
WndProc endp  
end start 

  
  :بررسي كد فوق

  
char WPARAM 20h                         ; the character the program receives from keyboard  

  
از آنجائيكـه ايـن كـاراكتر بـه     . كنـد     اين متغيري است كه حرف دريافت شده از صفحه كليـد را در خـود ذخيـره مـي           

همچنين مقدار اوليـه  .  انتخاب كرديمWPARAM نوع آنرا ، ولت جهت سه،شود  رويه پنجره ارسال مي   WPARAMپارامتر  
20h   است را به آن نسبت داده ايم ، زيرا در اولين باري كه پنجره ناحيـه كلاينـت خـود را بـه روز در           " فاصله " كه معادل با 
ا نمـايش  به همين جهـت قـصد داريـم در ابتـدا صـرفا يـك فاصـله ر       .ورودي از طرف صفحه كليد وجود ندارد ،  آورد   مي

  : در ادامه داريم. دهيم
  

    .ELSEIF uMsg==WM_CHAR  
        push wParam  
        pop  char  
        invoke InvalidateRect, hWnd,NULL,TRUE  

 charكـاراكتر در متغيـر    .  رسيده ، قطعه كد فوق به رويه پنجره اضافه گرديده اسـت        WM_CHAR   جهت پردازش پيغام    
همانطور كه در مثال قبل نيز ذكـر گرديـد ، بـا فراخـواني ايـن      . شود   فراخواني ميInvalidateRectته و سپس تابع  قرارگرف

.  را بـه رويـه پنجـره ارسـال كنـد     WM_PAINT تـا پيغـام   گرديد مستطيل پنجره غيرمعتبر شده و ويندوز وادار خواهد       تابع ، 
  :تعريف اين تابع به صورت زير است

InvalidateRect  proto  hWnd:HWND,\  
                                  lpRect:DWORD,\  
                                  bErase:DWORD  

  
    lpRect اگـر بجـاي آن   . گـردد  گري به مستطيلي است كه غيرمجاز اعلام مي  اشارهNull     وارد شـود كـل ناحيـه 

  . غيرمعتبر اعلام خواهد شدمربوطه 

http://mohandesyar.com


  قسمت دوم- آشنايي با برنامه نويسي اسمبلي ويندوز

 ٢٩

    bErase         اگـر ايـن پـرچم بـه     . به ويندوز اعلام مي كند كه آيا نياز است زمينه نيـز پـاك گـردد يـا خيـرTRUE 
  . ، زمينه مستطيل درخواستي را پاك خواهد كردBeginPaintتنظيم شود ويندوز با فراخواني تابع 

  
بـا فراخـواني    ( نيـز انجـام داد  WM_CHARتـوان در پاسـخ بـه پيغـام            انجام عمليات ترسيم كاراكتر بر روي پنجره را مـي         

GetDC   و ReleaseDC(  .     كليـدي فـشرده شـده باشـد ،     شود كه  اما از آنجائيكه اين قسمت از كد تنها زماني فراخواني مي
بنـابراين بهتـر اسـت ترسـيم كـاراكتر در      . هنگام نياز به ترسيم مجدد كل ناحيه كلاينت پنجـره ، ايـن امـر رخ نخواهـد داد      

  :در ادامه داريم. صورت گيرد WM_PAINTپاسخ به پيغام 
invoke TextOut,hdc,0,0,ADDR char,1  

  
بنابراين كد قـرار گرفتـه   . گردد  به پنجره ارسال ميWM_PAINTپيغام شود ،    فراخواني ميInvalidateRect   زمانيكه تابع   

سـمت چـپ ، بـالاي پنجـره ترسـيم      در فـشرده شـده   با فشردن هـر كليـدي ، كـاراكتر     . گردد   در قسمت ياد شده اجرا مي     
 حـداكثر رسـاندن آن نيـز ، كـاراكتر بـر روي صـفحه حفـظ           يـا  همچنين با تغيير سايز پنجـره و يـا بـه حـداقل و             . گردد مي
  .   قرار گرفته استWM_PAINT مي آن در قسمت مربوط بهيترسكد گردد زيرا  مي

  كيبيترهاي  مقايسه كد زبان سي و كد اسمبلي شرط
  
  

  
  
 همانطور كه در شـكل نيـز مـشخص اسـت تبـديل ايـن       .گردند هاي تركيبي تشكيل مي    هاي واقعي عموما از شرط     برنامه    

  . باشد ميكد به زبان اسمبلي نسبتا پيچيده 

  طريق ماوس دريافت ورودي از -۵مثال 
  
  

  . كلاينت پنجره ترسيم نماييم   در اين برنامه قصد داريم در محل كليك چپ ماوس، متني را در ناحيه 
  

 را رديـابي كـرده و   )هاي راست و چپ و دوبار كليـك كـردن   مانند حركت ماوس ، كليك (هاي ماوس     ويندوز فعاليت 
هاي صفحه كليد كه صرفا به پنجره اي ارسـال   بر خلاف پيغام. كند هايي را متناسب با آنها به پنجره مربوطه ارسال مي         پيغام

گـر آن بـر روي آنهـا     هـايي كـه اشـاره    هاي مربوط به ماوس به تمـامي پنجـره    پيغام ركز را در اختيار دارد،د كه تم نگرد مي
مـستطيل  توانند مربوط به ناحيـه خـارج از    حتي اين پيغامها مي. )خواه اين پنجره فعال باشد يا خير     (شود است ، فرستاده مي   

توان صرفا پيغامهاي مرتبط به ناحيه كلاينـت پنجـره را دريافـت و     ميخوشبختانه در اكثر اوقات . كلاينت پنجره نيز باشند    
  .  و از مابقي پيغامها صرفنظر نمودپردازش كرد

  
  :   به ازاي هر دكمه ي ماوس ، دو پيغام وجود دارد

WM_LBUTTONDOWN , WM_RBUTTONDOWN and WM_LBUTTONUP ,  WM_RBUTTONUP  
  
  :غامهاي زير نيز وجود دارنداگر ماوسي داراي سه دكمه بود ، پي   

WM_MBUTTONDOWN and WM_MBUTTONUP  
  

  :شود كند پيغام زير از طرف ويندوز صادر مي عبور مي   هنگاميكه ماوس از روي ناحيه كلاينت 
WM_MOUSEMOVE  

  
  

ــام    ــامي پيغ ــا هنگ ــره تنه ــك پنج ــاوس          ي ــدن م ــك ش ــار كلي ــه دوب ــوط ب ــاي مرب  WM_LBUTTONDBCLK or(ه

WM_RBUTTONDBCLK ( ــي ــت م ــت      را درياف ــرچم حال ــره داراي پ ــلاس پنج ــه ك ــد ك ــدCS_DBLCLKSكن در . باش
  .كند هاي ماوس را دريافت مي فشرده شدن ويا رها شدن دكمهغيراينصورت صرفا پيغامهاي 

  
حـاوي  آن  low wordدر ايـن پـارامتر،   .باشـد  مختـصات مـاوس مـي    حاوي lParam   براي تمامي پيغامهاي رسيده، پارامتر 

  wParamهمچنـين پـارامتر   . )نـسبت بـه گوشـه سـمت چـپ بـالا       ( اسـت yحاوي مختـصات  آن  high word و xتصات مخ
  .  استctrl و shiftهاي  هاي ماوس و همچنين دكمه حالات دكمه بيانگر

  :كد برنامه 
  

.386  

.model flat,stdcall  
option casemap:none  

WinMain proto :DWORD,:DWORD,:DWORD,:DWORD  

include \masm32\include\windows.inc  
include \masm32\include\user32.inc  
include \masm32\include\kernel32.inc  
include \masm32\include\gdi32.inc  
includelib \masm32\lib\user32.lib  
includelib \masm32\lib\kernel32.lib  
includelib \masm32\lib\gdi32.lib  

.data  
ClassName db "SimpleWinClass",0  
AppName  db "Our First Window",0  
MouseClick db 0         ; 0=no click yet  

.data?  
hInstance HINSTANCE ?  
CommandLine LPSTR ?  
hitpoint POINT <>  

.code  
start:  
    invoke GetModuleHandle, NULL  
    mov    hInstance,eax  
    invoke GetCommandLine 
    mov CommandLine,eax  
    invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT  
    invoke ExitProcess,eax  

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:DWORD  
    LOCAL wc:WNDCLASSEX  
    LOCAL msg:MSG  
    LOCAL hwnd:HWND  
    mov   wc.cbSize,SIZEOF WNDCLASSEX  
    mov   wc.style, CS_HREDRAW or CS_VREDRAW  
    mov   wc.lpfnWndProc, OFFSET WndProc  
    mov   wc.cbClsExtra,NULL  
    mov   wc.cbWndExtra,NULL  
    push  hInst  
    pop   wc.hInstance  
    mov   wc.hbrBackground,COLOR_WINDOW+1  
    mov   wc.lpszMenuName,NULL  
    mov   wc.lpszClassName,OFFSET ClassName  
    invoke LoadIcon,NULL,IDI_APPLICATION  
    mov   wc.hIcon,eax  
    mov   wc.hIconSm,eax  
    invoke LoadCursor,NULL,IDC_ARROW  
    mov   wc.hCursor,eax  
    invoke RegisterClassEx, addr wc  
    invoke CreateWindowEx,NULL,ADDR ClassName,ADDR AppName,\  
           WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\  
           CW_USEDEFAULT,CW_USEDEFAULT,CW_USEDEFAULT,NULL,NULL,\  

           hInst,NULL  
    mov   hwnd,eax  
    invoke ShowWindow, hwnd,SW_SHOWNORMAL  
    invoke UpdateWindow, hwnd  
    .WHILE TRUE  
                invoke GetMessage, ADDR msg,NULL,0,0  
                .BREAK .IF (!eax)  
                invoke DispatchMessage, ADDR msg  
    .ENDW  
    mov     eax,msg.wParam  
    ret  
WinMain endp  

WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM  
    LOCAL hdc:HDC  
    LOCAL ps:PAINTSTRUCT  

    .IF uMsg==WM_DESTROY  
        invoke PostQuitMessage,NULL  
    .ELSEIF uMsg==WM_LBUTTONDOWN  
        mov eax,lParam  
        and eax,0FFFFh  
        mov hitpoint.x,eax  
        mov eax,lParam  
        shr eax,16  
        mov hitpoint.y,eax  
        mov MouseClick,TRUE  
        invoke InvalidateRect,hWnd,NULL,TRUE  
    .ELSEIF uMsg==WM_PAINT  
        invoke BeginPaint,hWnd, ADDR ps  
        mov    hdc,eax  
        .IF MouseClick  
            invoke lstrlen,ADDR AppName  
            invoke TextOut,hdc,hitpoint.x,hitpoint.y,ADDR AppName,eax  
        .ENDIF  
        invoke EndPaint,hWnd, ADDR ps  
    .ELSE  
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam  
        ret  
    .ENDIF  
    xor    eax,eax  
    ret  
WndProc endp  
end start  

  
  

  :بررسي كد فوق
  

.ELSEIF uMsg==WM_LBUTTONDOWN  
        mov eax,lParam  
        and eax,0FFFFh  
        mov hitpoint.x,eax  
        mov eax,lParam  
        shr eax,16  
        mov hitpoint.y,eax  

        mov MouseClick,TRUE  
        invoke InvalidateRect,hWnd,NULL,TRUE  

  
افـت شـود،    دري WM_LBUTTONDOWNهنگاميكـه پيغـام     . ماند  منتظر فشردن شدن دكمه چپ ماوس مي      ،     رويه پنجره   

ايـن مختـصات در سـاختار زيـر     . اسـت  در مستطيل ناحيـه كلاينـت پنجـره       صات محل كليك    ت حاوي مخ  lParamپارامتر  
  :شود ذخيره مي

POINT STRUCT  
    x   dd ?  
    y   dd ?  
POINT ENDS  

  
مـاوس بـر روي ناحيـه    يك كليك چپ بدين معنا كه حداقل  (كند  تنظيم ميTRUE را به MouseClick   و همچنين پرچم    

  :در ادامه داريم. ) از آن در قسمت ترسيمي برنامه استفاده خواهيم كرد ،كلاينت پنجره انجام شده است
        Mov  eax,lParam  
        and  eax,0FFFFh  
        mov  hitpoint.x,eax  

  
 بيتي هـستند ، نيـاز اسـت    POINT ، ۳۲ ساختار و اعضاي است lParam پارامتر low word ، مساوي x   از آنجائيكه مختص 

  :پس از آن. )ANDتوسط  ( آن، صفر شودhitpoint.x ، high wordتا پيش از ذخيره سازي در 
        shr  eax,16  
        mov hitpoint.y,eax  

  
 مربـوط  low wordت در ، بايد آنرا توسط عمليات شـيف  استlParam پارامتر high word ، مساوي y   از آنجائيكه مختص 

  . ، قرار دادhitpoint.yقبل از ذخيره سازي در ،   eaxبه 
 سـبب فراخـواني اجبـاري آن    InvalidateRectكـه فراخـواني تـابع      ( WM_PAINTهاي رسـيده      پيغام   در قسمت پردازش    

  .گردد ، متني در محل كليك ماوس ترسيم مي )گردد مي
  

        .IF MouseClick  
            invoke  lstrlen,ADDR AppName  
            invoke  TextOut, hdc, hitpoint.x, hitpoint.y, ADDR AppName, eax  
        .ENDIF  

  
، عمليــات ترســيم  بــودن آن TRUE  بررســي شــده و در صــورت MouseClickپــرچم   ، ابتــداWM_PAINTدر قــسمت    

  .گيرد صورت مي
  .شود استفاده مي ، TextOut دريافت طول رشته مورد استفاده در تابع  جهتlstrlen   از تابع 

  عملگرهاي منطقيهايي با  شرطمقايسه كد زبان سي و كد اسمبلي 
  
  

  
  
  

 .كننـد  شرط ميـسر مـي  يك قالب شرط هاي تركيبي را در امكان بيان    زبانهاي سطح بالاتر با كمك عملگرهاي منطقي ،     
، عملگرها را با عملگرهـاي بيتـي اشـتباه نكنيـد    نوع اين  ( هستندOR و   ANDن نوع شروط عمدتا     دو عملگر بكار رفته در اي     

   . )شود   مشخص مي&& با AND ، اين نوع Cبراي مثال در زبان 
  

از دو پـرش شـرطي بـه يـك برچـسب تـشكيل       همانگونه كـه در شـكل نيـز مـشخص شـده اسـت ،        ،     كد اسمبلي معادل    
   . گرديده است

  
بـراي پيـاده   .شود  مشخص مي||اين عملگر در زبان سي با . )شكل بعد  ( است  ORها ، استفاده از      يگر اين نوع شرط      نوع د 

، به ازاي هر شرط يك پرش شرطي و در نهايت استفاده از يك پـرش غيرشـرطي   سازي اين نوع شرط ها به زبان اسمبلي         
  .لازم است

  : تر و سريعتري مانند شكل زير را براي اين منظور ارائه داده اندهاي بهينه  روش++VCكامپايلرهايي مانند 

   كار با منوها-۶مثال 
  

  .   در اين برنامه طرز اضافه كردن منوها به پنجره برنامه را خواهيم آموخت
  

ط آن كـاربر در  توس ـ. دهد ارائه مي، هاي يك برنامه    و ليستي را از توانايي        منو يكي از مهمترين اجزاي يك برنامه است       
از آنجائيكه هـدف منوهـا اسـتفاده    . ابتداي كار نيازي نخواهد داشت تا حتما راهنماي همراه برنامه را به دقت مطالعه نمايد          

دو آيـتم  .سريع و ساده از برنامه توسط كاربر است ، بايد در هنگام ايجاد آنها استانداردهايي را در اين بـاره مراعـات نمـود    
 editساير آيتم هاي خـود را بـين   .مربوط به راهنماي برنامه باشد،  بوده و آخرين آيتم Edit و Fileامه بايد اوليه در منوي برن   

  . و راهنماي برنامه قرار دهيد
 ,dialog box, string table, icon, bitmap انواع مختلفي از منـابع ماننـد   .شود محسوب مي)  resource(   منو ، نوعي منبع 

menuجود هستند و غيره مو .  
 ، linkسـپس ايـن فايـل و سـورس برنامـه در حـين مرحلـه         .شـوند  ايجاد مي  rc   منابع عموما در فايلي جداگانه و با پسوند        

  .ها خواهد بود حاوي كد و ريسورس، فايل اجرايي نهايي . شوند تركيب مي
بـديهي اسـت كـه آمـاده سـازي دسـتي آنهـا         البتـه  .توان آماده كـرد     فايل ريسورس را توسط هر نوع ويرايشگر متني مي       

هـاي   برنامـه  و غيـره،   ++Visual C++, Borland Cبـه همـراه كامپايلرهـايي ماننـد     . كاري وقت گيـر و طاقـت فرسـا اسـت    
resource editorنيز موجود هستند  .  

  :)Cشبيه به تعريف يك ساختار در زبان  (شود    يك فايل منبع منو به صورت زير تعريف مي
MyMenu  MENU  
{  
    [menu list here]  
}  

  
  .!)جهت سهولت كار برنامه نويسان پاسكال (نيز استفاده كرد   end و beginتوان از  ها مي    بجاي استفاده از براكت
روش تعريف آن نيـز بـه صـورت    . شوند  مشخص مي ) منوي جهنده ( POPUP و يا    MENUITEM   ليست منوها با عبارات     

  :زير است
MENUITEM   "&text", ID [,options]  

  
 IDاز .گـردد   بـراي آن مـي  underline  پيش از هر حرفي ، سبب قـرار گـرفتن يـك      &   لازم به ذكر است كه قرار گرفتن        

   ،)بـه برنامـه مـي رسـد    ، هنگاميكـه پيغـام مربـوط بـه آن      (جهت مشخص كردن آيتمي كه بر روي آن كليك شده اسـت             
 مـشخص شـده انـد    options كـه بـه صـورت    ي مـوارد . بكارگرفته شده بايد منحصربفرد باشـد IDبنابراين . شود استفاده مي 

  : اختياري بوده و به شرح زير هستند 

• GRAYED :      آيتم هايي كه به اين صورت مشخص شوند به صورت غيرفعـال و خاكـستري نمـايش داده
 .گردند  نميWM_COMMANDاين آيتم ها سبب ارسال پيغام . شوند مي

• INACTIVE :        سـبب ارسـال پيغـام     (اين آيتم به صورت معمول نمايش داده شده و غيرفعال خواهـد بـود
WM_COMMANDشود  نمي(. 

• MENUBREAK :  در خطي جديد ظاهر خواهند شد، اين آيتم و موارد بعدي . 

• HELP:  اين آيتم و موارد پس از آن راست چين خواهند شد.  
  

 INACTIVE and تنهـا دقـت داشـته باشـيد كـه      . اسـتفاده كـرد  orمـي تـوان از      براي تركيب حالت هـاي مختلـف فـوق    

GRAYED توان با هم بكار برد  را نمي.  
  :   قالب زير را دارندPOPUP    موارد 

  POPUP  "&text" [,options]  
{  
  [menu list]  
} 

  
  .آيد ها پديد مي نمايانگر ليستي از آيتم، منويي جهنده     با اينكار

  
كه سبب ترسيم خطي افقـي در ليـست    ( نيز وجود داردMENUITEM SEPARATOR با نام MENUITEM نوع ويژه اي از   

  . )گردد منوها مي
  

  : دو راه براي انتساب منو به برنامه وجود دارد.باشد    مرحله بعد ، استفاده از فايل منبع تعريف شده در برنامه مي
 ، مربـوط بـه سـاختار     lpszMenuNameتـوان بـه عـضو        اشته باشيد، آنـرا مـي      د FirstMenuاگر منويي به نام      )الف

WNDCLASSEXانتساب داد .  

.DATA  

MenuName  db "FirstMenu",0  

…........................  
…........................  
.CODE  

…........................  
mov   wc.lpszMenuName, OFFSET MenuName  
…........................  

   :CreateWindowEx تابع menu handleارسال آن به پارامتر ) ب

.DATA  

MenuName  db "FirstMenu",0  
hMenu HMENU ?  

...........................  

...........................  

.CODE  

...........................  
invoke LoadMenu, hInst, OFFSET MenuName  
mov   hMenu, eax  
invoke CreateWindowEx,NULL,OFFSET ClsName,\  
            OFFSET Caption, WS_OVERLAPPEDWINDOW,\  
            CW_USEDEFAULT,CW_USEDEFAULT,\  
            CW_USEDEFAULT,CW_USEDEFAULT,\  
            NULL,\  
           hMenu,\  
            hInst,\  
            NULL\  
...........................  

 
   تفاوت اين دو روش چيست؟:سوال

هر پنجره اي كه بر مبناي اين كلاس ايجـاد شـود   .منوي ايجاد شده ، منوي پيش فرض پنجره خواهد شد، در حالت اول      
       . منويي يكسان خواهد داشت

روش دوم اسـتفاده  خواهيد كه هر پنجره ايجاد شده از يك كلاس پنجره ، منوهايي متفاوت داشته باشـد بايـد از         اگر مي 
  . كنيد

  
  
  

 توسـط وينـدوز بـه رويـه پنجـره      WM_COMMAND پيغـام     در ادامه لازم به ذكر است كه با انتخاب هر گزينه اي از منو،           
  .  گزينه انتخابي استID حاوي wParam مربوط به low wordدر اين حالت .گردد برنامه ارسال مي

  :كد برنامه 
  

.386  

.model flat,stdcall  
option casemap:none  

WinMain proto :DWORD,:DWORD,:DWORD,:DWORD  

include \masm32\include\windows.inc  
include \masm32\include\user32.inc  
include \masm32\include\kernel32.inc  
includelib \masm32\lib\user32.lib  
includelib \masm32\lib\kernel32.lib  

.data  
ClassName db "SimpleWinClass",0  
AppName  db "Our First Window",0  
MenuName db "FirstMenu",0                ; The name of our menu in the resource file.  
Test_string db "You selected Test menu item",0  
Hello_string db "Hello, my friend",0  
Goodbye_string db "See you again, bye",0  

.data?  
hInstance HINSTANCE ?  
CommandLine LPSTR ?  

.const  
IDM_TEST equ 1                    ; Menu IDs  
IDM_HELLO equ 2  
IDM_GOODBYE equ 3  
IDM_EXIT equ 4  

.code  
start:  
    invoke GetModuleHandle, NULL  
    mov    hInstance,eax  
    invoke GetCommandLine 
    mov CommandLine,eax  
    invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT  
    invoke ExitProcess,eax  

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:DWORD  
    LOCAL wc:WNDCLASSEX  
    LOCAL msg:MSG  
    LOCAL hwnd:HWND  
    mov   wc.cbSize,SIZEOF WNDCLASSEX  
    mov   wc.style, CS_HREDRAW or CS_VREDRAW  
    mov   wc.lpfnWndProc, OFFSET WndProc  
    mov   wc.cbClsExtra,NULL  
    mov   wc.cbWndExtra,NULL  
    push  hInst  
    pop   wc.hInstance  
    mov   wc.hbrBackground,COLOR_WINDOW+1  
    mov   wc.lpszMenuName,OFFSET MenuName        ; Put our menu name here  
    mov   wc.lpszClassName,OFFSET ClassName  
    invoke LoadIcon,NULL,IDI_APPLICATION  

    mov   wc.hIcon,eax  
    mov   wc.hIconSm,eax  
    invoke LoadCursor,NULL,IDC_ARROW  
    mov   wc.hCursor,eax  
    invoke RegisterClassEx, addr wc  
    invoke CreateWindowEx,NULL,ADDR ClassName,ADDR AppName,\  
           WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\  
           CW_USEDEFAULT,CW_USEDEFAULT,CW_USEDEFAULT,NULL,NULL,\  
           hInst,NULL  
    mov   hwnd,eax  
    invoke ShowWindow, hwnd,SW_SHOWNORMAL  
    invoke UpdateWindow, hwnd  
    .WHILE TRUE  
                invoke GetMessage, ADDR msg,NULL,0,0  
                .BREAK .IF (!eax)  
                invoke DispatchMessage, ADDR msg  
    .ENDW  
    mov     eax,msg.wParam  
    ret  
WinMain endp  

WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM  
    .IF uMsg==WM_DESTROY  
        invoke PostQuitMessage,NULL  
    .ELSEIF uMsg==WM_COMMAND  
        mov eax,wParam  
        .IF ax==IDM_TEST  
            invoke MessageBox,NULL,ADDR Test_string,OFFSET AppName,MB_OK  
        .ELSEIF ax==IDM_HELLO  
            invoke MessageBox, NULL,ADDR Hello_string, OFFSET AppName,MB_OK  
        .ELSEIF ax==IDM_GOODBYE  
            invoke MessageBox,NULL,ADDR Goodbye_string, OFFSET AppName, MB_OK  
        .ELSE  
            invoke DestroyWindow,hWnd  
        .ENDIF  
    .ELSE  
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam  
        ret  
    .ENDIF  
    xor    eax,eax  
    ret  
WndProc endp  
end start  
 

**************************************************************************************************************************  

Menu.rc  
**************************************************************************************************************************  

#define IDM_TEST 1  
#define IDM_HELLO 2  
#define IDM_GOODBYE 3  
#define IDM_EXIT 4  

FirstMenu MENU  
{  
 POPUP "&PopUp"  
        {  
         MENUITEM "&Say Hello",IDM_HELLO  

http://mohandesyar.com


  قسمت دوم- آشنايي با برنامه نويسي اسمبلي ويندوز

 ٤٢

         MENUITEM "Say &GoodBye", IDM_GOODBYE  
         MENUITEM SEPARATOR  
         MENUITEM "E&xit",IDM_EXIT  
        }  
 MENUITEM "&Test", IDM_TEST  
}  

  
  :بررسي كد فوق

  
  :   در ابتدا بررسي فايل منبع آن 

#define  IDM_TEST            1                /* equal to IDM_TEST equ 1*/  
#define  IDM_HELLO          2  
#define  IDM_GOODBYE    3  
#define  IDM_EXIT              4  

  
تـوان   بديهي است كه از هر عددي مي . هاي مختلف تعريف شده اند     ريف گزينه  هاي مورد استفاده براي تعا     ID   در اينجا   

  :در ادامه . دناستفاده كرد تا زمانيكه در اين مجموعه منحصربفرد باش
FirstMenu MENU  

  :سپس .   تعريف منو با واژه كليدي مربوطه   
 POPUP "&PopUp"  
        {  
         MENUITEM "&Say Hello",IDM_HELLO  
         MENUITEM "Say &GoodBye", IDM_GOODBYE  
         MENUITEM SEPARATOR  
         MENUITEM "E&xit",IDM_EXIT  
        }  

  
  : در ادامه .   در بالا منويي جهنده با عضو سوم جدا كننده تعريف شده است

 MENUITEM  "&Test", IDM_TEST  
  

  .كند عريف ميدر منوي اصلي ت   نوار منو را 
  

  :   در ادامه كد برنامه را بررسي خواهيم كرد
MenuName db "FirstMenu",0                ; The name of our menu in the resource file.  
Test_string db "You selected Test menu item",0  
Hello_string db "Hello, my friend",0  
Goodbye_string db "See you again, bye",0  

  
. دقيقا مـشخص نمـود  را بايد نام منوي مورد استفاده ،    از آنجائيكه در يك فايل منبع مي توان چندين منو را تعريف كرد     

  :در ادامه. اند   نيز تعريف شدهكند، اي خاص كليك مي  استفاده هنگاميكه كاربر بر روي گزينهسپس پيغامهاي مورد

IDM_TEST equ 1                    ; Menu IDs  
IDM_HELLO equ 2  
IDM_GOODBYE equ 3  
IDM_EXIT equ 4  

  
اين مقادير بايد بـا مقـدارهاي تعريـف شـده در فايـل منبـع دقيقـا        . در رويه پنجرهها جهت استفاده   هاي گزينهID   تعريف  

  :سپس . يكسان باشد

    .ELSEIF uMsg==WM_COMMAND  
        mov eax,wParam  
        .IF ax==IDM_TEST  
            invoke MessageBox,NULL,ADDR Test_string,OFFSET AppName,MB_OK  
        .ELSEIF ax==IDM_HELLO  
            invoke MessageBox, NULL,ADDR Hello_string, OFFSET AppName,MB_OK  
        .ELSEIF ax==IDM_GOODBYE  
            invoke MessageBox,NULL,ADDR Goodbye_string, OFFSET AppName, MB_OK  
        .ELSE  
            invoke DestroyWindow,hWnd  
        .ENDIF  

  
همانطور كه پيشتر نيز ذكر گرديد در ايـن حالـت   .  را پردازش خواهيم كردWM_COMMAND   در رويه پنجره پيغامهاي  

low word مربوط به wParam حاوي IDيعني مقدار قرار گرفته در  ( گزينه انتخابي استax(.  
  .شود ه شدن پنجره مي سبب بستDestroyWindow   اگر كاربر گزينه خروج را انتخاب كرده باشد ، تابع 

  
  :به صورت زير اصلاح شودبايد   در آخر اگر بخواهيم از روش دوم نمايش منو استفاده كنيم ، كد برنامه 

  
.data?  
hInstance HINSTANCE ?  
CommandLine LPSTR ?  
hMenu HMENU ?                    ; handle of our menu  
 
 
        invoke LoadMenu, hInst, OFFSET MenuName  
        mov    hMenu,eax  
        INVOKE CreateWindowEx,NULL,ADDR ClassName,ADDR AppName,\  
           WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\  
           CW_USEDEFAULT,CW_USEDEFAULT,CW_USEDEFAULT,NULL,hMenu,\  
           hInst,NULL  

  
تـابع   ، CreateWindowExپـيش از فراخـواني   . اي بـه منـوي مـا ايجـاد خواهـد كـرد        دسـتگيره HMENUري از نـوع       متغي

LoadMenu،   براي استفاده در تابع   دستگيره لازم راCreateWindowEx ، برخواهد گرداند .  

  :نحوه كامپايل فايل ريسورس و الحاق آن به فايل اجرايي نهايي
  

بـا  .  قـرار گرفتـه انـد   C:\masm32\prog در آدرس p7.rc و فايل منبع آن به نام   p7.asm برنامه به نام     asm   براي مثال فايل    
  :شود فايل اجرايي برنامه ساخته مي  فايل ريسورس كامپايل شده و نهايتا،اجراي خطوط زير در خط فرمان 

C:\masm32\bin\ml.exe /c /coff /Cp p7.asm 

C:\masm32\bin\rc.exe p7.rc 

C:\masm32\bin\link.exe /SUBSYSTEM:WINDOWS /LIBPATH:C:/masm32/prog/ p7.obj   p7.res 

pause 

  
  

  
  

   نمايي از اجراي برنامه كار با منوها–۳شكل 
  
  

پس از اجـراي  .  حاوي چهار خط فوق ايجاد كنيدbat   بهتر است جهت سهولت اجراي دستورهاي فوق ، فايلي با پسوند        
  . ي پس از ديگري اجرا خواهند شدمراحل يك، آن 

  عملگر منطقيبيش از دو  شرط بايك مقايسه كد زبان سي و كد اسمبلي 
  

  
  

  : توان مشاهده نمود     حالت سريعتر و بهينه تر را نيز در شكل زير مي

  Child Window Controls  كار با-۷مثال 
  

. هـاي مـا تعريـف كـرده اسـت      ش تعريف شده را جهت استفاده در برنامههاي از پي هاي متعددي از پنجره       ويندوز كلاس 
 نـام  child window controls ، بـه همـين جهـت    شـود   استفاده ميdialog boxهاي  در اغلب اوقات از آنها بعنوان كامپوننت

يكـه حالـت آنهـا    هاي ماوس و صفحه كليد مربوط بـه خـود را پـردازش كـرده و هنگام      پيغامchild windowيك .اند گرفته
  . سازد ي والد را باخبر مي تغيير كرد پنجره

بـراي  .باشـد   و غيـره مـي  listbox ، checkbox ، radio button  ها ، هاي از پيش تعريف شده شامل دكمه    مثالهايي از پنجره
لازم . ايجـاد كـرد  CreateWindow or CreateWindowEx  توابعها نياز است تا در ابتدا آنها را بوسيله استفاده از اين كنترل

زيرا اين امر توسـط وينـدوز پـيش تـر صـورت      . به ذكر است كه نيازي به رجيستر كردن كلاس اين پنجره ها وجود ندارد   
اي  خواهيـد دكمـه   براي مثال اگر مي.  نامي از پيش تعريف شده انتخاب شود"بايد"پارامتر نام كلاس در اينجا . گرفته است 

 ساير پارامترهـاي  . معرفي نماييدCreateWindowExدر تابع   ، را بعنوان پارامتر نام كلاس پنجره   buttonرا ايجاد كنيد بايد     
هـايي كـه بـر روي فـرم قـرار        بايد در بين ساير كنتـرل IDاين . باشد  كنترل ميIDي والد و  اي به پنجره مورد نياز ، دستگيره  

  . گيرند يكتا باشد مي
كنتـرل سـاير   . را بـه آن ارسـال خواهـد كـرد     WM_CREATEشود پيغـام     نجره والد ايجاد مي      هنگاميكه كنترلي بر روي پ    

 مربـوط بـه   low word كنتـرل توسـط    ID. كنـد   بـه پنجـره والـد ارسـال مـي     WM_COMMANDهـاي خـود را بوسـيله     پيغـام 
wParam كد باخبرسازي توسط ،high word مربوط به wParam دستگيره به پنجره توسط  وlParamاين حـين ارسـال    در 

  .  ويندوز مراجعه كرد APIهر كنترل داراي كدباخبري سازي متفاوتي است و بهتر است در اين مورد به مرجع . گردد مي
  . ارسال كندSendMessageهاي لازم را توسط فراخواني تابع  پيغام، ها  تواند به كنترل    همچنين پنجره والد نيز مي

  
  

  :كد برنامه 
  
هنگاميكـه بـر روي دكمـه    . اي را ايجاد خواهيم كرد كه داراي يك كنترل اديت و يك دكمـه اسـت   ال پنجره در اين مث     

همچنـين از منوهـا نيـز در    .  حاوي متن نوشته در كنترل اديت نمايش داده خواهد شـد message boxشود يك  كليك مي
  .گردد اين برنامه استفاده مي

  
.386  
.model flat,stdcall  
option casemap:none  

WinMain proto :DWORD,:DWORD,:DWORD,:DWORD  

include \masm32\include\windows.inc  
include \masm32\include\user32.inc  
include \masm32\include\kernel32.inc  
includelib \masm32\lib\user32.lib  
includelib \masm32\lib\kernel32.lib  

.data  
ClassName db "SimpleWinClass",0  
AppName  db "Our First Window",0  
MenuName db "FirstMenu",0  
ButtonClassName db "button",0  
ButtonText db "My First Button",0  
EditClassName db "edit",0  
TestString db "Wow! I'm in an edit box now",0  

.data?  
hInstance HINSTANCE ?  
CommandLine LPSTR ?  
hwndButton HWND ?  
hwndEdit HWND ?  
buffer db 512 dup(?)                    ; buffer to store the text retrieved from the edit box  

.const  
ButtonID equ 1                                ; The control ID of the button control  
EditID equ 2                                    ; The control ID of the edit control  
IDM_HELLO equ 1  
IDM_CLEAR equ 2  
IDM_GETTEXT equ 3  
IDM_EXIT equ 4  

.code  
start:  
    invoke GetModuleHandle, NULL  
    mov    hInstance,eax  
    invoke GetCommandLine 
    mov CommandLine,eax  
    invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT  
    invoke ExitProcess,eax  

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:DWORD  
    LOCAL wc:WNDCLASSEX  
    LOCAL msg:MSG  
    LOCAL hwnd:HWND  
    mov   wc.cbSize,SIZEOF WNDCLASSEX  
    mov   wc.style, CS_HREDRAW or CS_VREDRAW  
    mov   wc.lpfnWndProc, OFFSET WndProc  
    mov   wc.cbClsExtra,NULL  
    mov   wc.cbWndExtra,NULL  
    push  hInst  
    pop   wc.hInstance  
    mov   wc.hbrBackground,COLOR_BTNFACE+1  
    mov   wc.lpszMenuName,OFFSET MenuName  
    mov   wc.lpszClassName,OFFSET ClassName  
    invoke LoadIcon,NULL,IDI_APPLICATION  
    mov   wc.hIcon,eax  
    mov   wc.hIconSm,eax  
    invoke LoadCursor,NULL,IDC_ARROW  
    mov   wc.hCursor,eax  
    invoke RegisterClassEx, addr wc  
    invoke CreateWindowEx,WS_EX_CLIENTEDGE,ADDR ClassName, \  

                        ADDR AppName, WS_OVERLAPPEDWINDOW,\  
                        CW_USEDEFAULT, CW_USEDEFAULT,\  
                        300,200,NULL,NULL, hInst,NULL  
    mov   hwnd,eax  
    invoke ShowWindow, hwnd,SW_SHOWNORMAL  
    invoke UpdateWindow, hwnd  
    .WHILE TRUE  
        invoke GetMessage, ADDR msg,NULL,0,0  
        .BREAK .IF (!eax)  
        invoke TranslateMessage, ADDR msg  
        invoke DispatchMessage, ADDR msg  
    .ENDW  
    mov     eax,msg.wParam  
    ret  
WinMain endp  

WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM  
    .IF uMsg==WM_DESTROY  
        invoke PostQuitMessage,NULL  
    .ELSEIF uMsg==WM_CREATE  
        invoke CreateWindowEx,WS_EX_CLIENTEDGE, ADDR EditClassName,NULL,\  
                        WS_CHILD or WS_VISIBLE or WS_BORDER or ES_LEFT or\  
                        ES_AUTOHSCROLL,\  
                        50,35,200,25,hWnd,8,hInstance,NULL  
        mov  hwndEdit,eax  
        invoke SetFocus, hwndEdit  
        invoke CreateWindowEx,NULL, ADDR ButtonClassName,ADDR ButtonText,\  
                        WS_CHILD or WS_VISIBLE or BS_DEFPUSHBUTTON,\  
                        75,70,140,25,hWnd,ButtonID,hInstance,NULL  
        mov  hwndButton,eax  
    .ELSEIF uMsg==WM_COMMAND  
        mov eax,wParam  
        .IF lParam==0  
            .IF ax==IDM_HELLO  
                invoke SetWindowText,hwndEdit,ADDR TestString  
            .ELSEIF ax==IDM_CLEAR  
                invoke SetWindowText,hwndEdit,NULL  
            .ELSEIF  ax==IDM_GETTEXT  
                invoke GetWindowText,hwndEdit,ADDR buffer,512  
                invoke MessageBox,NULL,ADDR buffer,ADDR AppName,MB_OK  
            .ELSE  
                invoke DestroyWindow,hWnd  
            .ENDIF  
        .ELSE  
            .IF ax==ButtonID  
                shr eax,16  
                .IF ax==BN_CLICKED  
                    invoke SendMessage,hWnd,WM_COMMAND,IDM_GETTEXT,0  
                .ENDIF  
            .ENDIF  
        .ENDIF  
    .ELSE  
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam  
        ret  
    .ENDIF  
     xor    eax,eax  
    ret  
WndProc endp  
end start  

  :فايل منبع مورد استفاده 

#define IDM_HELLO 1 

#define IDM_CLEAR 2 

#define IDM_GETTEXT 3 

#define IDM_EXIT 4 

 

FirstMenu MENU 

{ 

 POPUP "&Test Controls" 

        { 

         MENUITEM "Say Hello",IDM_HELLO 

         MENUITEM "Clear Edit Box",IDM_CLEAR 

         MENUITEM "Get Text", IDM_GETTEXT 

         MENUITEM SEPARATOR 

         MENUITEM "E&xit",IDM_EXIT 

        } 

} 

  
  :بررسي كد فوق

  
.ELSEIF uMsg==WM_CREATE  
        invoke CreateWindowEx,WS_EX_CLIENTEDGE, \  
                        ADDR EditClassName,NULL,\  
                        WS_CHILD or WS_VISIBLE or WS_BORDER or ES_LEFT\  
                        or ES_AUTOHSCROLL,\  
                        50,35,200,25,hWnd,EditID,hInstance,NULL  
        mov  hwndEdit,eax  
        invoke SetFocus, hwndEdit  
        invoke CreateWindowEx,NULL, ADDR ButtonClassName,\  
                        ADDR ButtonText,\  
                        WS_CHILD or WS_VISIBLE or BS_DEFPUSHBUTTON,\  
                        75,70,140,25,hWnd,ButtonID,hInstance,NULL  
        mov  hwndButton,eax 

  

http://mohandesyar.com


  قسمت دوم- آشنايي با برنامه نويسي اسمبلي ويندوز

 ٥٠

  .شوند  ايجاد ميWM_CREATEها در حين پردازش پيغام     كنترل
نـام  . گـردد  نمـايش سـطحي فرورفتـه مـي     سـبب  WS_EX_CLIENTEDGE با حالـت     CreateWindowEx   فراخواني تابع   

  . باشد  براي دكمه ميbutton براي اديت باكس و editكلاس هر كنترل از پيش تعريف شده بوده و معادل 
بـراي مثـال   . لاوه بر حالتهاي ممكن براي پنجـره والـد ، داراي حالتهـاي مخـصوص بـه خـود نيـز مـي باشـد                   هر كنترل ع  

تـوان    براي مطالعه بيشتر اين مـوارد مـي  .شوند  شروع مي_ES و حالتهاي اديت باكس با _BSدكمه با اضافي يك  حالتهاي  
  .  مراجعه كرد MSDNبه 

اين مـورد مـشكلي را ايجـاد نخواهـد كـرد      . بكار گرفته شده است،  گيره منو    كنترل بجاي دست   ID   لازم به ذكر است كه      
  . توانند داراي منو باشند ها نمي زيرا كنترل

  . هاي آتي در يك متغير ذخيره كرد توان دستگيره آنرا براي استفاده مي،    بعد از ايجاد هر كنترلي 
در اين حالـت بلافاصـله پـس از نمـايش پنجـره ،      . فاده گرديد  جهت انتقال تمركز به اديت باكس است       SetFocus   از تابع   

  .توان به سادگي از اديت باكسي داراي فوكوس و انتخاب شده ، كمك گرفت مي
  . كند ي والد ارتباط برقرار مي  با پنجرهWM_COMMAND هر كنترل توسط پيغام ،   همانطوركه ذكر شد

    .ELSEIF uMsg==WM_COMMAND  
        mov eax,wParam  
        .IF lParam==0  

ي والـد كمـك     جهت بر قراري ارتبـاط بـا پنجـره   WM_COMMAND   از مثالهاي قبل بخاطر داريم كه منوها نيز از پيغام      
  توان بين اين دو مورد تميز قائل شد؟ اما چگونه مي.گيرند مي

  :   پاسخ در جدول زير مشخص شده است
  

 Low word of wParam High word of wParam lParam 
Menu Menu ID 0 0 
Control Control ID Notification code Child Window Handle 

  
  

تـوان اسـتفاده كـرد زيـرا       نمـي wParamاز .  مساوي صفر بود پيغام از طرف منو صـادر شـده اسـت       lParam   بنابراين اگر   
  :امه در اد. وجود دارد، احتمال صفر بودن كد باخبرسازي كنترل 

            .IF ax==IDM_HELLO  
                invoke SetWindowText,hwndEdit,ADDR TestString  
            .ELSEIF ax==IDM_CLEAR  
                invoke SetWindowText,hwndEdit,NULL  
            .ELSEIF  ax==IDM_GETTEXT  
                invoke GetWindowText,hwndEdit,ADDR buffer,512  
                invoke MessageBox,NULL,ADDR buffer,ADDR AppName,MB_OK  

تـا    بـود تـوان اسـتفاده كـرد و اگـر نيـاز        ميSetWindowTextهاي متني درون اديت باكس از تابع           براي قرار دادن رشته   
  . كمك گرفت بعنوان ورودي nullمي توان از پارامتر ، محتواي آنرا خالي نمود 

از آن بـراي تغييـر عنـوان يـك     .  ويندوز محـسوب مـي شـود   API يكي از توابع با كاربرد عمومي       SetWindowText   تابع  
  .پنجره و يا متن روي يك دكمه مي توان استفاده كرد

  :ر ادامه د.   كمك گرفتGetWindowTextتوان از تابع     براي دريافت متن درون يك اديت باكس مي
  

  .IF ax==ButtonID  
                shr eax,16  
                .IF ax==BN_CLICKED  
                    invoke SendMessage,hWnd,WM_COMMAND,IDM_GETTEXT,0  
                .ENDIF  
            .ENDIF  

  
 دكمـه  ID بـا  wParam مربوط به low wordدر ابتدا .  اجرا خواهد شددكمه برنامه كليك شود   كد بالا هنگاميكه بر روي 

  . گردد مقايسه مي، ) BN_CLICKED( با كد باخبر سازي دكمه wParam مربوط به high wordسپس . شود مقايسه مي

  
  

  ها  تصويري از اجراي برنامه كار با كنترل-۴شكل 
  

جتناب شود ، بهتر است بـا پيغـامي هماننـد پيغـام منـوي مربوطـه كـه بـه            براي اينكه از دوباره نويسي كد در اين قسمت ا       
  . اينكار انجام پذير استSendMessageبا كمك تابع .  را انجام دادMessage Box، عمليات نمايش شود پنجره ارسال مي

  
 ورودي از صفحه كليـد    فراخواني نشود ، عمليات دريافتTranslateMessage   در پايان لازم به ذكر است كه اگر تابع         

  . نجام نخواهد شداو ترجمه آن به متن قابل استفاده در اديت باكس 

   منطقيهاي عملگرتركيبات مختلفي ازمقايسه كد زبان سي و كد اسمبلي يك شرط با 
  

  
  

بهينـه  كـد اسـمبلي ارائـه شـده     . نماييـد   را ملاحظـه مـي  andدر تصوير فوق نحوه پياده سازي يك شرط مركب از چندين           
  .باشد سازي شده نيز مي

  ها بعنوان پنجره اصلي  استفاده از ديالوگ باكس-۸مثال 
  

از يـك  ، ورودي ) فوكـوس (    اگر به مثال قبلي بيشتر دقت كرده باشيد حتما متوجه شـده ايـد كـه امكـان تغييـر تمركـز                  
اه انتخاب يك كنتـرل در ايـن حالـت ، انتخـاب كنتـرل        تنها ر .  ميسر نيست   tabكنترل به كنترل ديگر توسط فشردن كليد        

  . توسط كليك ماوس است
ديالوگ باكس چيزي بيـشتر از يـك پنجـره    . ها بيشتر آشنا شويم     پيش از توضيحات بيشتر نياز است تا با ديالوگ باكس         

دير داخلـي اسـتفاده   هـا از يـك م ـ   همچنين وينـدوز بـراي كـار بـا ديـالوگ بـاكس      . ها نيست متداول جهت كاربر با كنترل 
  . ها و امثال آن شود به اين صورت ديگر نيازي نيست تا برنامه نويس درگير مديريت انتقال تمركز به ساير كنترل. كند مي

بهتـر اسـت بـراي تعريـف اينگونـه فايلهـاي       . شود انجام مي) ريسورس(يك ديالوگ باكس توسط فايلهاي منبع          تعريف  
هـاي يـك    لازم به ذكر است كه تمامي ريسورس. يا در بسياري از كامپايلرها استفاده شود    ريسورس از ابزارهاي بصري مه    

  . گيرند برنامه در آخر در يك فايل منبع قرار مي
) modalبـرخلاف حالـت   ( modelessدر حالـت   .modal and modeless:    دو نـوع كلـي ديـالوگ بـاكس وجـود دارد      

 نيـز  modalپنجره هاي . )MS-Wordبراي مثال پنجره جستجو در  ( ها نيز انتقال داد  توان تمركز ورودي را به ساير پنجره       مي
تمركـز   امكـان تغييـر     ،application modalدر حالت  . application modal and system modal: داراي دو حالت هستند 

امـا در  . تـوان انتخـاب كـرد    سـادگي مـي  هـا را بـه    هاي ساير برنامه  پنجرهولي ،برنامه نيستيك هاي   ورودي به ساير پنجره   
بنابراين ابتدا بايد عمليات و پاسـخ دهـي   . هاي برنامه هاي ديگر نيز نيست انتخاب ساير پنجره امكان system modalحالت 

  .ها پرداخت توان به ساير برنامه شود و سپس ميو تمام به اين پنجره انجام 
هـايي بـا     ديـالوگ بـاكس   شده و  ايجاد   CreateDialogParamخواني تابع   با فرا  modelessهايي با حالت        ديالوگ باكس 

ــابع    ــط ت ــودال توس ــت م ــي DialogBoxParamحال ــاد م ــد  ايج ــه  .گردن ــر ب ــب اگ ــودال ، قال ــاكس م ــالوگ ب ــت دي   حال
DS_SYSMODAL ، دايالوگ باكس از نوع  نيز اضافه شودsystem modalخواهد شد .  

 بايـد اسـتفاده   SendDlgItemMessageهـا از تـابع    ار گرفتـه بـر روي ديـالوگ بـاكس    هـاي قـر      براي تبادل پيغام با كنترل    
  :تعريف آن به صورت زير است.كرد

SendDlgItemMessage proto hwndDlg:DWORD,\  
                                             idControl:DWORD,\  
                                             uMsg:DWORD,\  
                                             wParam:DWORD,\  
                                             lParam:DWORD 

  
  :   براي مثال جهت دريافت متن از اديت كنترل به صورت زير مي توان عمل كرد

  
call SendDlgItemMessage, hDlg, ID_EDITBOX, WM_GETTEXT, 256, ADDR text_buffer 

  
  .  مراجعه كردMSDN يافتن ليست پيغامهاي مجاز آن بايد به براي   

ــز دارد    ــژه اي ني ــع وي ــا يــك ســري از كنتــرل هــا تواب ــراي كــار ب ــال .    همچنــين وينــدوز ب ــراي مث  و GetDlgItemTextب
CheckDlgButton    برنامـه نويـسي را سـاده تـر كـرده و      ، ايـن نـوع توابـع    . ها كار با دكمه ها و تكست باكس    جهت  و غيره

 بنــابراين تنهــا در حــالتي از .كنــد  و غيــره ، بــي نيــاز مــيlParamبرنامــه نــويس را از بكــارگيري جزئيــات بيــشتري ماننــد  
SendDlgItemMessageاستفاده نماييد كه توابع مخصوص كنترل وجود نداشته باشند   .  

  :كند  ارسال مي كه فرمت زير را داردcallbackا به يك تابع هاي مربوط به ديالوگ باكس ر    ويندوز پيغام
  

DlgProc  proto hDlg:DWORD ,\  
                        iMsg:DWORD ,\  
                        wParam:DWORD ,\  
                        lParam:DWORD 

  
رويـه  .باشـد   مـي FALSE و يـا  LRESULT ، TRUEوجي آن بجـاي    اما خر   ،ديالوگ باكس همانند رويه پنجره است          رويه  

 ما را هنگاميكـه پيغـامي بـه آن    callbackاين رويه ، تابع . شود  حقيقي و اصلي ديالوگ باكس توسط خود ويندوز اداره مي 
 و اگـر  دادهار قر  EAX را در TRUEرا پردازش مي نمايد بايد خروجي رسيده اين رويه اگر پيغام  . برسد با خبر خواهد كرد    

  .  بعنوان خروجي تابع قرار دهدEAX را در FALSEقصد پردازش آنرا ندارد بايد مقدار 
نمايـد بـه      پيغامهـايي را كـه پـردازش نمـي     رويه ديالوگ باكس تعريف شده در بالا رويه حقيقي آن نيست،        از آنجائيكه   

DefWindowProcكند  ارسال نمي.  
  

ز آن بعنوان پنجـره اصـلي برنامـه و يـا اسـتفاده از آن بعنـوان       استفاده ا. لوگ باكس وجود دارد   دو كاربرد مجزا براي ديا   
  . وسيله ورود اطلاعات

  
  :    پياده سازي ديالوگ باكس به صورت پنجره اصلي برنامه به دو شكل ممكن است

  
 رجيـستر  RegisterClassExفراخـواني  قالـب كلاسـي كـه بـا      توان از قالب ديـالوگ بـاكس بعنـوان           مي .۱

در .  در اين حالت ديالوگ باكس بعنوان يك پنجره معمـول رفتـار خواهـد كـرد        .، استفاده كرد  شود  مي
 ، lpfnWndProcديالوگ باكس پيغام ها را از طريق رويه پنجره مشخص شده توسط پـارامتر   اين حالت   

دوز هنگـام ايجـاد    اين است كه وينروشمزيت اين . )نه از طريق رويه ديالوگ باكس  (كند دريافت مي 
 همچنـين وينـدوز اعمـالي ماننـد     .كند قرار گرفته بر روي آنرا نيز ايجاد ميديالوگ باكس ، كنترل هاي    
اينجـا ديـالوگ بـاكس بـدون ايجـاد      . مديريت خواهـد كـرد   و غيره را     tabانتقال تمركز ورودي توسط     

  . شود پنجره والد ايجاد مي

  :)مربوط به روش اول(كد برنامه 
  

.386  

.model flat,stdcall  
option casemap:none  
WinMain proto :DWORD,:DWORD,:DWORD,:DWORD  
include \masm32\include\windows.inc  
include \masm32\include\user32.inc  
include \masm32\include\kernel32.inc  
includelib \masm32\lib\user32.lib  
includelib \masm32\lib\kernel32.lib  

.data  
ClassName db "DLGCLASS",0  
MenuName db "MyMenu",0  
DlgName db "MyDialog",0  
AppName db "Our First Dialog Box",0  
TestString db "Wow! I'm in an edit box now",0  

.data?  
hInstance HINSTANCE ?  
CommandLine LPSTR ?  
buffer db 512 dup(?)  

.const  
IDC_EDIT        equ 3000  
IDC_BUTTON      equ 3001  
IDC_EXIT        equ 3002  
IDM_GETTEXT     equ 32000  
IDM_CLEAR       equ 32001  
IDM_EXIT        equ 32002  

.code  
start:  
    invoke GetModuleHandle, NULL  
    mov    hInstance,eax  
    invoke GetCommandLine 
    mov CommandLine,eax  
    invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT  
    invoke ExitProcess,eax  

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:DWORD  
    LOCAL wc:WNDCLASSEX  
    LOCAL msg:MSG  
    LOCAL hDlg:HWND  
    mov   wc.cbSize,SIZEOF WNDCLASSEX  
    mov   wc.style, CS_HREDRAW or CS_VREDRAW  
    mov   wc.lpfnWndProc, OFFSET WndProc  
    mov   wc.cbClsExtra,NULL  
    mov   wc.cbWndExtra,DLGWINDOWEXTRA  
    push  hInst  
    pop   wc.hInstance  
    mov   wc.hbrBackground,COLOR_BTNFACE+1  
    mov   wc.lpszMenuName,OFFSET MenuName  
    mov   wc.lpszClassName,OFFSET ClassName  
    invoke LoadIcon,NULL,IDI_APPLICATION  
    mov   wc.hIcon,eax  
    mov   wc.hIconSm,eax  

    invoke LoadCursor,NULL,IDC_ARROW  
    mov   wc.hCursor,eax  
    invoke RegisterClassEx, addr wc  
    invoke CreateDialogParam,hInstance,ADDR DlgName,NULL,NULL,NULL  
    mov   hDlg,eax  
    invoke ShowWindow, hDlg,SW_SHOWNORMAL  
    invoke UpdateWindow, hDlg  
    invoke GetDlgItem,hDlg,IDC_EDIT  
    invoke SetFocus,eax  
    .WHILE TRUE  
        invoke GetMessage, ADDR msg,NULL,0,0  
        .BREAK .IF (!eax)  
       invoke IsDialogMessage, hDlg, ADDR msg  
        .IF eax ==FALSE  
            invoke TranslateMessage, ADDR msg  
            invoke DispatchMessage, ADDR msg  
        .ENDIF  
    .ENDW  
    mov     eax,msg.wParam  
    ret  
WinMain endp  

WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM  
    .IF uMsg==WM_DESTROY  
        invoke PostQuitMessage,NULL  
    .ELSEIF uMsg==WM_COMMAND  
        mov eax,wParam  
        .IF lParam==0  
            .IF ax==IDM_GETTEXT  
                invoke GetDlgItemText,hWnd,IDC_EDIT,ADDR buffer,512  
                invoke MessageBox,NULL,ADDR buffer,ADDR AppName,MB_OK  
            .ELSEIF ax==IDM_CLEAR  
                invoke SetDlgItemText,hWnd,IDC_EDIT,NULL  
            .ELSE  
                invoke DestroyWindow,hWnd  
            .ENDIF  
        .ELSE  
            mov edx,wParam  
            shr edx,16  
            .IF dx==BN_CLICKED  
                .IF ax==IDC_BUTTON  
                    invoke SetDlgItemText,hWnd,IDC_EDIT,ADDR TestString  
                .ELSEIF ax==IDC_EXIT  
                    invoke SendMessage,hWnd,WM_COMMAND,IDM_EXIT,0  
                .ENDIF  
            .ENDIF  
        .ENDIF  
    .ELSE  
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam  
        ret  
    .ENDIF  
    xor    eax,eax  
    ret  
WndProc endp  
end start  

  :محتويات فايل منبع برنامه فوق
  

#include "resource.h"  

#define IDC_EDIT                                       3000  
#define IDC_BUTTON                                3001  
#define IDC_EXIT                                       3002  

#define IDM_GETTEXT                             32000  
#define IDM_CLEAR                                  32001  
#define IDM_EXIT                                      32003  
   

MyDialog DIALOG 10, 10, 205, 60  
STYLE 0x0004 | DS_CENTER | WS_CAPTION | WS_MINIMIZEBOX |  
WS_SYSMENU | WS_VISIBLE | WS_OVERLAPPED | DS_MODALFRAME | DS_3DLOOK  
CAPTION "Our First Dialog Box"  
CLASS "DLGCLASS"  
BEGIN  
    EDITTEXT         IDC_EDIT,   15,17,111,13, ES_AUTOHSCROLL | ES_LEFT  
    DEFPUSHBUTTON   "Say Hello", IDC_BUTTON,    141,10,52,13  
    PUSHBUTTON      "E&xit", IDC_EXIT,  141,26,52,13, WS_GROUP  
END  
   

MyMenu  MENU  
BEGIN  
    POPUP "Test Controls"  
    BEGIN  
        MENUITEM "Get Text", IDM_GETTEXT  
        MENUITEM "Clear Text", IDM_CLEAR  
        MENUITEM "", , 0x0800 /*MFT_SEPARATOR*/  
        MENUITEM "E&xit", IDM_EXIT  
    END  
END  

  
  

  :)مربوط به روش اول (بررسي كد فوق
  

ايـن روش برنامـه   . دهد    اين مثال نحوه ثبت قالب ديالوگ را بعنوان كلاس پنجره و ايجاد پنجره اي متداول را نمايش مي   
  . شود ها به صورت خودكار انجام مي نويسي را ساده تر مي نمايد زيرا ايجاد كنترل

  
MyDialog DIALOG 10, 10, 205, 60  

  
 مـساوي سـاير پارامترهـا   . تعريـف شـده اسـت    ،   DIALOGبـه همـراه كلمـه كليـدي         ) MyDialog(در اينجا نـام ديـالوگ          
  :در ادامه. ) يكي نيستpixelبا  ( طول و عرض آن در واحد ديالوگ باكس استوصات گوشه بالا سمت چپ فرم تمخ

STYLE 0x0004 | DS_CENTER | WS_CAPTION | WS_MINIMIZEBOX |  
WS_SYSMENU | WS_VISIBLE | WS_OVERLAPPED | DS_MODALFRAME | DS_3DLOOK  

  
  .كند حالت ديالوگ باكس را تعريف مي   
  

CAPTION "Our First Dialog Box"  
  .شود پنجره نمايش داده ميعنوان متني است كه در نوار    

  
CLASS "DLGCLASS"  

  
تـوان    ، از قالب ديالوگ باكس بعنوان كلاس پنجره مـي classيدي  به كمك كلمه كل   .باشد    اين خط از كد بسيار مهم مي      

  :در ادامه . آيد نام كلاس است كلمه اي كه پس از آن مي. استفاده كرد
  

BEGIN  
    EDITTEXT         IDC_EDIT,   15,17,111,13, ES_AUTOHSCROLL | ES_LEFT  
    DEFPUSHBUTTON   "Say Hello", IDC_BUTTON,    141,10,52,13  
    PUSHBUTTON      "E&xit", IDC_EXIT,  141,26,52,13  
END  

  
نحوه تعريف عمـومي آن بـه   . كند   تعريف ميend و  beginهاي تعريف شده بر روي پنجره را بين          كنترل   قطعه كد فوق    

  :صورت زير است
control-type   "text"   ,controlID, x, y, width, height [,styles] 

  
  :باشد   ساختار كلاس مي،  قسمت جالب توجه اين كد 

mov   wc.cbWndExtra,DLGWINDOWEXTRA  
mov   wc.lpszClassName,OFFSET ClassName 

  
اما اگر قالب يك ديالوگ باكس را بخواهيم بعنـوان    .شود   ، نال در نظر گرفته مي      cbWndExtra   به صورت معمول عضو     

 در ادامـه بايـد دقـت    . مقـدار دهـي كنـيم   DLGWINDOWEXTRAبايد اين عـضو را بـا   يك كلاس پنجره رجيستر كنيم ،       
سـاير اعـضاي ايـن سـاختار ماننـد      .  يكـي باشـد  class يداشت كه نام كلاس بايد با نام تعريف شده آن توسط كلمه كليـد     

  . مثالهاي قبل مقدار دهي خواهند شد
  :در ادامه .كنيم   رجيستر ميRegisterClassEx    پس از پر كردن ساختار كلاس ، آنرا با فراخواني تابع

  
invoke CreateDialogParam,hInstance,ADDR DlgName,NULL,NULL,NULL  

 را modelessتـابع فـوق ديـالوگ باكـسي     . كنيم ديالوگ باكس خود را ايجاد مي   پس از رجيستر كردن كلاس پنجره ،      
پـارامتر اول دسـتگيره اي   . كند اما پر كردن دو پارامتر اول آن كفايت مي    ،پذيرد     آرگومان را مي   ۵اين تابع   .كند ايجاد مي 

  . )و نه به نام كلاس (گري است به نام قالب ديالوگ باكس به وهله برنامه بوده و پارامتر دوم اشاره
 را طبـق  WM_CREATEپيغام ،  و رويه پنجره  شدهتوسط ويندوز ايجاد    هاي آن       در اين لحظه ، ديالوگ باكس و كنترل       

  :در ادامه. كنند معمول دريافت مي
  

invoke GetDlgItem,hDlg,IDC_EDIT  
invoke SetFocus,eax 

  
  .دهيم    پس از ايجاد اديت باكس ، تمركز ورودي را به آن انتقال مي

  شـد  با شكست مواجـه خواهـد  GetDlgItem  تابعفراخواني قرار دهيم،  WM_CREATE   اگر اين قطعه كد را در قسمت          
 كنتـرل  GetDlgItem،   ID تابع .ايم  قرار دادهUpdateWindow بنابراين آنرا پس از فراخواني     .)زيرا هنوز ايجاد نشده است    (

 يـك كنتـرل   IDكـه بـا دانـستن    اسـت  ايـن روشـي   (. گردانـد   باز مي كنترل را را دريافت كرده و دستگيره پنجره وابسته به         
  )گشت دادرا بازمربوطه توان دستگيره پنجره  مي

  
         invoke IsDialogMessage, hDlg, ADDR msg  
        .IF eax ==FALSE  
            invoke TranslateMessage, ADDR msg  
            invoke DispatchMessage, ADDR msg  
        .ENDIF  

  
وگ بـاكس اجـازه مـي دهـيم تـا رخ داد هـاي         ، بـه مـديريت درونـي ديـال       IsDialogMessage   به كمك فراخواني تابع     

  .  استTRUEبازگشتي اين تابع . صفحه كليد را براي ما پردازش كند
 جهت دريافت رشـته  GetDlgItemTextاستفاده از تابع هاي قبلي ،       همچنين يكي ديگر از تفاوت هاي اين برنامه با برنامه         

  .شود وگ باكس ها سبب سادگي كار مياستفاده از اين تابع هنگام كار با ديال. ورودي است
  
  
  

در ايـن حالـت حلقـه ي پيغامهـا و يـا      .  ايجاد خواهيم كردmodal ديالوگ باكسي به صورت  در مثال دومي از اين دست ،  
  !زيرا نيازي به آنها نيست. رويه پنجره را نخواهيد يافت

  :)مربوط به روش دوم(كد برنامه 
  

.386  

.model flat,stdcall  
option casemap:none  

DlgProc proto :DWORD,:DWORD,:DWORD,:DWORD  

include \masm32\include\windows.inc  
include \masm32\include\user32.inc  
include \masm32\include\kernel32.inc  
includelib \masm32\lib\user32.lib  
includelib \masm32\lib\kernel32.lib  

.data  
DlgName db "MyDialog",0  
AppName db "Our Second Dialog Box",0  
TestString db "Wow! I'm in an edit box now",0  

.data?  
hInstance HINSTANCE ?  
CommandLine LPSTR ?  
buffer db 512 dup(?)  

.const  
IDC_EDIT            equ 3000  
IDC_BUTTON     equ 3001  
IDC_EXIT            equ 3002  
IDM_GETTEXT  equ 32000  
IDM_CLEAR       equ 32001  
IDM_EXIT           equ 32002  
   

.code  
start:  
    invoke GetModuleHandle, NULL  
    mov    hInstance,eax  
    invoke DialogBoxParam, hInstance, ADDR DlgName,NULL, addr DlgProc, NULL  
    invoke ExitProcess,eax  

DlgProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM  
    .IF uMsg==WM_INITDIALOG  
        invoke GetDlgItem, hWnd,IDC_EDIT  
        invoke SetFocus,eax  
    .ELSEIF uMsg==WM_CLOSE  
        invoke SendMessage,hWnd,WM_COMMAND,IDM_EXIT,0  
    .ELSEIF uMsg==WM_COMMAND  
        mov eax,wParam  
        .IF lParam==0  
            .IF ax==IDM_GETTEXT  
                invoke GetDlgItemText,hWnd,IDC_EDIT,ADDR buffer,512  
                invoke MessageBox,NULL,ADDR buffer,ADDR AppName,MB_OK  
            .ELSEIF ax==IDM_CLEAR  
                invoke SetDlgItemText,hWnd,IDC_EDIT,NULL  
            .ELSEIF ax==IDM_EXIT  
                invoke EndDialog, hWnd,NULL  
            .ENDIF  

        .ELSE  
            mov edx,wParam  
            shr edx,16  
            .if dx==BN_CLICKED  
                .IF ax==IDC_BUTTON  
                    invoke SetDlgItemText,hWnd,IDC_EDIT,ADDR TestString  
                .ELSEIF ax==IDC_EXIT  
                    invoke SendMessage,hWnd,WM_COMMAND,IDM_EXIT,0  
                .ENDIF  
            .ENDIF  
        .ENDIF  
    .ELSE  
        mov eax,FALSE  
        ret  
    .ENDIF  
    mov eax,TRUE  
    ret  
DlgProc endp  
end start  

 
dialog.rc (part 2) 

 
 
#include "resource.h"  

#define IDC_EDIT                                       3000  
#define IDC_BUTTON                                3001  
#define IDC_EXIT                                       3002  

#define IDR_MENU1                                  3003  

#define IDM_GETTEXT                              32000  
#define IDM_CLEAR                                   32001  
#define IDM_EXIT                                       32003  
   

MyDialog DIALOG 10, 10, 205, 60  
STYLE 0x0004 | DS_CENTER | WS_CAPTION | WS_MINIMIZEBOX |  
WS_SYSMENU | WS_VISIBLE | WS_OVERLAPPED | DS_MODALFRAME | DS_3DLOOK  
CAPTION "Our Second Dialog Box"  
MENU IDR_MENU1  
BEGIN  
    EDITTEXT         IDC_EDIT,   15,17,111,13, ES_AUTOHSCROLL | ES_LEFT  
    DEFPUSHBUTTON   "Say Hello", IDC_BUTTON,    141,10,52,13  
    PUSHBUTTON      "E&xit", IDC_EXIT,  141,26,52,13  
END  
   

IDR_MENU1  MENU  
BEGIN  
    POPUP "Test Controls"  
    BEGIN  
        MENUITEM "Get Text", IDM_GETTEXT  
        MENUITEM "Clear Text", IDM_CLEAR  
        MENUITEM "", , 0x0800 /*MFT_SEPARATOR*/  
        MENUITEM "E&xit", IDM_EXIT  
    END  
END  

  :)مربوط به روش دوم (بررسي كد فوق
  

    DlgProc proto :DWORD,:DWORD,:DWORD,:DWORD  
  

  :مشخص شده است، گيرد  مورد استفاده قرار مي  addr  كه در كد زير با اپراتور DlgProc  در خط فوق تعريف رويه     
  

    invoke  DialogBoxParam, hInstance, ADDR DlgName,NULL, addr  DlgProc, NULL  
  

اين تابع تـا زمـاني كـه ديـالوگ بـاكس تخريـب نـشود بـه پايـان          .    تابع فوق ديالوگ باكسي مودال را ايجاد خواهد كرد    
  :در ادامه . نخواهد رسيد

    .IF uMsg==WM_INITDIALOG  
        invoke GetDlgItem, hWnd,IDC_EDIT  
        invoke SetFocus,eax  
    .ELSEIF uMsg==WM_CLOSE  
        invoke SendMessage,hWnd,WM_COMMAND,IDM_EXIT,0  

 اولـين  . را دريافـت نخواهـد كـرد   WM_CREATE   رويه ديالوگ باكس شبيه به رويه پنجره بوده با اين تفاوت كه پيغـام            
تـوان قـرار    ه را در ايـن قـسمت مـي   كدهاي آغازين برنام ـ.باشد  ميWM_INITDIALOGپيغامي را كه دريافت خواهد كرد     

 TRUEگيـرد بـا     قـرار مـي  EAXلازم به ذكر است ، در صورتيكه اينجا پيغامي پردازش شود بايد خروجي رويه كـه در       . داد
  . مقدار دهي شود

ــي  WM_CLOSE   هنگاميكــه پيغــام  ــاكس فرســتاده م ــالوگ ب ــه دي ــاكس    ب ــالوگ ب ــي دي ــديريت درون ــام ،شــود ، م  پيغ
WM_DESTROY   بنـابراين بـراي عكـس العمـل نـشان دادن بـه بـسته شـدن            . كند  براي رويه ديالوگ باكس ارسال نمي      را

در اينجـا عمـل كليـك شـدن بـر روي گزينـه       . را پـردازش كـرد   WM_CLOSE ديالوگ باكس توسط كاربر، بايد پيغـام    
  . خروج منو را شبيه سازي كرده ايم

فراخواني اين تـابع سـريعا سـبب تخريـب ديـالوگ بـاكس        . است EndDialog   تنها راه تخريب ديالوگ باكس فراخواني       
 و باقيمانـده كـد يكـي    كنـد   تنظيم ميTRUEاين تابع تنها پرچم مربوطه را در مديريت دروني ديالوگ باكس به         . شود نمي

  .پس از ديگري اجرا خواهند شد
  

  :   بررسي فايل منبع اين مثال
اينكـار بـراي الحـاق كـردن منـو بـه       .  استفاده شده اسـت IDR_MENU1از مقدار ،     بجاي بكارگيري رشته بعنوان نام منو      

قالب ديـالوگ بـاكس ، از كلمـه كليـدي      لازم به ذكر است كه در       .  لازم است  DialogBoxParamتوسط   ديالوگ باكس 
MENUشود آي دي مربوطه استفاده مي  به همراه .  

  .توان آيكون آنرا تنظيم كرد مي ، WM_INITDIALOGحين ديالوگ باكس در  به WM_SETICON   با ارسال پيغام 

  Switchمقايسه كد زبان سي و كد اسمبلي يك 
  

  
  

  : زير ارائه گرديدswitch   كد اسمبلي فوق جهت 

سـوئيچ  . كنـيم  ، از سـوئيچ اسـتفاده مـي    يـك عملگـر انتظـار داريـم          مقـادير مختلـف    ازرفتارهاي مختلفي را        هنگاميكه  
 همچنـين امكـان ارائـه يـك     .باشـد  هاي مناسب به اين مقادير مـي  مقادير براي يك عملگر و واكنش    جدولي از    درحقيقت

  . واكنش براي چندين مقدار نيز ميسر است
  

هـر كـدام از قطعـات يـك     .اسـت گرهـا   ها براي پياده سازي سوئيچ ها ، ايجاد جدولي از اشـاره     يكي از بهينه ترين روش  
گــري بـه هـر كـدام از ايـن قطعـات در يـك جـدول ذخيــره         اشـاره انـه كامپايـل شـده و سـپس     سـوئيچ بـه صـورت جداگ   

بعنـوان انديـسي در ايـن    مربوطـه  هاي سوئيچ ، عملگر  اجراي قطعه كد قرار گرفته در يكي از شرط        نياز به   هنگام  .شوند  مي
  . ردو پردازشگر به سادگي به قطعه كد مربوطه پرش خواهد كرفتار كرده گرها  جدول اشاره

  
  :كند    شكل زير ديدگاه يك كامپايلر را از يك سوئيچ و نحوه ي اجراي آنرا بيان مي

  
  
  

http://mohandesyar.com


  قسمت دوم- آشنايي با برنامه نويسي اسمبلي ويندوز

 ٦٥

  
  
  

  :باشد    اين درخت ، پياده سازي سوئيچ زير مي
  

switch (Value) 
{ 
          case 120: 
                         Code... 
                         break; 
          case 140: 
                         Code... 
                         break; 
          case 501: 
                         Code... 
                         break; 
          case 1001: 
                         Code... 

                         break; 
          case 1100: 
                         Code... 
                         break; 
          case 1400: 
                         Code... 
                         break; 
          case 2000: 
                         Code... 
                         break; 
          case 3400: 
                         Code... 
                         break; 
          case 4100: 
                         Code... 
                         break; 
}; 

  )موارد تكميلي(ها   استفاده از ديالوگ باكس-۹مثال 
  

  . را خواهيم آموخت common dialog boxesه از    در اين مثال نحوه استفاد
هـدف از ارائـه آنهـا نيـز     . هاي از پيش تعريـف شـده جهـت اسـتفاده در برنامـه هـاي شـما دارد                 ويندوز يك سري پنجره   

  :ها شامل موارد زير هستند اين پنجره. باشد ها مي هاي كاربر برنامه استاندارد كردن رابط
file, print, color, font, and search dialog boxes  

  
 بـه همـين جهـت بـراي اسـتفاده از آنهـا نيـاز اسـت تـا          .  قرار دارنـد comdlg32.dllهاي استاندارد در      اين ديالوگ باكس  

comdlg32.libبه برنامه الحاق شود  .  
ــاكس   ــالوگ ب ــن دي ــراي فراخــواني اي ــي     ب ــتاندارد م ــاي اس ــرد ما    ه ــتفاده ك ــار اس ــع مخــصوص اينك ــوان از تواب ــد ت نن

GetOpenFileName    ، ــراي نمــايش پنجــره گــشودن فايلهــا ــراي نمــايش ديــالوگ بــاكس ذخيــره بعنــوان         ب از ...  ب
GetSaveFileName كـه در  (هر كدام از اين توابع براي كار نياز به اشاره گري بـه سـاختاري ويـژه دارنـد     .  و غيرهMSDN 

  . )قابل مطالعه هستند
  :  به صورت زير استGetOpenFileName تعريف تابع . را داريمopen file dialog    در اين مثال قصد نمايش 

GetOpenFileName proto lpofn:DWORD  
  

 باشـد يعنـي كـاربر    TRUEاگر خروجي ايـن تـابع    . OPENFILENAME   تنها پارامتر اين تابع اشاره گري است به ساختار          
  :ف اين ساختار به صورت زير استتعري. فايلي را انتخاب كرده است ، در غير اينصورت خير

OPENFILENAME  STRUCT  
   lStructSize DWORD  ?  
   hwndOwner HWND  ?  
   hInstance HINSTANCE ?  
   lpstrFilter LPCSTR  ?  
   lpstrCustomFilter LPSTR  ?  
   nMaxCustFilter DWORD  ?  
   nFilterIndex DWORD  ?  
   lpstrFile LPSTR  ?  
   nMaxFile DWORD  ?  
   lpstrFileTitle LPSTR  ?  
   nMaxFileTitle DWORD  ?  
   lpstrInitialDir LPCSTR  ?  
   lpstrTitle LPCSTR  ?  
   Flags  DWORD  ?  
   nFileOffset WORD  ?  
   nFileExtension WORD  ?  
   lpstrDefExt LPCSTR  ?  
   lCustData LPARAM  ?  
   lpfnHook DWORD  ?  
   lpTemplateName LPCSTR  ? 
OPENFILENAME  ENDS  

  :   توضيحات اعضاي مهم آن به شرح زير است
  

 lStructSize . به بايتOPENFILENAMEاندازه ساختار 

 open file dialog box.   hwndOwnerدستگيره اي به پنجره 

 hInstance .كند  را ايجاد ميopen file dialog boxدستگيره اي به وهله برنامه اي كه 

 بـه  مهـاي مختـو   معرفـي شـده توسـط رشـته     (گيرد د استفاده قرار ميفيلتري كه براي نمايش فايلها مور     
  :براي مثال. )نال

     FilterString   db "All Files (*.*)",0, "*.*",0  
                         db "Text Files (*.txt)",0,"*.txt",0,0  

  
يده توسـط وينـدوز بـراي       لازم به ذكر است كه تنها الگوي معرفي شده در دومين زوج تعريف گرد    

همچنين يك صفر اضافي هم در انتهـاي تعريـف رشـته فيلتـر          .گيرد  اعمال فيلتر مورد استفاده قرار مي     
 .بايد قرار گيرد تا خاتمه ي كل آنرا مشخص كند

lpstrFilter 

كند كه كدام زوج معرفي شده بعنوان فيلتـر بايـد در آغـاز نمـايش ديـالوگ بـاكس بعنـوان                  تعيين مي 
 . شود اين انديس از يك شروع مي.  فرض انتخاب شوندپيش

nFilterIndex 

گر به بافري است كه جهت ذخيره كردن نام و مـسير فايـل انتخـاب شـده توسـط كـاربر ، بكـار                  اشاره
 lpstrFile  . باشدbytes 260حداقل طول آن بايد . رود مي

 nMaxFile  .است lpstrFileاندازه بافر 

 lpstrTitle .ديالوگ باكس نمايش داده شدهگري به عنوان  اشاره

 Flags  .كند ويژگي ها و حالات ديالوگ باكس را تعيين مي

اين عضو حـاوي انـديس اولـين كـاراكتر     پس از اينكه كاربري فايلي را جهت گشودن انتخاب كرد ،   
 باشـد،  c:\windows\system\lz32.dllنام فايل به همراه مسير كامل آن      براي مثال اگر    . شود  مينام فايل   

  .  خواهد شد۱۸اين عضو مساوي 
nFileOffset 

پس از اينكه كاربري فايلي را جهت گشودن انتخاب كرد ، اين عضو حـاوي انـديس اولـين كـاراكتر       
 .شود پسوند فايل مي

nFileExtension 

  
  

توسـط كـاربر ، نـام فايـل بـه      لوگ باكس استاندارد گشودن فايـل و انتخـاب فايـل      در مثال اين قسمت پس از نمايش ديا   
  .  نمايش داده خواهد شدmessage boxهمراه مسير و پسوند آن توسط يك 

  :كد برنامه
  

.386  

.model flat,stdcall  
option casemap:none  
WinMain proto :DWORD,:DWORD,:DWORD,:DWORD  
include \masm32\include\windows.inc  
include \masm32\include\user32.inc  
include \masm32\include\kernel32.inc  
include \masm32\include\comdlg32.inc  
includelib \masm32\lib\user32.lib  
includelib \masm32\lib\kernel32.lib  
includelib \masm32\lib\comdlg32.lib  

.const  
IDM_OPEN equ 1  
IDM_EXIT equ 2  
MAXSIZE equ 260  
OUTPUTSIZE equ 512  

.data  
ClassName db "SimpleWinClass",0  
AppName  db "Our Main Window",0  
MenuName db "FirstMenu",0  
ofn   OPENFILENAME <>  
FilterString db "All Files",0,"*.*",0  
             db "Text Files",0,"*.txt",0,0  
buffer db MAXSIZE dup(0)  
OurTitle db "-=Our First Open File Dialog Box=-: Choose the file to open",0  
FullPathName db "The Full Filename with Path is: ",0  
FullName  db "The Filename is: ",0  
ExtensionName db "The Extension is: ",0  
OutputString db OUTPUTSIZE dup(0)  
CrLf db 0Dh,0Ah,0  

.data?  
hInstance HINSTANCE ?  
CommandLine LPSTR ?  

.code  
start:  
    invoke GetModuleHandle, NULL  
    mov    hInstance,eax  
    invoke GetCommandLine 
    mov CommandLine,eax  
    invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT  
    invoke ExitProcess,eax  

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:DWORD  
    LOCAL wc:WNDCLASSEX  
    LOCAL msg:MSG  
    LOCAL hwnd:HWND  
    mov   wc.cbSize,SIZEOF WNDCLASSEX  
    mov   wc.style, CS_HREDRAW or CS_VREDRAW  
    mov   wc.lpfnWndProc, OFFSET WndProc  
    mov   wc.cbClsExtra,NULL  
    mov   wc.cbWndExtra,NULL  

    push  hInst  
    pop   wc.hInstance  
    mov   wc.hbrBackground,COLOR_WINDOW+1  
    mov   wc.lpszMenuName,OFFSET MenuName  
    mov   wc.lpszClassName,OFFSET ClassName  
    invoke LoadIcon,NULL,IDI_APPLICATION  
    mov   wc.hIcon,eax  
    mov   wc.hIconSm,eax  
    invoke LoadCursor,NULL,IDC_ARROW  
    mov   wc.hCursor,eax  
    invoke RegisterClassEx, addr wc  
    invoke CreateWindowEx,WS_EX_CLIENTEDGE,ADDR ClassName,ADDR AppName,\  
           WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\  
           CW_USEDEFAULT,300,200,NULL,NULL,\  
           hInst,NULL  
    mov   hwnd,eax  
    invoke ShowWindow, hwnd,SW_SHOWNORMAL  
    invoke UpdateWindow, hwnd  
    .WHILE TRUE  
        invoke GetMessage, ADDR msg,NULL,0,0  
        .BREAK .IF (!eax)  
        invoke TranslateMessage, ADDR msg  
        invoke DispatchMessage, ADDR msg  
    .ENDW  
    mov     eax,msg.wParam  
    ret  
WinMain endp  

WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM  
    .IF uMsg==WM_DESTROY  
        invoke PostQuitMessage,NULL  
    .ELSEIF uMsg==WM_COMMAND  
        mov eax,wParam  
        .if ax==IDM_OPEN  
            mov ofn.lStructSize,SIZEOF ofn  
            push hWnd  
            pop  ofn.hwndOwner  
            push hInstance  
            pop  ofn.hInstance  
            mov  ofn.lpstrFilter, OFFSET FilterString  
            mov  ofn.lpstrFile, OFFSET buffer  
            mov  ofn.nMaxFile,MAXSIZE  
            mov  ofn.Flags, OFN_FILEMUSTEXIST or \  
                OFN_PATHMUSTEXIST or OFN_LONGNAMES or\  
                OFN_EXPLORER or OFN_HIDEREADONLY  
            mov  ofn.lpstrTitle, OFFSET OurTitle  
            invoke GetOpenFileName, ADDR ofn  
            .if eax==TRUE  
                invoke lstrcat,offset OutputString,OFFSET FullPathName  
                invoke lstrcat,offset OutputString,ofn.lpstrFile  
                invoke lstrcat,offset OutputString,offset CrLf  
                invoke lstrcat,offset OutputString,offset FullName  
                mov  eax,ofn.lpstrFile  
                push ebx  
                xor  ebx,ebx  
                mov  bx,ofn.nFileOffset  
                add  eax,ebx  
                pop  ebx  
                invoke lstrcat,offset OutputString,eax  
                invoke lstrcat,offset OutputString,offset CrLf  
                invoke lstrcat,offset OutputString,offset ExtensionName  

                mov  eax,ofn.lpstrFile  
                push ebx  
                xor ebx,ebx  
                mov  bx,ofn.nFileExtension  
                add eax,ebx  
                pop ebx  
                invoke lstrcat,offset OutputString,eax  
                invoke MessageBox,hWnd,OFFSET OutputString,ADDR AppName,MB_OK  
                invoke RtlZeroMemory,offset OutputString,OUTPUTSIZE  
            .endif  
        .else  
            invoke DestroyWindow, hWnd  
        .endif  
    .ELSE  
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam  
        ret  
    .ENDIF  
    xor    eax,eax  
    ret  
WndProc endp  
 end start  

  
  

  : بررسي كد فوق
  
  

            Mov  ofn.lStructSize,SIZEOF ofn  
            push hWnd  
            pop   ofn.hwndOwner  
            push hInstance  
            pop   ofn.hInstance  

  
  : در ادامه .اند   مقدار دهي شدهofn   در اينجا اعضاي ساختار     

mov  ofn.lpstrFilter, OFFSET FilterString  
  
   FilterStringشود  نام فيلتري است كه به صورت زير تعريف مي:  
  

FilterString  db "All Files",0,"*.*",0  
                   db "Text Files",0,"*.txt",0,0  

  
  :شود  جهت قرار گيري نام و مسير فايل تعريف ميآن ، مورد نياز و اندازه بافر    در ادامه 

  
            mov  ofn.lpstrFile, OFFSET buffer  
            mov  ofn.nMaxFile,MAXSIZE  

براي توضيحات بيـشتر  . نامهاي انتخابي اكثر آنها بسيار واضح است      (شوند  هاي ديالوگ باكس تعريف مي     ويژگي   سپس  
  : ) مراجعه كردMSDNتوان به  مي

  
            mov  ofn.Flags, OFN_FILEMUSTEXIST or \  
                                      OFN_PATHMUSTEXIST or OFN_LONGNAMES or\  
                                      OFN_EXPLORER or OFN_HIDEREADONLY  

  
  

  :مشخص كردن عنوان ديالوگ باكس:    در ادامه 
mov  ofn.lpstrTitle, OFFSET OurTitle  

  
  :تابع نمايشي   سپس فراخواني 

Invoke  GetOpenFileName, ADDR ofn  

  
در ايـن زمـان ديـالوگ بـاكس نمـايش داده      .گيرد  گري به ساختار تعريف شده را بعنوان آرگومان ورودي مي              كه اشاره 

ايـن  .داندگر  كليك ننمايد ، چيزي را بر نميcancelي  تابع تا زمانيكه كاربر فايلي را انتخاب نكند و يا روي دكمه        .شود  مي
در غيراينـصورت  .دهـد    قـرار مـي  eax را بعنـوان خروجـي در      TRUEدر صورتي كه كاربر فايلي را انتخـاب نماييـد ،            تابع  

  :در ادامه .  خواهد بودFALSEخروجي 

            .if eax==TRUE  
                invoke lstrcat,offset OutputString,OFFSET FullPathName  
                invoke lstrcat,offset OutputString,ofn.lpstrFile  
                invoke lstrcat,offset OutputString,offset CrLf  
                invoke lstrcat,offset OutputString,offset FullName  

  
ابتـدا  . كنـيم   آمـاده مـي  message box در رشـته خروجـي را بـراي نمـايش      انتخـاب كـرد ،     پس از اينكه كاربر فايلي را 

  . كنيم اتصال رشته ها به هم استفاده مي براي lstrcat  تابعاز آماده كرده و سپس OutputStringاي از حافظه را براي  قطعه
  :سپس.  استفاده كرد)CrLf )carriage return-line feed   براي قرار دادن خطوط در چندين خط مختلف مي توان از 

  
 mov   eax,ofn.lpstrFile                 

                push  ebx  
                xor     ebx,ebx  
                mov   bx,ofn.nFileOffset  
                add    eax,ebx  
                pop    ebx  
                invoke  lstrcat,offset OutputString,eax  

 WORDاندازه اولـي    .  استofn.lpstrFileانديسي در   حاوي   nFileOffset.    خطوط فوق نياز به توضيحات بيشتري دارند      
  low wordدر   را بايـد nFileOffsetبنـابراين  . و امكان افزودن آنها به يكديگر موجود نيـست  استDWORDو اندازه دومي 

  :در ادامه . افزودlpstrFile قرار داد و سپس آنرا به مقدار  ebxمربوط به 
  

invoke MessageBox,hWnd,OFFSET OutputString,ADDR AppName,MB_OK  
  

  : سپس .  نمايش خواهيم دادmessage box   پيغام را توسط 

invoke RtlZerolMemory,offset OutputString,OUTPUTSIZE  

ــي   ــدار ده ــل از مق ــا OutputString   قب ــد   ب ــر، باي ــته اي ديگ ــواي مرش ــرد حت ــاك ك ــرا پ ــابع    . آن ــت از ت ــين جه ــه هم ب
RtlZeroMemoryاستفاده گرديده است  .  

  مقايسه كد زبان سي و كد اسمبلي يك حلقه
  
  

 پياده سازي حلقه در زبان اسمبلي نمايش حلقه در زبان سي

c = 0; 
while (c < 1000) 
{ 
     array[c] = c; 
     c++; 
} 

 
mov ecx, DWORD PTR [array] 
xor eax, eax 
LoopStart: 
    mov DWORD PTR [ecx+eax*4], eax 
    add eax, 1 
    cmp eax, 1000 
jl   LoopStart  

  

  
باشد ، با اين تفاوت كه مرتبا قطعـه كـدي    تفاده از دستورات شرطي مي   پياده سازي يك حلقه در زبان اسمبلي همانند اس    

  .كند دن شرط آن، اجرا ميرا تا زمان صادق بو
  

  : در آن  break   حلقه به همراه يك 
  

 پياده سازي حلقه در زبان اسمبلي نمايش حلقه در زبان سي

do 
{ 
    if (array[c]) 
        break; 
 
     array[c] = c; 
     c++; 
} while (c < 1000); 
 

 
mov eax, DWORD PTR [c] 
mov ecx, DWORD PTR [array] 
LoopStart: 
     cmp DWORD PTR [ecx+eax*4], 0 
     jne AfterLoop 
     mov DWORD PTR [ecx+eax*4], eax 
     add eax, 1 
     cmp eax, 1000 
jl LoopStart 
AfterLoop: 

  مديريت حافظه و كار با فايلها -۱۰مثال 
  

نيـز    common dialog boxesهمچنين از .  حافظه را خواهيم آموخت   در اين قسمت طرز كار ابتدايي با فايلها و مديريت
  . استفاده خواهد شد

  
  :يادآوري 

  
فـضاي آدرس دهـي   ،  هـر پروسـه در وينـدوز    .باشـد     از ديد يك برنامه ، مديريت حافظه در ويندوز سـاده و واضـح مـي       

در ايـن  .  تخـت معـروف اسـت   ي  مـدل حافظـه  مدل حافظه در اين حالـت ، بـه    . را دارا است   گيگابايت   ۴اي معادل     حافظه
.  بيـت هـستند  ۳۲كنند و آفست ها  ، به يك آدرس آغازين اشاره مي) گرها  يا انتخاب( حالت، تمام رجيسترهاي سگمنت    

توانـد   بـه هـر آدرسـي بـدون نيـاز بـه تغييـر مقـدار انتخابگرهـا مـي          ، بنابراين هر برنامه در فضاي آدرس دهي حافظه خـود       
  .كند مديريت حافظه را بسيار ساده مي، اين مدل . كنددسترسي پيدا 

 توابع مديريت حافظـه وجـود    دو نوع عمده ازWin16تحت .  وجود ندارند "near" or "far"گرهاي     در اينجا ديگر اشاره
هـاي تخـصيص داده شـده در سـاير سـگمنت هـا سـروكار داشـتند           بـا حافظـه  Globalنوع هاي  .Global and Local: دارند

 تحـت  . بودنـد near پردازشـگر سـروكار داشـتند بنـابراين از نـوع      local heap با Localنوع هاي .  بودندfarبنابراين از نوع 
Win32بنابراين فراخواني .  اين دو نوع يكي هستندGlobalAlloc or LocalAllocندگردان   يك نتيجه را بر مي .  

  :   مراحل تخصيص و بكارگيري حافظه به شرح زير هستند
اي را بـه قطعـه حافظـه      ايـن تـابع دسـتگيره    .GlobalAlloc  تـابع  اي از حافظه با فراخـواني      تخصيص قطعه  •

 . گرداند تخصيص داده شده بر مي

اي بـه قطعـه   اين تابع دسـتگيره     . GlobalLockقفل كردن اين قطعه تخصيص داده شود با فراخواني تابع            •
 . گرداند حافظه مورد نظر را دريافت و سپس اشاره گري به آنرا بر مي

گـر ايجـاد شـده بـه قطعـه        ايـن تـابع اشـاره    .GlobalUnlockرهاكردن قفل قطعه حافظه با فراخواني تابع         •
 . كند حافظه را غيرمعتبر مي

بـه  را ايـن تـابع دسـتگيره اي     . GlobalFreeآزاد سازي حافظه تخصيص داده شده توسط فراخواني تابع     •
  . كند  دريافت مي،قطعه حافظه مورد نظر 

  
  . بيتي انجام مي دهند۳۲ اين توابع نيز اعمال يكساني را تحت ويندوزهاي local   همانطور كه ذكر گرديد نگارش 

http://mohandesyar.com


  قسمت دوم- آشنايي با برنامه نويسي اسمبلي ويندوز

 ٧٦

ايـن حالـت   در .  ، سـاده تـر مـي گـردد    GlobalAlloc در تـابع  GMEM_FIXED   همچنين روش فوق با بكـارگيري پـرچم      
در اين حالت ديگر نيازي بـه فراخـواني دو تـابع قفـل     . گري به قطعه حافظه تخصيص داده شده خواهد خروجي تابع اشاره  

  . كردن قطعه حافظه و رها كردن قفل مربوطه نيست
  . م كردجهت آشنايي بيشتر ، استفاده خواهي    البته در اين برنامه از روش سنتي اينكار بدون بكارگيري پرچم فوق 

  
 APIمراحل مورد نياز يكي هـستند و تنهـا بايـد وقفـه هـا را بـه توابـع         .  شبيه داس مي باشد    Win32   استفاده از فايلها تحت     

  :اين مراحل به شرح زير هستند. تبديل كرد
از آن . ايـن تـابع چنـدين منظـوره مـي باشـد       . CreateFileگشودن و يا ايجاد فايل توسط فراخواني تابع         .۱

پس از اجراي موفقيت آميز ، دستگيره اي بـه     . توان استفاده كرد   شودن فايلها ، پورتها و غيره مي      براي گ 
از ايـن دسـتگيره بـراي انجـام عمليـات روي فايـل و يـا وسـيله         . گردانـد  فايل و يا وسيله مورد نظر برمـي    

 .   استفاده كردSetFilePointerبراي انتقال اشاره گر به فايل مي توان از تابع . گيريم مربوطه كمك مي

ايـن توابـع قطعـه اي از     .ReadFile or WriteFileانجام عمليات خواندن و يا نوشتن بـا فراخـواني توابـع     .۲
بنـابراين بايـد قطعـه تخـصيص داده شـده از حافظـه ، بـه        . دهنـد   حافظه را به فايل و يا از فايل انتقال مـي      

 .ها بزرگ باشد اندازه كافي براي نگهداري داده

   .CloseHandle فايل  با فراخواني بستن .۳
  
  

  :كد برنامه
  
در اين برنامه قصد گشودن يك فايل متني و سپس نمايش محتويات آن در يك اديت باكس به همراه قابليـت ويـرايش             

  .را داريمدر آن و ذخيره سازي متن تغيير كرده 
  

.386  

.model flat,stdcall  
option casemap:none  
WinMain proto :DWORD,:DWORD,:DWORD,:DWORD  
include \masm32\include\windows.inc  
include \masm32\include\user32.inc  
include \masm32\include\kernel32.inc  
include \masm32\include\comdlg32.inc  
includelib \masm32\lib\user32.lib  
includelib \masm32\lib\kernel32.lib  
includelib \masm32\lib\comdlg32.lib  

.const  
IDM_OPEN equ 1  
IDM_SAVE equ 2  

IDM_EXIT equ 3  
MAXSIZE equ 260  
MEMSIZE equ 65535  

EditID equ 1                            ; ID of the edit control  

.data  
ClassName db "Win32ASMEditClass",0  
AppName  db "Win32 ASM Edit",0  
EditClass db "edit",0  
MenuName db "FirstMenu",0  
ofn   OPENFILENAME <>  
FilterString db "All Files",0,"*.*",0  
             db "Text Files",0,"*.txt",0,0  
buffer db MAXSIZE dup(0)  

.data?  
hInstance HINSTANCE ?  
CommandLine LPSTR ?  
hwndEdit HWND ?                               ; Handle to the edit control  
hFile HANDLE ?                                   ; File handle  
hMemory HANDLE ?                            ;handle to the allocated memory block  
pMemory DWORD ?                            ;pointer to the allocated memory block  
SizeReadWrite DWORD ?                   ; number of bytes actually read or write  

.code  
start:  
    invoke GetModuleHandle, NULL  
    mov    hInstance,eax  
    invoke GetCommandLine 
    mov CommandLine,eax  
    invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT  
    invoke ExitProcess,eax  

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:SDWORD  
    LOCAL wc:WNDCLASSEX  
    LOCAL msg:MSG  
    LOCAL hwnd:HWND  
    mov   wc.cbSize,SIZEOF WNDCLASSEX  
    mov   wc.style, CS_HREDRAW or CS_VREDRAW  
    mov   wc.lpfnWndProc, OFFSET WndProc  
    mov   wc.cbClsExtra,NULL  
    mov   wc.cbWndExtra,NULL  
    push  hInst  
    pop   wc.hInstance  
    mov   wc.hbrBackground,COLOR_WINDOW+1  
    mov   wc.lpszMenuName,OFFSET MenuName  
    mov   wc.lpszClassName,OFFSET ClassName  
    invoke LoadIcon,NULL,IDI_APPLICATION  
    mov   wc.hIcon,eax  
    mov   wc.hIconSm,eax  
    invoke LoadCursor,NULL,IDC_ARROW  
    mov   wc.hCursor,eax  
    invoke RegisterClassEx, addr wc  
    invoke CreateWindowEx,WS_EX_CLIENTEDGE,ADDR ClassName,ADDR AppName,\  
           WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\  
           CW_USEDEFAULT,300,200,NULL,NULL,\  
           hInst,NULL  
    mov   hwnd,eax  
    invoke ShowWindow, hwnd,SW_SHOWNORMAL  

    invoke UpdateWindow, hwnd  
    .WHILE TRUE  
        invoke GetMessage, ADDR msg,NULL,0,0  
        .BREAK .IF (!eax)  
        invoke TranslateMessage, ADDR msg  
        invoke DispatchMessage, ADDR msg  
    .ENDW  
    mov     eax,msg.wParam  
    ret  
WinMain endp  

WndProc proc uses ebx hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM  
    .IF uMsg==WM_CREATE  
        invoke CreateWindowEx,NULL,ADDR EditClass,NULL,\  
                   WS_VISIBLE or WS_CHILD or ES_LEFT or ES_MULTILINE or\  
                   ES_AUTOHSCROLL or ES_AUTOVSCROLL,0,\  
                   0,0,0,hWnd,EditID,\  
                   hInstance,NULL  
        mov hwndEdit,eax  
        invoke SetFocus,hwndEdit  
;==============================================  
;        Initialize the members of OPENFILENAME structure  
;==============================================  
        mov ofn.lStructSize,SIZEOF ofn  
        push hWnd  
        pop  ofn.hWndOwner  
        push hInstance  
        pop  ofn.hInstance  
        mov  ofn.lpstrFilter, OFFSET FilterString  
        mov  ofn.lpstrFile, OFFSET buffer  
        mov  ofn.nMaxFile,MAXSIZE  
    .ELSEIF uMsg==WM_SIZE  
        mov eax,lParam  
        mov edx,eax  
        shr edx,16  
        and eax,0ffffh  
        invoke MoveWindow,hwndEdit,0,0,eax,edx,TRUE  
    .ELSEIF uMsg==WM_DESTROY  
        invoke PostQuitMessage,NULL  
    .ELSEIF uMsg==WM_COMMAND  
        mov eax,wParam  
        .if lParam==0  
            .if ax==IDM_OPEN  
                mov  ofn.Flags, OFN_FILEMUSTEXIST or \  
                                OFN_PATHMUSTEXIST or OFN_LONGNAMES or\  
                                OFN_EXPLORER or OFN_HIDEREADONLY  
                invoke GetOpenFileName, ADDR ofn  
                .if eax==TRUE  
                    invoke CreateFile,ADDR buffer,\  
                                GENERIC_READ or GENERIC_WRITE ,\  
                                FILE_SHARE_READ or FILE_SHARE_WRITE,\  
                                NULL,OPEN_EXISTING,FILE_ATTRIBUTE_ARCHIVE,\  
                                NULL  
                    mov hFile,eax  
                    invoke GlobalAlloc,GMEM_MOVEABLE or GMEM_ZEROINIT,MEMSIZE  
                    mov  hMemory,eax  
                    invoke GlobalLock,hMemory  
                    mov  pMemory,eax  
                    invoke ReadFile,hFile,pMemory,MEMSIZE-1,ADDR SizeReadWrite,NULL  
                    invoke SendMessage,hwndEdit,WM_SETTEXT,NULL,pMemory  
                    invoke CloseHandle,hFile  

                    invoke GlobalUnlock,pMemory  
                    invoke GlobalFree,hMemory  
                .endif  
                invoke SetFocus,hwndEdit  
            .elseif ax==IDM_SAVE  
                mov ofn.Flags,OFN_LONGNAMES or\  
                                OFN_EXPLORER or OFN_HIDEREADONLY  
                invoke GetSaveFileName, ADDR ofn  
                    .if eax==TRUE  
                        invoke CreateFile,ADDR buffer,\  
                                                GENERIC_READ or GENERIC_WRITE ,\  
                                                FILE_SHARE_READ or FILE_SHARE_WRITE,\  
                                                NULL,CREATE_NEW,FILE_ATTRIBUTE_ARCHIVE,\  
                                                NULL  
                        mov hFile,eax  
                        invoke GlobalAlloc,GMEM_MOVEABLE or GMEM_ZEROINIT,MEMSIZE  
                        mov  hMemory,eax  
                        invoke GlobalLock,hMemory  
                        mov  pMemory,eax  
                        invoke SendMessage,hwndEdit,WM_GETTEXT,MEMSIZE-1,pMemory  
                        invoke WriteFile,hFile,pMemory,eax,ADDR SizeReadWrite,NULL  
                        invoke CloseHandle,hFile  
                        invoke GlobalUnlock,pMemory  
                        invoke GlobalFree,hMemory  
                    .endif  
                    invoke SetFocus,hwndEdit  
                .else  
                    invoke DestroyWindow, hWnd  
                .endif  
            .endif  
        .ELSE  
            invoke DefWindowProc,hWnd,uMsg,wParam,lParam  
            ret  
.ENDIF  
xor    eax,eax  
ret  
WndProc endp  
end start  

  
  

  :بررسي كد فوق
  
     

       invoke CreateWindowEx,NULL,ADDR EditClass,NULL,\    
                   WS_VISIBLE or WS_CHILD or ES_LEFT or ES_MULTILINE or\  
                   ES_AUTOHSCROLL or ES_AUTOVSCROLL,0,\  
                   0,0,0,hWnd,EditID,\  
                   hInstance,NULL  
        mov hwndEdit,eax  

  
 ,x, yلازم بـه ذكـر اسـت كـه مقـادير      . يك كنترل تكست باكس را ايجاد خواهيم كرد،  WM_CREATEدر قسمت ما    

width,heightكل پنجره را توسط آن خواهيم پوشاند، ند زيرا در ادامه با تغيير اندازه آن  آن مساوي صفر وارد شده ا .  

  جهـت  ShowWindow به فراخواني تابع   ي براي تكست باكس ، ديگر نياز      WS_VISIBLE   همچنين با بكارگيري حالت     
  :ه ادامدر . توان استفاده كرد از اين نكته براي پنجره والد نيز مي. داشتآن نخواهيم نمايش 

  
;==============================================  
;        Initialize the members of OPENFILENAME structure  
;==============================================  
        mov ofn.lStructSize,SIZEOF ofn  
        push hWnd  
        pop  ofn.hWndOwner  
        push hInstance  
        pop  ofn.hInstance  
        mov  ofn.lpstrFilter, OFFSET FilterString  
        mov  ofn.lpstrFile, OFFSET buffer  
        mov  ofn.nMaxFile,MAXSIZE  

  
  نيـز  save as dialog boxمـايش  از آنجائيكه از ايـن سـاختار در ن  . را مقدار دهي اوليه خواهيم كردofn   در ادامه اعضاي 

  . مقدار دهي شدندGetSaveFileName و GetOpenFileNameاستفاده خواهيم كرد ، تنها مقادير مشترك بين توابع 
  :سپس . باشد  مكان بسيار مناسبي براي مقدار دهي اوليه متغيرها ميWM_CREATE   قسمت 

    .ELSEIF uMsg==WM_SIZE  
        mov eax,lParam  
        mov edx,eax  
        shr edx,16  
        and eax,0ffffh  
        invoke MoveWindow,hwndEdit,0,0,eax,edx,TRUE  

  
همچنـين ايـن پيغـام در    . صـادر خواهـد شـد   WM_SIZE   هنگاميكه اندازه ناحيه كلاينت پنجره برنامه تغييـر كنـد ، پيغـام             

 CS_VREDRAWبراي دريافت اين پيغام ، كـلاس پنجـره بايـد داراي حالتهـاي     . شود  ميابتداي ايجاد پنجره نيز دريافت

and CS_HREDRAWما از اين پيغام جهت تغيير اندازه اديت باكس برنامه به اندازه ناحيه كلاينـت پنجـره كمـك    .  باشد 
طـول و    مـساوي  lParamوط بـه   مربhigh wordمقدار . و عرض جاري پنجره را بدست آورددر ابتدا بايد طول. مي گيريم
low wordسپس از اين اطلاعات در تابع .  آن معادل عرض پنجره استMoveWindow     جهت تغييـر انـدازه اديـت بـاكس 
  :در ادامه. گيريم كمك مي

                .if ax==IDM_OPEN  
                mov  ofn.Flags, OFN_FILEMUSTEXIST or \  
                                OFN_PATHMUSTEXIST or OFN_LONGNAMES or\  
                                OFN_EXPLORER or OFN_HIDEREADONLY  
                invoke GetOpenFileName, ADDR ofn  

  :سپس . دهيم مايش مي را نOpenFile، ديالوگ باكس    هنگاميكه كاربر از منوي برنامه گزينه گشودن را انتخاب كند 

                    .if eax==TRUE    
                    invoke CreateFile,ADDR buffer,\  
                                GENERIC_READ or GENERIC_WRITE ,\  
                                FILE_SHARE_READ or FILE_SHARE_WRITE,\  
                                NULL,OPEN_EXISTING,FILE_ATTRIBUTE_ARCHIVE,\  
                                NULL  
                    mov hFile,eax  

  
همچنـين در اينجـا مـشخص    . جهـت گـشودن آن اسـتفاده خـواهيم كـرد      ، CreateFile   پس از انتخاب فايـل ، از تـابع            

پـس از گـشودن فايـل ، دسـتگيره اي بـه آن را در يـك       . ايل را جهت خواندن و نوشتن در آن گشوده ايـم          كنيم كه ف    مي
  :به صورت زير است تعريف اين تابع . دهيم متغير عمومي جهت استفاده هاي بعدي قرار مي

CreateFile proto lpFileName:DWORD,\  
                           dwDesiredAccess:DWORD,\  
                           dwShareMode:DWORD,\  
                           lpSecurityAttributes:DWORD,\  
                           dwCreationDistribution:DWORD\,  
                           dwFlagsAndAttributes:DWORD\,  
                           hTemplateFile:DWORD  

dwDesiredAccess :      اگـر مـساوي   . مقدار صفر آن به معناي سطح دسترسي خوانـدن و نوشـتن اسـت
GENERIC_READ         ــدن ــت خوان ــرفا جه ــل ص ــرد ، فاي ــرار گي ــي ، ق ــاز م ــساوي    ب ــر م ــود و اگ ش
GENERIC_WRITE شود باز مي، ، فايل براي نوشتن در آن   قرار گيرد .  

dwShareMode :   و يا عدم دسترسي ساير پروسه هاي فعال را در مورد عمليات بر روي فايـل       دسترسي
اگر مساوي صفر قرار گيرد ايـن فايـل بـا سـاير پروسـه هـا بـه اشـتراك           . كند گشوده شده ، مشخص مي    

خوانـدن از  قرار گيرد، به ساير پروسه هـا اجـازه      FILE_SHARE_READاگر مساوي. شود گذاشته نمي 
هـا اجـازه    ديگر پروسه  قرار گيرد     FILE_SHARE_WRITEاهد شد و اگر مساوي    فايل گشوده داده خو   

  . نوشتن در فايل گشوده شده در برنامه را مي يابند
lpSecurityAttributes : بيتي اهميتي ندارد۳۲ تحت ويندوزهاي .  

dwCreationDistribution:             در  عملياتي را كه تابع در صورت وجود و يا عدم وجود فايـل ذكـر شـده 
lpFileNameاين موارد به شرح زير هستند.  انجام دهد ، مشخص مي كند بايد:  

CREATE_NEW :      اگـر فايـل از پـيش وجـود     . فايل جديـدي را ايجـاد خواهـد كـرد
  .داشته باشد تابع با يك خطا متوقف مي شود

CREATE_ALWAYS :      اگـر ايـن فايـل از پـيش     .فايل جديدي را ايجاد خواهـد كـرد
  . گردد گزيني مي باشد ، جايوجود داشته 

OPEN_EXISTING :   اگر فايل از پيش وجود نداشته باشد تـابع بـا      .گشايد  فايل را مي 
  .يك خطا متوقف مي شود

OPEN_ALWAYS :      اگر فايـل موجـود نباشـد و        .گشايد  فايل را در صورت وجود مي
  .فايل جديدي ايجاد خواهد شد  پيشتر مشخص شده باشد ، CREATE_NEWحالت 

TRUNCATE_EXISTING :   سپس كل محتويات آنرا تخليه كرده       .گشايد  فايل را مي 
ســـطح دسترســـي در ايـــن حالـــت بايـــد  .شـــود و انـــدازه فايـــل صـــفر بايـــت مـــي 

GENERIC_WRITE اگر فايل موجود نباشد تابع با يك خطـا متوقـف خواهـد             .  باشد
  . شد

  
dwFlagsAndAttributes : ادي، فقط خواندني و غيره مي باشدبيانگر خواص فايل مانند آرشيو ، ع.  

  
  :   در ادامه 

                    invoke  GlobalAlloc,GMEM_MOVEABLE or GMEM_ZEROINIT,MEMSIZE  
                    mov   hMemory,eax  
                    invoke  GlobalLock,hMemory  
                    mov   pMemory,eax  

 
.  تخصيص مي دهـيم ReadFile and WriteFile  هنگاميكه فايلي گشوده شد ، قطعه اي از حافظه را براي استفاد در توابع  

 ، به ويندوز اجازه انتقال قطعـه حافظـه مـورد اسـتفاده جهـت يكپـارچگي بيـشتر            GMEM_MOVEABLEبا مشخص كردن    
  . شود افظه تخصيص داده شده در ابتدا با صفر پر ميقطعه ح ، GMEM_ZEROINITبا تنظيم پرچم . دهيم حافظه را مي

تخـصيص داده شـده   قطعـه حافظـه   دسـتگيره اي بـه    حـاوي  eax ، رجيستر GlobalAlloc   پس از به پايان رسيدن كار تابع     
  :سپس.اشاره گري به قطعه حافظه بدست خواهد آمد ، GlobalLockسپس با فراخواني . خواهد بود

                    invoke ReadFile,hFile,pMemory,MEMSIZE-1,ADDR SizeReadWrite,NULL  
                    invoke SendMessage,hwndEdit,WM_SETTEXT,NULL,pMemory  

 فايـل  زمانيكـه . كنـيم   جهت خواندن اطلاعات از فايـل اسـتفاده مـي   ReadFileاز تابع    هنگاميكه قطعه حافظه آماده شد ،       
اولـين  . بنابراين آماده خوانـدن اولـين بايـت فايـل بـه بعـد هـستيم       . گر آن در آفست صفر قرار دارد د ، اشارهشو گشوده مي 
دومين پارامتر اشاره گري اسـت بـه قطعـه اي    .  ، دستگيره اي به فايل مورد نظر جهت خواندن است          ReadFileپارامتر تابع   

تر بعـدي تعــداد بايتهـايي اسـت كــه بايـد از فايـل خوانــده      پـارام . از حافظـه كـه داده هـا را در خــود ذخيـره خواهـد كــرد     
  .گيرد تعداد بايتهاي واقعي خوانده شده از فايل درون آن قرار مي، آدرس متغيري است كه چهارمين آرگومان .شوند

 ، اديـت بـاكس را مقـدار دهـي خـواهيم      WM_SETTEXTتوسط ارسـال پيغـام   ها ،       پس از پركردن قطعه حافظه از داده      
داده هـاي خوانـده    پس از اين ، اديت باكس  . مورد استفاده  قطعه حافظه  حاوي اشاره گري است به       lParamدراينجا  . كرد

  . دهد شده را نمايش مي

                    invoke CloseHandle,hFile  
                    invoke GlobalUnlock,pMemory  

                    invoke GlobalFree,hMemory  
                .endif  

قصد ذخيره سازي داده هاي ويرايش شده در اديـت بـاكس را در فـايلي ديگـر داريـم ، بنـابراين فايـل            با توجه به اينكه   
تـوان حافظـه اختـصاص     البتـه مـي  . كنـيم  مورد نظر را پس از خواندن ، خواهيم بست و منابع تخصيص يافته را نيز آزاد مي         

  :در ادامه. ها مجددا استفاده كرد  در اينجا آزاد نكرد و از آن در قسمت ذخيره سازي دادهيافته را

                invoke SetFocus,hwndEdit  

  .    پس از نمايش پنجره ، تمركز ورودي به اديت باكس منتقل خواهد شد
 نمـايش  ."..ذخيـره كـردن بعنـوان   "يـالوگ بـاكس      هنگاميكه ذخيره كردن فايل ، پس از انتخاب گزينه مربوطه از منو، د           

  :داده خواهد شد
mov ofn.Flags,OFN_LONGNAMES or\  
                                OFN_EXPLORER or OFN_HIDEREADONLY  

  
 OFN_FILEMUSTEXIST and OFN_PATHMUSTEXISTهـاي    پرچم    از آنجائيكه قصد ايجاد فايلي جديد را داريم ،

 dwCreationDistribution و همچنين پـارامتر  )زيرا اجازه ايجاد فايلي كه وجود ندارد را نخواهد داد          (ار نخواهيم برد  را بك 
تنهـا تفـاوت در   .  باقيمانده كد تفاوتي بـا آنچـه كـه گذشـت نـدارد            . تنظيم شود  CREATE_NEW بايد به    CreateFileتابع  

  :قسمت زير است

                        invoke SendMessage,hwndEdit,WM_GETTEXT,MEMSIZE-1,pMemory  
                        invoke WriteFile,hFile,pMemory,eax,ADDR SizeReadWrite,NULL  

خروجـي  .  جهت دريافت اطلاعات از اديت باكس و انتقال آن به حافظه استفاده شـده اسـت           WM_GETTEXTپيغام  از     
ها درحافظه ، آنها را در فايل خـواهيم   پس از قرار گرفتن داده. اندازه داده ها درون بافر است رفته ،    قرار گ  eaxتابع كه در    

  . نوشت

  Memory Mapped Files -۱۱مثال 
  

 ،اگر حجم فايل مورد نظـر بـراي خوانـدن   .    اگر مثال برنامه قبل را بررسي كرده باشيد احتمالا نقايصي را در آن يافته ايد         
و يا اگر رشته اي را كه بدنبال آن هستيد ، نيمـه اول آن در    ،   اندازه قطعه حافظه تخصيص يافته شده براي آن بيشتر باشد          از

   ؟چكار بايد كرد  انتهاي قطعه حافظه قرار گرفته باشد ،
 ديگـري خوانـد و      پاسخ سنتي به سوال اول اين است كه بايد به صورت مداوم تا انتهاي فايـل ، داده هـا را يكـي پـس از      

  . موارد لازم را در نظر داشت،پاسخ سوال دوم اين است كه بايد براي حالت خاص انتهاي فايل. مورد استفاده قرار داد
   شايد بتوان قطعه بسيار بزرگي از حافظه را براي ذخيره كردن محتويات كل يك فايل تخصيص داد ، اما به يـك چنـين                 

 Fileبا كمك . راه حلي براي اين موضوع است) نگاشت فايل( File mapping .شود ته مي گفresource hog ، هايي برنامه

mapping  و با استفاده از يـك اشـاره گـر    توان تصور كرد كه كل فايل هم اكنون بدرون حافظه بارگذاري شده است      مي 
 APIتفاده از توابـع مـديريت حافظـه    ديگـر نيـازي بـه اس ـ   . توان اطلاعاتي را از آن خواند و يا در آن ذخيره كرد            حافظه مي 

تـوان   ها بـين پروسـه هـاي مختلـف نيـز مـي        براي به اشتراك گذاشتن دادهFile mappingهمچنين از . ويندوز نخواهد بود
 ويل واقعي ديگر درگير نخواهـد بـود و ايـن حالـت شـبيه بـه قطعـه حافظـه رزر         ، فاFile mappingبا كمك . استفاده كرد

البته در اين حالت براي به اشتراك گذاشـتن داده هـا، بايـد بـه نكـات      . توانند آنرا ببينند   ها مي  ام پروسه است كه تم    اي شده
  .مربوط به تردها و همزماني بين آنها دقت داشت ، درغير اينصورت برنامه در مدت كوتاهي از كار خواهد افتاد

  ، File mappingتوسـط  ) افظه به اشتراك گذاشته شـده ناحيه ح  (shared memory region   در اين مثال در مورد ايجاد 
  ، از روش PE loaderبراي مثـال يـك   . آشنايي با نحوه نگاشت فايل در حافظه است، بلكه هدف ما . بحث نخواهيم كرد

ا اين روش بـسيار مناسـب اسـت زيـرا تنه ـ    . كند براي بارگذاري فايلهاي اجرايي به درون حافظه استفاده مي   ،  نگاشت فايل   
  . قسمت مورد نياز از فايل خوانده خواهد شد

گـردد ،   مـي نگاشـت  هنگاميكه يك فايـل بـه درون حافظـه    . هاي خود را نيز دارا است      ها محدوديت      روش نگاشت فايل  
عمليـاتي  بنابراين روش نگاشت فايلها براي كار با فايلهاي فقـط خوانـدني و يـا    . اندازه آن در حين عمليات نبايد تغيير كند 

  .آل است ، بسيار ايده گذارند كه بر روي حجم فايل تاثيري نمي
تـوان از ايـن روش اسـتفاده نمـود و بـديهي اسـت كـه در ايـن          البته در حالتهايي كه اندازه فايل تغيير مي كننـد نيـز مـي            

  . يد ايجاد كردها بايد اندازه جديد را محاسبه نمود و فايل نگاشت شده بدرون حافظه را بر اساس اندازه جد حالت
  

  :مراحل زير بايد صورت گيرند، ها بدرون حافظه      براي استفاده از روش نگاشت فايل
 .شود   فايل مورد نظر براي نگاشت ، گشوده ميCreateFileبا فراخواني تابع  §

 ايـن   . با استفاده از دستگيره بدست آمـده از مرحلـه قبـل            CreateFileMappingفراخواني   §
 .نمايد  فايل را از فايل گشوده شده در قسمت قبل ، ايجاد ميتابع شيء نگاشت

 جهــت نگاشــت كــل و يــا قــسمتي انتخــابي از فايــل بــدورن MapViewOfFileفراخــواني  §
 .گرداند اين تابع اشاره گري را به ابتداي ناحيه نگاشت شده باز مي. حافظه

 لگر فوق جهت خواندن و يا نوشتن داده ها در فاي استفاده از اشاره §

  براي پايان دادن به نگاشت فايلUnmapViewOfFileفراخواني  §

 .شود  ، فايل نگاشت شده بسته ميCloseHandleبا كمك تابع  §

   .CreateFile جهت بستن فايل اصلي باز شده توسط تابع CloseHandleفراخواني  §
  

  :كد برنامه
  

.386  

.model flat,stdcall  
WinMain proto :DWORD,:DWORD,:DWORD,:DWORD  
include \masm32\include\windows.inc  
include \masm32\include\user32.inc  
include \masm32\include\kernel32.inc  
include \masm32\include\comdlg32.inc  
includelib \masm32\lib\user32.lib  
includelib \masm32\lib\kernel32.lib  
includelib \masm32\lib\comdlg32.lib  

.const  
IDM_OPEN equ 1  
IDM_SAVE equ 2  
IDM_EXIT equ 3  
MAXSIZE equ 260  

.data  
ClassName db "Win32ASMFileMappingClass",0  
AppName  db "Win32 ASM File Mapping Example",0  
MenuName db "FirstMenu",0  
ofn   OPENFILENAME <>  
FilterString db "All Files",0,"*.*",0  
             db "Text Files",0,"*.txt",0,0  
buffer db MAXSIZE dup(0)  
hMapFile HANDLE 0                            ; Handle to the memory mapped file, must be  
                                                                    ;initialized with 0 because we also use it as  
                                                                    ;a flag in WM_DESTROY section too  

.data?  
hInstance HINSTANCE ?  
CommandLine LPSTR ?  
hFileRead HANDLE ?                               ; Handle to the source file  
hFileWrite HANDLE ?                                ; Handle to the output file  
hMenu HANDLE ?  
pMemory DWORD ?                                 ; pointer to the data in the source file  
SizeWritten DWORD ?                               ; number of bytes actually written by WriteFile  

.code  
start:  
        invoke GetModuleHandle, NULL  

        mov    hInstance,eax  
        invoke GetCommandLine  
        mov CommandLine,eax  
        invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT  
        invoke ExitProcess,eax  

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:DWORD  
    LOCAL wc:WNDCLASSEX  
    LOCAL msg:MSG  
    LOCAL hwnd:HWND  
    mov   wc.cbSize,SIZEOF WNDCLASSEX  
    mov   wc.style, CS_HREDRAW or CS_VREDRAW  
    mov   wc.lpfnWndProc, OFFSET WndProc  
    mov   wc.cbClsExtra,NULL  
    mov   wc.cbWndExtra,NULL  
    push  hInst  
    pop   wc.hInstance  
    mov   wc.hbrBackground,COLOR_WINDOW+1  
    mov   wc.lpszMenuName,OFFSET MenuName  
    mov   wc.lpszClassName,OFFSET ClassName  
    invoke LoadIcon,NULL,IDI_APPLICATION  
    mov   wc.hIcon,eax  
    mov   wc.hIconSm,eax  
    invoke LoadCursor,NULL,IDC_ARROW  
    mov   wc.hCursor,eax  
    invoke RegisterClassEx, addr wc  
    invoke CreateWindowEx,WS_EX_CLIENTEDGE,ADDR ClassName,\  
                ADDR AppName, WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\  
               CW_USEDEFAULT,300,200,NULL,NULL,\  
    hInst,NULL  
    mov   hwnd,eax  
    invoke ShowWindow, hwnd,SW_SHOWNORMAL  
    invoke UpdateWindow, hwnd  
    .WHILE TRUE  
        invoke GetMessage, ADDR msg,NULL,0,0  
        .BREAK .IF (!eax)  
        invoke TranslateMessage, ADDR msg  
        invoke DispatchMessage, ADDR msg  
    .ENDW  
    mov     eax,msg.wParam  
    ret  
WinMain endp  

WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM  
    .IF uMsg==WM_CREATE  
        invoke GetMenu,hWnd                       ;Obtain the menu handle  
        mov  hMenu,eax  
        mov ofn.lStructSize,SIZEOF ofn  
        push hWnd  
        pop  ofn.hWndOwner  
        push hInstance  
        pop  ofn.hInstance  
        mov  ofn.lpstrFilter, OFFSET FilterString  
        mov  ofn.lpstrFile, OFFSET buffer  
        mov  ofn.nMaxFile,MAXSIZE  
    .ELSEIF uMsg==WM_DESTROY  
        .if hMapFile!=0  
            call CloseMapFile  
        .endif  
        invoke PostQuitMessage,NULL  

    .ELSEIF uMsg==WM_COMMAND  
        mov eax,wParam  
        .if lParam==0  
            .if ax==IDM_OPEN  
                mov  ofn.Flags, OFN_FILEMUSTEXIST or \  
                                OFN_PATHMUSTEXIST or OFN_LONGNAMES or\  
                                OFN_EXPLORER or OFN_HIDEREADONLY  
                                invoke GetOpenFileName, ADDR ofn  
                .if eax==TRUE  
                    invoke CreateFile,ADDR buffer,\  
                                                GENERIC_READ ,\  
                                                0,\  
                                                NULL,OPEN_EXISTING,FILE_ATTRIBUTE_ARCHIVE,\  
                                                NULL  
                    mov hFileRead,eax  
                    invoke CreateFileMapping,hFileRead,NULL,PAGE_READONLY,0,0,NULL  
                    mov     hMapFile,eax  
                    mov     eax,OFFSET buffer  
                    movzx  edx,ofn.nFileOffset  
                    add      eax,edx  
                    invoke SetWindowText,hWnd,eax  
                    invoke EnableMenuItem,hMenu,IDM_OPEN,MF_GRAYED  
                    invoke EnableMenuItem,hMenu,IDM_SAVE,MF_ENABLED  
                .endif  
            .elseif ax==IDM_SAVE  
                mov ofn.Flags,OFN_LONGNAMES or\  
                                OFN_EXPLORER or OFN_HIDEREADONLY  
                invoke GetSaveFileName, ADDR ofn  
                .if eax==TRUE  
                    invoke CreateFile,ADDR buffer,\  
                                                GENERIC_READ or GENERIC_WRITE ,\  
                                                FILE_SHARE_READ or FILE_SHARE_WRITE,\  
                                                NULL,CREATE_NEW,FILE_ATTRIBUTE_ARCHIVE,\  
                                                NULL  
                    mov hFileWrite,eax  
                    invoke MapViewOfFile,hMapFile,FILE_MAP_READ,0,0,0  
                    mov pMemory,eax  
                    invoke GetFileSize,hFileRead,NULL  
                    invoke WriteFile,hFileWrite,pMemory,eax,ADDR SizeWritten,NULL  
                    invoke UnmapViewOfFile,pMemory  
                    call   CloseMapFile  
                    invoke CloseHandle,hFileWrite  
                    invoke SetWindowText,hWnd,ADDR AppName  
                    invoke EnableMenuItem,hMenu,IDM_OPEN,MF_ENABLED  
                    invoke EnableMenuItem,hMenu,IDM_SAVE,MF_GRAYED  
                .endif  
            .else  
                invoke DestroyWindow, hWnd  
            .endif  
        .endif  
    .ELSE  
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam  
        ret  
    .ENDIF  
    xor    eax,eax  
    ret  
WndProc endp  

CloseMapFile PROC  
        invoke CloseHandle,hMapFile  
        mov    hMapFile,0  

        invoke CloseHandle,hFileRead  
        ret  
CloseMapFile endp  

end start 
  

  :بررسي كد فوق
  

invoke CreateFile,ADDR buffer,\  
                                                GENERIC_READ ,\  
                                                0,\  
                                                NULL,OPEN_EXISTING,FILE_ATTRIBUTE_ARCHIVE,\  
                                                NULL  

  
از .  كـرد كند ، سعي در بـازكردن آن خـواهيم      هنگاميكه كاربر فايلي را توسط ديالوگ باكس گشودن فايل انتخاب مي       

 dwShareModeگشودن فايل به صورت فقط خوانـدني اسـتفاده شـده و بـا قـرار دادن       جهت  ،   GENERIC_READپرچم  
  :در ادامه.شود  داده نمي،ي تغيير فايل در حين انجام عمليات  به ساير پروسه ها اجازهصفر ،  مساوي

  
invoke CreateFileMapping,hFileRead,NULL,PAGE_READONLY,0,0,NULL  

  
 تعريـف  .كنـيم   ايجـاد مـي  ،   ، فايلي نگاشت شده به حافظهCreateFileMappingتوسط تابع      سپس از فايل گشوده شده،    
  :اين تابع به صورت زير است

CreateFileMapping proto hFile:DWORD,\  
                                         lpFileMappingAttributes:DWORD,\  
                                         flProtect:DWORD,\  
                                         dwMaximumSizeHigh:DWORD,\  
                                         dwMaximumSizeLow:DWORD,\  
                                         lpName:DWORD  

 dwMaximumSizeHighتوسط پارامترهـاي  .كند ر است كه تابع فوق كل فايل را بدرون حافظه نگاشت نمي   لازم به ذك

and dwMaximumSizeLow،  گردد  قسمت مورد نياز انتخاب و بدرون حافظه نگاشت مي .  
بـا  . ه بـسط داده خواهـد شـد   به اندازه ذكر شد، فايل واقعي    اگر اندازه اي بزرگتر از اندازه فايل واقعي را انتخاب نماييد            

  . قرار دادن صفر بعنوان هر دو پارامتر ذكر شده ، اندازه فايل واقعي و فايل نگاشت شده يكي خواهد گرديد
  

، فايـل نگاشـت شـده بـا خـواص امنيتـي پـيش فـرض          قـرار گيـرد   NULL مساوي   lpFileMappingAttributes   اگر پارامتر   
  . شود ويندوز ايجاد مي

 PAGE_READONLYبراي مثال مقـدار  . كند تعيين مي  نوع محافظت مورد نياز براي فايل مپ شده را            flProtect    پارامتر
لازم بـه ذكـر اسـت كـه ايـن خاصـيت بـا         . شود تا فايل نگاشت شده به صورت فقط خواندني قابل استفاده شـود             سبب مي 

فراخـواني تـابع نگاشـت فايـل بـا شكـست       ينـصورت  در غيرا.  نبايد متنـاقض باشـد   CreateFileخاصيت ذكر شده در تابع     
  .مواجه خواهد شد

  
اگر مي خواهيـد ايـن فايـل را بـا سـاير پروسـه هـا بـه          . كند  اشاره مي در حافظه   نام فايل نگاشت شده      به   lpName   پارامتر  

اي اسـت كـه از فايـل    اما در مثال مـا برنامـه ايجـاد شـده تنهـا برنامـه          . نماييداشتراك بگذاريد بايد نامي براي آن مشخص        
  : در ادامه.استفاده خواهد كرد بنابراين از اين پارامتر صرفنظر خواهيم كرد

                    mov     eax,OFFSET buffer  
                    movzx  edx,ofn.nFileOffset  
                    add      eax,edx  
                    invoke SetWindowText,hWnd,eax  

  
نـام و مـسير كامـل فايـل در     . دهـيم     پس از انجام موفقيت آميز مراحل قبل ، عنوان پنجره را به نام فايل باز شده تغيير مـي           

بنـابراين بايـد مقـدار    خـواهيم نمـايش دهـيم ،      فايـل را در عنـوان پنجـره مـي         از آنجائيكه تنها نـام      . بافر ذخيره خواهند شد   
nFileOffset تار  ساخOPENFILENAME سپس. اضافه نماييم را به آدرس بافر:   

                    invoke EnableMenuItem,hMenu,IDM_OPEN,MF_GRAYED  
                    invoke EnableMenuItem,hMenu,IDM_SAVE,MF_ENABLED  

  
منـوي   ربر چندين فايل را با هـم بـاز كنـد بنـابراين     پس از گشودن فايل اوليه ، نمي خواهيم كا    بعنوان اقدامي احتياطي ،     

و اگـر كـاربر قـصد    .كنـيم  غيرفعال و منوي ذخيره كردن فايـل را فعـال مـي           EnableMenuItemبا كمك   گشودن فايل را    
  :بستن برنامه را داشت به صورت زير منابع تخصيص يافته را بايد آزاد نمائيم

    .ELSEIF uMsg==WM_DESTROY  
        .if hMapFile!=0  
            call CloseMapFile  
        .endif  
        invoke PostQuitMessage,NULL  

  
  : فوق به صورت زير تعريف شده استدستورات در CloseMapFile  تابع 

CloseMapFile PROC  
        invoke CloseHandle,hMapFile  
        mov    hMapFile,0  
        invoke CloseHandle,hFileRead  
        ret  
CloseMapFile endp  

ذخيـره سـازي   " نمايش ديـالوگ بـاكس اسـتاندارد    پس ازها را از منو انتخاب كند ،      اگر كاربر گزينه ذخيره سازي داده     
  :مه براي ذخيره سازيكنيم و در ادا ، از كاربر نام فايل جديدي را براي ذخيره سازي داده ها دريافت مي...." بعنوان 

  
invoke MapViewOfFile,hMapFile,FILE_MAP_READ,0,0,0  
mov pMemory,eax  

  
قـسمت دلخـواهي از فايـل نگاشـت شـده بـه        ، MapViewOfFile   بالافاصله پس از ايجاد فايل خروجي ، به كمـك تـابع            

  :شود اين تابع به صورت زير تعريف مي. كنيم حافظه را به حافظه نگاشت مي

MapViewOfFile proto hFileMappingObject:DWORD,\  
                                   dwDesiredAccess:DWORD,\  
                                   dwFileOffsetHigh:DWORD,\  
                                   dwFileOffsetLow:DWORD,\  
                                   dwNumberOfBytesToMap:DWORD  

  :   توضيحات پارامترهاي آن به صورت زير است
dwDesiredAccess :           بـراي  . عملياتي را كه مـي خـواهيم روي فايـل انجـام دهـيم مـشخص مـي كنـد

  .گردد   استفاده ميFILE_MAP_READحالت فقط خواندني از 
dwFileOffsetHigh and dwFileOffsetLow :   خـواهيم   آغـازين قـسمتي از فايـل را كـه مـي     آفـست

در ايـن مثـال چـون كـل فايـل را مـي خـواهيم مـورد            . كننـد  بدرون حافظه نگاشت كنيم ، مشخص مـي       
  . ايم  استفاده قرار دهيم ، آنها را مساوي صفر قرار داده

dwNumberOfBytesToMap :   ــدرون حافظــه نگاشــت كــرد مــشخص  تعــداد بايتهــايي را كــه بايــد ب
  .شود با مساوي صفر قرار دادن اين آرگومان ، كل فايل خوانده مي. كنند مي

  
گـري   حاصل اين فراخواني اشـاره .شود بدرون حافظه بارگذاري مي ، قسمت دلخواه     MapViewOfFile   پس از فراخواني    

  :سپس . به قطعه حافظه براي كار با آن خواهد بود

                    invoke GetFileSize,hFileRead,NULL  

 گيگابايـت بيـشتر   ۴اگـر انـدازه فايـل از    . گيـرد  قـرار مـي    eaxخروجي تابع در    آمده و    اندازه فايل به اين ترتيب بدست       
 از آنجائيكـه قـصد كـار بـا اينگونـه فايلهـاي       .گـردد   ذخيره ميFileSizeHighWord  اندازه فايل درhigh DWORD،  باشد

  : در ادامه.ظر كردحجيم را نداريم مي توان از آن صرفن

                    invoke WriteFile,hFileWrite,pMemory,eax,ADDR SizeWritten,NULL  

سـپس منـابع تخـصيص داده شـده آزاد     . بـدرون فايـل اسـتفاده شـد     حافظـه ،  درهاي نگاشت شـده   دادهاز آن براي نوشتن   
  :گردند مي

                    invoke UnmapViewOfFile,pMemory  

                    call   CloseMapFile  
                    invoke CloseHandle,hFileWrite  

  :   بازيابي عنوان اصلي پنجره
invoke SetWindowText,hWnd,ADDR AppName  

  
  .كنيم ذخيره كردن منو را غيرفعال و گزينه گشودن فايل را فعال مي    سپس گزينه 

                    invoke EnableMenuItem,hMenu,IDM_OPEN,MF_ENABLED  
                    invoke EnableMenuItem,hMenu,IDM_SAVE,MF_GRAYED  

  پروسه -۱۲مثال 
  

  چيست؟) process(پروسه 
و همچنـين شـامل سـاير     ك تـرد ، ي ـپروسه برنامه اي است در حال اجرا ، حاوي فضاي آدرس دهي مجازي اختـصاصي       

  . دنمي باش قابل مشاهده آنكه براي منابع سيستمي 
  

 )هـاي (مـاژول  فـضاي آدرس دهـي ،   : ، يك پروسه شامل چندين شيء است مشخص است    همانگونه كه از تعريف نيز  
حداقل داراي يك تـرد  يد باهر پروسه . كند اجرايي و هر چيزي كه يك فايل اجرايي آنرا ايجاد و يا جهت استفاده باز مي       

، بـه ازاي  كنـد   هنگاميكه ويندوز پروسه اي را ايجاد مـي . ترد در حقيقت صف اجرايي است   چيست؟  ) thread(ترد  . باشد
مـاژول شـروع بـه كـار     ) اينستراكـشن (اين ترد از اجراي اولـين دسـتورالعمل   . خلق خواهد كردهر پروسه تنها يك ترد را     

  . تواند ايجاد نمايد از به تردهاي بيشتري داشته باشد ، آنها را مياگر يك پروسه ني. خواهد كرد
فضاي آدرس دهي اختصاصي براي اين تـرد ايجـاد   كند ،  ي بر ايجاد يك ترد دريافت مي      ن   زمانيكه ويندوز پيغامي را مبت    

ن پروسـه ايجـاد   سـپس تـردي اوليـه بـراي اي ـ     .نمايـد  و سپس فايل اجرايي را به اين فضاي آدرسي دهي نگاشت مـي            كرده  
  .گردد مي

  
ون برنامه هـا ، پروسـه اي جديـد را ايجـاد     راز د ، CreateProcessتوان با فراخواني تابع   بيتي ، مي۳۲   تحت ويندوزهاي   

  :تعريف اين تابع به صورت زير است. نمود

CreateProcess proto lpApplicationName:DWORD,\  
                                 lpCommandLine:DWORD,\ 
                                 lpProcessAttributes:DWORD,\  
                                 lpThreadAttributes:DWORD,\  
                                 bInheritHandles:DWORD,\  
                                 dwCreationFlags:DWORD,\  
                                 lpEnvironment:DWORD,\  
                                 lpCurrentDirectory:DWORD,\  
                                 lpStartupInfo:DWORD,\  
                                 lpProcessInformation:DWORD  

lpApplicationName :         اگـر ايـن     .)با و يا بدون مـسير فايـل        (نام فايل اجرايي مورد نظر براي اجرا است
  مـشخص  lpCommandLine قرار دهـيم، بايـد نـام فايـل اجرايـي را در پـارامتر         NULLپارامتر را مساوي    

  .نمود
lpCommandLine : ــال .  كــه برنامــه دريافــت و پــردازش مــي كنــد يآرگومــان خــط فرمــان ــراي مث : ب

notepad.exe readme.txt  
lpProcessAttributes and lpthreadAttributes:   كنـد  خواص امنيتي پروسه و ترد اوليه را مشخص مـي .

    .گيرند خواص امنيتي پيش فرض ويندوز مورد استفاده قرار مي وارد شوند ، NULLاگر اين دو پارامتر 

bInheritHandles :    بـاز شـده    هـاي    ديـد تمـام دسـتگيره      آيا پروسه ج   درجهت تعيين اينكه  پرچمي است
  .لي را به ارث ببرد يا خيرتوسط پروسه فع

dwCreationFlags : بـراي مثـال بلافاصـله    . كننـد  رفتار پروسه ايجاد شده را مشخص مـي     هايي كه     پرچم
همچنـين حـق تقـدم    . پس از ايجاد پروسه، معلق گرديده ، اصلاح شده و سپس مورد استفاده قرار گيرد         

 مـورد  NORMAL_PRIORITY_CLASS و بـه صـورت معمـول    تـوان تعيـين كـرد     نيـز مـي  تردهاي آنـرا  
  . گيرد استفاده قرار مي
lpEnvironment :    اگـر مـساوي   . اشاره گري است به قطعه محيطـيNULL     قـرار گيـرد ، ايـن خـواص از 

  . برنامه والد به ارث خواهند رسيد
lpCurrentDirectory : ري پروسه اي كـه ايجـاد خواهـد شـد     دايركتوري جاگري است به  اشاره)child 

process .(خواهيد كه اين مورد را نيز از برنامه والد به ارث ببرد آنرا مساوي  اگر ميNULL  قرار دهيد .  
lpStartupInfo : گري است به ساختار      اشارهSTARTUPINFO      اصـلي    كـه بيـانگر نحـوه نمـايش پنجـره

تـوان ايـن پــارامتر را بـا كمـك تــابع       نظـر نيـست ، مــي  اگــر حالـت ويـژه اي مــد  . پروسـه جديـد اسـت   
GetStartupInfo مقدار دهي كردبرنامه والد فرض   از مقادير پيش.  

lpProcessInformation :   اشـاره گــري اســت بــه ســاختارPROCESS_INFORMATION  كــه دريافــت
  :شود ف مياين ساختار به شكل زير تعري. شناسايي پروسه ايجاد شده استكننده ي اطلاعات 

  
PROCESS_INFORMATION STRUCT  
    hProcess          HANDLE ?             ; handle to the child process  
    hThread            HANDLE ?             ; handle to the primary thread of the child process  
    dwProcessId     DWORD ?             ; ID of the child process  
    dwThreadId      DWORD ?            ; ID of the primary thread of the child process  
PROCESS_INFORMATION ENDS 

  
  

    Process handle and process ID : دو مفهوم مجزا هستند .process ID  ، يكID     منحـصر بفـرد از پروسـه در سيـستم 
 وينـدوز بازگـشت داده   APIتوسـط وينـدوز بـراي اسـتفاده سـاير توابـع       مقداري است كـه  ،  process handle .عامل است

  . و منحصربفرد نيستشود مي
بـراي  . شـود  خروجي تابع بلافاصله بازگشت داده مـي  پروسه اي جديد ايجاد شده و    ، CreateProcess   پس از فراخواني    

قالـب  .   اسـتفاده كـرد  GetExitCodeProcessتـوان از تـابع      ل است يا خير مي    بررسي اينكه آيا پروسه ايجاد شده هنوز فعا       
  :اين تابع به صورت زير است

GetExitCodeProcess proto hProcess:DWORD, lpExitCode:DWORD  

ساوي  اگر اين خروجي م ـ. گيرد   قرار مي lpExitCodeوضعيت پروسه در پارامتر       ،اگر اين فراخواني موفقيت آميز باشد          
STILL_ACTIVEباشد ، بدين معنا است كه پروسه ايجاد شده هنوز در حال اجرا است .  

  
  :تعريف اين تابع به شرح زير است. اجبار خاتمه دادتوان پروسه را    ميTerminateProcess   با فراخواني تابع 

TerminateProcess proto hProcess:DWORD, uExitCode:DWORD  
  
 هـاي  dll زيـرا   باشـد ،  اما روش صحيحي براي خاتمه يك پروسه نمي  توان بكار برد ،    واهي را اينجا مي    دلخ يكد خروج    

  . سازد  را باخبر نميپروسهضميمه شده به 
  
  

  :كد برنامه
  

  اجـرا  msgbox.exeشـده و برنامـه   اي جديـد خلـق    پروسـه شـود ،   از منو انتخاب مي  create process  گزينههنگاميكه    
) در صورتيكه در حال اجرا باشـد (مي توان برنامه را   منو، terminate processپروسه، توسط گزينه پس از ايجاد .ودش مي

  .خاتمه بخشيد
  

.386  

.model flat,stdcall  
option casemap:none  
WinMain proto :DWORD,:DWORD,:DWORD,:DWORD  
include \masm32\include\windows.inc  
include \masm32\include\user32.inc  
include \masm32\include\kernel32.inc  
includelib \masm32\lib\user32.lib  
includelib \masm32\lib\kernel32.lib  

.const  
IDM_CREATE_PROCESS equ 1  
IDM_TERMINATE equ 2  
IDM_EXIT equ 3  

.data  
ClassName db "Win32ASMProcessClass",0  
AppName  db "Win32 ASM Process Example",0  
MenuName db "FirstMenu",0  
processInfo PROCESS_INFORMATION <>  
programname db "msgbox.exe",0  

.data?  
hInstance HINSTANCE ?  
CommandLine LPSTR ?  
hMenu HANDLE ?  
ExitCode DWORD ?       ; contains the process exitcode status from GetExitCodeProcess call.  

.code  
start:  
        invoke GetModuleHandle, NULL  
        mov    hInstance,eax  
        invoke GetCommandLine  
        mov CommandLine,eax  
        invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT  
        invoke ExitProcess,eax  

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:DWORD  
    LOCAL wc:WNDCLASSEX  
    LOCAL msg:MSG  
    LOCAL hwnd:HWND  
    mov   wc.cbSize,SIZEOF WNDCLASSEX  
    mov   wc.style, CS_HREDRAW or CS_VREDRAW  
    mov   wc.lpfnWndProc, OFFSET WndProc  
    mov   wc.cbClsExtra,NULL  
    mov   wc.cbWndExtra,NULL  
    push  hInst  
    pop   wc.hInstance  
    mov   wc.hbrBackground,COLOR_WINDOW+1  
    mov   wc.lpszMenuName,OFFSET MenuName  
    mov   wc.lpszClassName,OFFSET ClassName  
    invoke LoadIcon,NULL,IDI_APPLICATION  
    mov   wc.hIcon,eax  
    mov   wc.hIconSm,eax  
    invoke LoadCursor,NULL,IDC_ARROW  
    mov   wc.hCursor,eax  
    invoke RegisterClassEx, addr wc  
    invoke CreateWindowEx,WS_EX_CLIENTEDGE,ADDR ClassName,ADDR AppName,\  
           WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\  
           CW_USEDEFAULT,300,200,NULL,NULL,\  
           hInst,NULL  
    mov   hwnd,eax  
    invoke ShowWindow, hwnd,SW_SHOWNORMAL  
    invoke UpdateWindow, hwnd  
    invoke GetMenu,hwnd  
    mov  hMenu,eax  
    .WHILE TRUE  
                invoke GetMessage, ADDR msg,NULL,0,0  
                .BREAK .IF (!eax)  
                invoke TranslateMessage, ADDR msg  
                invoke DispatchMessage, ADDR msg  
    .ENDW  
    mov     eax,msg.wParam  
    ret  
WinMain endp  

WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM  
    LOCAL startInfo:STARTUPINFO  
    .IF uMsg==WM_DESTROY  
        invoke PostQuitMessage,NULL  
    .ELSEIF uMsg==WM_INITMENUPOPUP  
        invoke GetExitCodeProcess,processInfo.hProcess,ADDR ExitCode  
        .if eax==TRUE  
            .if ExitCode==STILL_ACTIVE  
                invoke EnableMenuItem,hMenu,IDM_CREATE_PROCESS,MF_GRAYED  
                invoke EnableMenuItem,hMenu,IDM_TERMINATE,MF_ENABLED  
            .else  
                invoke EnableMenuItem,hMenu,IDM_CREATE_PROCESS,MF_ENABLED  

                invoke EnableMenuItem,hMenu,IDM_TERMINATE,MF_GRAYED  
            .endif  
        .else  
            invoke EnableMenuItem,hMenu,IDM_CREATE_PROCESS,MF_ENABLED  
            invoke EnableMenuItem,hMenu,IDM_TERMINATE,MF_GRAYED  
        .endif  
    .ELSEIF uMsg==WM_COMMAND  
        mov eax,wParam  
        .if lParam==0  
            .if ax==IDM_CREATE_PROCESS  
                .if processInfo.hProcess!=0  
                    invoke CloseHandle,processInfo.hProcess  
                    mov processInfo.hProcess,0  
                .endif  
                invoke GetStartupInfo,ADDR startInfo  
                invoke CreateProcess,ADDR programname,NULL,NULL,NULL,FALSE,\  
                                        NORMAL_PRIORITY_CLASS,\  
                                        NULL,NULL,ADDR startInfo,ADDR processInfo  
                invoke CloseHandle,processInfo.hThread  
            .elseif ax==IDM_TERMINATE  
                invoke GetExitCodeProcess,processInfo.hProcess,ADDR ExitCode  
                .if ExitCode==STILL_ACTIVE  
                    invoke TerminateProcess,processInfo.hProcess,0  
                .endif  
                invoke CloseHandle,processInfo.hProcess  
                mov processInfo.hProcess,0  
            .else  
                invoke DestroyWindow,hWnd  
            .endif  
        .endif  
    .ELSE  
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam  
        ret  
    .ENDIF  
    xor    eax,eax  
    ret  
WndProc endp  
end start  

  
  :بررسي كد فوق

  
هنگاميكه كـاربر بـر روي   . كند  را براي استفاده آتي ذخيره ميmenu handleبرنامه پنجره اصلي را ايجاد كرده و سپس    

  .  را پردازش خواهيم كردWM_INITMENUPOPUP پيغام   در منو كليك مي نمايد ،Processگزينه 
  

    .ELSEIF uMsg==WM_INITMENUPOPUP  
        invoke GetExitCodeProcess,processInfo.hProcess,ADDR ExitCode  
        .if eax==TRUE  
            .if ExitCode==STILL_ACTIVE  
                invoke EnableMenuItem,hMenu,IDM_CREATE_PROCESS,MF_GRAYED  
                invoke EnableMenuItem,hMenu,IDM_TERMINATE,MF_ENABLED  
            .else  
                invoke EnableMenuItem,hMenu,IDM_CREATE_PROCESS,MF_ENABLED  
                invoke EnableMenuItem,hMenu,IDM_TERMINATE,MF_GRAYED  
            .endif  
        .else  

            invoke EnableMenuItem,hMenu,IDM_CREATE_PROCESS,MF_ENABLED  
            invoke EnableMenuItem,hMenu,IDM_TERMINATE,MF_GRAYED  
        .endif 

  
پارامترهـاي ايـن   آيا پروسه جديد در حال اجرا است ؟   بررسي مي شود كه      GetExitCodeProcessتوسط تابع        در ابتدا   

 باشد بدين معنا است كـه پروسـه   FALSEاگر خروجي تابع ذكر شده    . اند   مقدار دهي شده   CreateProcessبع توسط تابع    تا
  :در ادامه. جديد هنوز ايجاد نشده است

            .if ax==IDM_CREATE_PROCESS  
                .if processInfo.hProcess!=0  
                    invoke CloseHandle,processInfo.hProcess  
                    mov processInfo.hProcess,0  
                .endif  
                invoke GetStartupInfo,ADDR startInfo  
                invoke CreateProcess,ADDR programname,NULL,NULL,NULL,FALSE,\  
                                        NORMAL_PRIORITY_CLASS,\  
                                        NULL,NULL,ADDR startInfo,ADDR processInfo  
                invoke CloseHandle,processInfo.hThread  

  
  

ــه    ــاربر گزينـ ــه كـ ــي  start process   هنگاميكـ ــاب مـ ــو انتخـ ــد   را از منـ ــدار ، نمايـ ــمقـ ــاختار hProcessضو عـ  سـ
PROCESS_INFORMATION   زيـرا ايـن سـاختار در    در ابتداي كار اين مقدار مـساوي صـفر اسـت          . نماييم  را بررسي مي

امـا  پروسه جديد ايجاد شده خاتمـه يافتـه ،   اگر اين مقدار مساوي صفر باشد يعني   .  برنامه تعريف شده است    data.قسمت  
  . نابراين اينكار را در ادامه انجام خواهيم داد بسته نشده ، بprocess handleهنوز 

ــا فراخــواني   ــاختار   ، GetStartupInfo   ب ــضاي س ــد   startupinfoاع ــد ش ــي خواهن ــدار ده ــابع  . مق ــي ت ــه خروج در ادام
CreateProcessدر برنامه هاي ديگر حتما بايد اينكار صورت گيرد.  بدليل پيچيده نشدن برنامه بررسي نشد .  

.  ، بـسته خواهـد شـد   processInfo دريافت شده از ساختار CreateProcess ، thread handle از فراخواني    بلافاصله پس
دسـتگيره تـرد در   خـواهيم از   بستن اين دستگيره بدين معنا نيست كه ترد خاتمه يافته ، بلكه به اين معنا است كه ديگر نمي            

  :سپس. را در برنامه خواهيم داشت) resource leak(ابع در صورت عدم انجام اينكار نشتي من. استفاده كنيمبرنامه

            .elseif ax==IDM_TERMINATE  
                invoke GetExitCodeProcess,processInfo.hProcess,ADDR ExitCode  
                .if ExitCode==STILL_ACTIVE  
                    invoke TerminateProcess,processInfo.hProcess,0  
                .endif  
                invoke CloseHandle,processInfo.hProcess  
                mov processInfo.hProcess,0  

شـود   بررسـي مـي   GetExitCodeProcessتوسط تابع ابتدا شود ،   از منو انتخاب ميterminate process   هنگاميكه گزينه 
به پروسه خاتمـه داده   TerminateProcess  با كمكهنوز فعال است يا خير؟ در صورت مثبت بودن پاسخ ،     كه آيا پروسه    

  .شود بسته مي همچنين دستگيره اين پروسه جديد نيز .شود مي

  Freshبا مقدماتي آشنايي 
  
هـا   عمـده اي از طراحـي رابـط كـاربر برنامـه     ايد كه قـسمت   هستيد و دريافتهتر كردن كد نويسي اسمبلر  اگر بدنبال ساده     

كه به ايـن منظـور تهيـه    ) IDE(هاي توسعه مجتمع    توان از محيط   د، مي نشو ها تكرار مي   همانند قالب يكساني در اكثر برنامه     
  :اين برنامه از آدرس زير قابل دريافت است.  ياد كردFreshتوان از  براي نمونه مي.  استفاده كرد،اند شده

  
http://fresh.flatassembler.net  

  

  
  

 Fresh نمايي از محيط -  ۵ شكل
Flat assembler visual programming IDE  

  
  
  

http://fresh.flatassembler.net

  )Multithreading Programming(برنامه نويسي چند ريسماني  -۱۳مثال 
  

. گوينـد   نيـز مـي  primary thread كـه بـه آن      در مثال قبل آموختيم كه هر پروسه حداقل از يك ترد تشكيل شـده اسـت  
از ديدگاه پياده سـازي ، يـك تـرد تـابعي اسـت كـه بـه صـورت         . تردهايي را ايجاد نمايدتواند    برنامه نيز مي    يك همچنين

توان چندين وهله از يك تـابع و يـا چنـدين تـابع مختلـف را همزمـان         بسته به نياز مي    .شود  همزمان با برنامه اصلي اجرا مي     
  . اجرا كرد

هـر تـرد   . دن ـبه تمام منابع يك پروسه مانند متغيرهـا و غيـره دسترسـي دار    شوند بنابراين     اجرا مي    تردها تحت يك پروسه     
همچنــين مجموعــه .د بــودنــصــرفا مخــتص بــه آن خواهمتغيرهــاي محلــي آن ، پــشته مخــصوص خــود را دارد ، بنــابراين 

ي ديگـر ، تـرد قبلـي مـي توانـد      رجيسترهاي مورد استفاده براي يك ترد نيز مختص به آن بـوده و در هنگـام اجـراي تـرد          
اين موارد به صـورت درونـي توسـط وينـدوز     . حالت قبلي خود را بخاطر آورده و در صورت نياز آنرا مجددا از سر گيرد       

    :توان تقسيم كرد تردها را به دو گروه مي. دنشو مديريت مي
  

نايي دريافـت پيغـام هـاي    و توااين نوع تردها پنجره مخصوص خود را ايجاد كرده   : تردهاي رابط كاربر   .۱
 . دنويندوز را دار

اين نوع تردها پنجره مخصوص خود را ايجاد نكرده و توانايي دريافت پيغـام هـاي وينـدوز             : كاريترد   .۲
  . دنشو از اين نوع تردها براي انجام كارهايي در پس زمينه برنامه استفاده مي.را ندارند

  
اجـازه دهيـد تـرد اصـلي برنامـه      : اين روند را دنبـال كنيـد  هاي چندريسماني  هبرنامدر هنگام پياده سازي     شود       توصيه مي 

بنـابراين  . كارهاي مربوط به رابط كاربر را مديريت كند و از اين مبحث صرفا جهت ايجاد تردهاي كاري اسـتفاده نماييـد          
مـدير كارهـا را بـه افـراد تحـت      .  كردترد اصلي برنامه شبيه به مدير و ساير تردها همانند دستياران و افراد او رفتار خواهند              

هـاي   افـراد كارهـاي محولـه را انجـام داده و گـزارش     . نمايد  خود محول كرده و تماس خود را با بيرون حفظ مي         يتمدير
بهتـر اسـت كارهـاي طـولاني را در تردهـاي كـاري انجـام داد و مـسؤوليت پاسـخ دهـي بـه            .دهند لازم را به مدير ارائه مي   

براي مثال اگر كارهاي طولاني را به ترد اصـلي برنامـه محـول    . ز بيرون را به ترد اصلي برنامه محول كردهاي رسيده ا   پيغام
  . تواند به پيغام هاي رسيده از بيرون پاسخ لازم و مناسب را بدهد نماييد ، تا پايان كار ترد ، برنامه نمي

   :تعريف اين تابع به شكل زير استروش . دترد ايجاد خواهد ش يك CreateThread   با فراخواني تابع 
  

CreateThread proto lpThreadAttributes:DWORD,\  
                                dwStackSize:DWORD,\  
                                lpStartAddress:DWORD,\  
                                lpParameter:DWORD,\  
                                dwCreationFlags:DWORD,\  
                                lpThreadId:DWORD  

lpThreadAttributes :   با مساويNULL       قرار دادن ايـن پـارامتر مـي تـوان از خـواص امنيتـي  
  .پيش فرض ويندوز در مورد ترد استفاده كرد

dwStackSize :      بـا مـساوي نـال قـرار دادن ايـن پـارامتر ،            .دكن  اندازه پشته ترد را مشخص مي 
  .با اندازه پشته ترد اصلي يكي خواهد شداندازه پشته 

lpStartAddress :    تابعي است كه كار تـرد را انجـام خواهـد     تابع ترد ،    . آدرس تابع ترد است
  . داد

lpParameter : پارامتري كه مايل هستيد به تابع ترد ارسال كنيد.  
dwCreationFlags :             بايـد اجـرا   اگر مساوي صفر قرار گيرد يعني ترد بلافاصله پس از ايجـاد

  .   استCREATE_SUSPENDEDبرخلاف آن . شود
lpThreadId : تابع CreateThread  آدرس idدهد  ترد ايجاد شده را در اين پارامتر قرار مي .  

  
تـرد ايجـاد شـده بازگـشت داده خواهـد         ) handle(دسـتگيره    موفقيت آميز باشـد ،       CreateThread   اگر فراخواني تابع       

، شـروع    تنظيم نشده باشـد CREATE_SUSPENDEDكار خويش را بلافاصله در صورتيكه پرچم   ،  همچنين تابع ترد    .شد
   .نمايد  شروع ميResumeThreadاگر اين پرچم تنظيم شده باشد ، تابع ترد كار خود را تنها با فراخواني تابع .خواهد كرد

 را جهـت تـابع تـرد بـه صـورت ضـمني         ExitThread خاتمـه يابـد، وينـدوز تـابع          RET تابع ترد بوسيله دسـتورالعمل          اگر
امـا بايـد دقـت داشـت كـه كـد       از درون تابع ترد نيز موجـود اسـت ،   ExitThread  امكان فراخواني تابع.كند فراخواني مي

    .رد بدست آوGetExitCodeThread توسط تابع توان ميخروجي ترد را 
. اينكـار را انجـام داد  توان    ميTerminateThread   اگر قصد خاتمه يك ترد را توسط تردي ديگر داريد ، با كمك تابع               

سـبب خاتمـه آنـي تـرد بـدون فرصـت دادن بـه آن جهـت بازگردانـدن منـابع             ،  البته بايد دقت داشت فراخواني ايـن تـابع          
  .گردد مي، به سيستمداده شده  تخصيص

  
  :ل سه روش براي تبادل اطلاعات بين تردها وجود دارد   حداق

 استفاده از متغيرهاي سراسري •

 هاي ويندوز استفاده از پيغام •

  هارخداد •
  

بنـابراين تردهـا از متغيرهـاي    .گذارنـد   بـين خـويش بـه اشـتراك مـي      )شامل متغيرهاي سراسري (را   ها پروسهمنابع     تردها  
بـه همـراه نكـات مربـوط بـه      البته اين مـورد را بايـد بـا دقـت كامـل            .كنند  ه مي سراسري جهت تبادل اطلاعات باهم استفاد     

  . بكار گرفتهمزماني تردها 

چـه اتفـاقي خواهـد افتـاد اگـر      . كننـد  به صورت مشترك اسـتفاده مـي  ،  عضو ۱۰از يك ساختار با   ،  فرض كنيد دو ترد        
شـد و ناگهـان وينـدوز كنتـرل را از ايـن تـرد بـه تـردي         هنگاميكه يك ترد در نيمه به روز رساني مقادير اعضاي ساختار با     

هـاي چنـد ريـسماني مـشكل تـر بـوده و        مديريت برنامه. هايي ناقص روبرو خواهد شد  ترد ديگر با داده   ديگر منتقل نمايد؟  
  .نياز به دقت بيشتري دارند

  
تردهـاي رابـط   تردهـا از نـوع   گر  ا. توان كمك گرفت      همچنين از پيغامهاي ويندوز جهت تبادل اطلاعات بين تردها مي         

  :براي مثال. كاربر باشند مي توان براي آنها با توجه به پيغامهاي موجود ويندوز ، پيغامهاي سفارشي طراحي كرد
  

WM_MYCUSTOMMSG equ WM_USER+100h  
  

نجـره مخـصوص بـه    تردكـاري پ زيـرا  . توان استفاده كـرد     اما اگر يكي از تردها ، تردكاري باشد ديگر از اين روش نمي         
  :توان استفاده كرد در اين موارد از شماي زير مي. بهره است از صف پيغامها بي وخود را نداشته 

  
  تردكاري    متغيرهاي سراسري    ترد رابط كاربر

  ترد رابط كاربر    پيغامهاي سفارشي     تردكاري
  

  .    و از اين روش در برنامه اين مثال استفاده خواهيم كرد
  

بـه ايـن نـوع اشـياء هماننـد نـوعي پـرچم نيـز            .  اسـت  هارخداداشياء  استفاده از   ،  رين روش تبادل اطلاعات بين تردها        آخ
در ايـن حالـت   . بـرد  ي كمـون بـه سـر مـي     تـرد در دوره  باشد،  unsignalled در حالت    رخداد اگر شيء    .توان نگريست   مي

وينـدوز تـرد را     قـرار گيـرد،   signalled در حالت رخدادشيء اگر .  به ترد اختصاص داده نخواهد شد    CPUقطعات زماني   
  . به حالت نرمال درآورده و ترد كار عادي خود را آغاز خواهد كرد

  
  

  :كد برنامه
  

از  را Savage Calculationگزينـه  .  را اجـرا كنيـد  thread1.exe    فايلهاي همراه اين مجموعه را دريافت و سـپس برنامـه   
لازم به ذكر اسـت كـه   . گردد  بار مي  600,000,000  به تعدادadd eax,eaxاينكار سبب اجراي . دبرنامه انتخاب كنيمنوي 

تـوان انجـام    براي حل اين مـشكل مـي  . توان انجام داد در طول انجام اين عمليات هيچگونه كاري با رابط كاربر برنامه نمي         

هـاي رسـيده بـه رابـط      پاسخ دهي بـه درخواسـت  برنامه اجازه انتقال داد و به ترد اوليه  برنامه  را به ترد كاري     اين محاسبات   
  . )افتد  اما بطور كامل از كار نميشدهر چند اينكار كندتر از معمول قابل انجام خواهد  (كاربر را داد

  

.386  

.model flat,stdcall  
option casemap:none  
WinMain proto :DWORD,:DWORD,:DWORD,:DWORD  
include \masm32\include\windows.inc  
include \masm32\include\user32.inc  
include \masm32\include\kernel32.inc  
includelib \masm32\lib\user32.lib  
includelib \masm32\lib\kernel32.lib  

.const  
IDM_CREATE_THREAD equ 1  
IDM_EXIT equ 2  
WM_FINISH equ WM_USER+100h  

.data  
ClassName db "Win32ASMThreadClass",0  
AppName  db "Win32 ASM MultiThreading Example",0  
MenuName db "FirstMenu",0  
SuccessString db "The calculation is completed!",0  

.data?  
hInstance HINSTANCE ?  
CommandLine LPSTR ?  
hwnd HANDLE ?  
ThreadID DWORD ?  

.code  
start:  
    invoke GetModuleHandle, NULL  
    mov    hInstance,eax  
    invoke GetCommandLine  
    mov CommandLine,eax  
    invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT  
    invoke ExitProcess,eax  

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:DWORD  
    LOCAL wc:WNDCLASSEX  
    LOCAL msg:MSG  
    mov   wc.cbSize,SIZEOF WNDCLASSEX  
    mov   wc.style, CS_HREDRAW or CS_VREDRAW  
    mov   wc.lpfnWndProc, OFFSET WndProc  
    mov   wc.cbClsExtra,NULL  
    mov   wc.cbWndExtra,NULL  
    push  hInst  
    pop   wc.hInstance  
    mov   wc.hbrBackground,COLOR_WINDOW+1  
    mov   wc.lpszMenuName,OFFSET MenuName  
    mov   wc.lpszClassName,OFFSET ClassName  
    invoke LoadIcon,NULL,IDI_APPLICATION  
    mov   wc.hIcon,eax  
    mov   wc.hIconSm,eax  

    invoke LoadCursor,NULL,IDC_ARROW  
    mov   wc.hCursor,eax  
    invoke RegisterClassEx, addr wc  
    invoke CreateWindowEx,WS_EX_CLIENTEDGE,ADDR ClassName,ADDR AppName,\  
           WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\  
           CW_USEDEFAULT,300,200,NULL,NULL,\  
           hInst,NULL  
    mov   hwnd,eax  
    invoke ShowWindow, hwnd,SW_SHOWNORMAL  
    invoke UpdateWindow, hwnd  
    .WHILE TRUE  
            invoke GetMessage, ADDR msg,NULL,0,0  
            .BREAK .IF (!eax)  
            invoke TranslateMessage, ADDR msg  
            invoke DispatchMessage, ADDR msg  
    .ENDW  
    mov     eax,msg.wParam  
    ret  
WinMain endp  

WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM  
    .IF uMsg==WM_DESTROY  
        invoke PostQuitMessage,NULL  
    .ELSEIF uMsg==WM_COMMAND  
        mov eax,wParam  
        .if lParam==0  
            .if ax==IDM_CREATE_THREAD  
                mov  eax,OFFSET ThreadProc  
                invoke CreateThread,NULL,NULL,eax,\  
                                        0,\  
                                        ADDR ThreadID  
                invoke CloseHandle,eax  
            .else  
                invoke DestroyWindow,hWnd  
            .endif  
        .endif  
    .ELSEIF uMsg==WM_FINISH  
        invoke MessageBox,NULL,ADDR SuccessString,ADDR AppName,MB_OK  
    .ELSE  
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam  
        ret  
    .ENDIF  
    xor    eax,eax  
    ret  
WndProc endp  

ThreadProc PROC USES ecx Param:DWORD  
        mov  ecx,600000000  
Loop1:  
        add  eax,eax  
        dec  ecx  
        jz   Get_out  
        jmp  Loop1  
Get_out:  
        invoke PostMessage,hwnd,WM_FINISH,NULL,NULL  
        ret  
ThreadProc ENDP  

end start  

  :بررسي كد فوق
  

  :كند ، برنامه تردي را به شكل زير ايجاد ميتخاب كند  را انCreate Thread    اگر كاربر از منوي برنامه گزينه 
  

            .if ax==IDM_CREATE_THREAD  
                mov  eax,OFFSET ThreadProc  
                invoke CreateThread,NULL,NULL,eax,\  
                                        NULL,0,\  
                                        ADDR ThreadID  
                invoke CloseHandle,eax  

  
پـس از اجـراي موفقيـت    .كنـد  برنامه ايجاد مـي  با ترد اصلي   ThreadProcتابع  جهت اجراي همزمان    تردي را   ،     كد فوق   

گيره ترد استفاده نخواهد شد ، آنـرا جهـت   همچنين از آنجائيكه در برنامه از دست. آميز ترد ، تابع بلافاصله اجرا خواهد شد    
  :در ادامه. )البته بستن آن به معناي خاتمه ترد نيست (جلوگيري از نشتي حافظه ، خواهيم بست

  
ThreadProc PROC USES ecx Param:DWORD  
        mov  ecx,600000000  
Loop1:  
        add  eax,eax  
        dec  ecx  
        jz   Get_out  
        jmp  Loop1  
Get_out:  
        invoke PostMessage,hwnd,WM_FINISH,NULL,NULL  
        ret  
ThreadProc ENDP  

  
 ايـن پيغـام بـه صـورت زيـر تعريـف       .شـود   به پنجره اصلي برنامه ارسـال مـي    WM_FINISH   پس از پايان كار ترد ، پيغام        

  :است شده
WM_FINISH equ WM_USER+100h 

  
  .  از منو انتخاب كردCreate Threadتوان چندين ترد را همزمان با انتخاب گزينه     مي

از پنجـره بـه تـرد    خواهيـد   اگـر مـي  ). ها به پنجره برنامه پيغامارسال (ها بين تردها يك طرفه است           در اين مثال تبادل داده    
  :دپيغام ها ارسال شود مي توان به صورت زير عمل كربرنامه كاري 

 ، به منوي اصلي برنامه  Kill Thread  گزينهاضافه كردن •

 بـه معنـاي متوقـف سـاختن تـرد و         TRUE. تعريف يـك متغيـر سراسـري بعنـوان يـك پـرچم               •
FALSEبه معناي ادامه ترد است . 

 ه و در صـورتيك  به صورتيكه اين پرچم را درون حلقـه بررسـي كنـد       ThreadProcاصلاح تابع    •
  .با بررسي پرچم مربوطه ، ترد را خاتمه دهد را انتخاب كرده بود ، Kill Threadكاربر گزينه 

  RadASMبا مقدماتي آشنايي 
  
   RadASMجهت سهولت كار با اسمبلرهاي مختلفي همانند masm/tasm/fasm/nasm/goasm/hla    طراحـي شـده اسـت  .

  : باشد اين برنامه از آدرس زير قابل دريافت مي
  

http://radasm.visualassembler.com  
  

  
  

 RadASM نمايي از محيط -۶ شكل

  
  : ، جهت عموم علاقمندان در آدرس زير قرار دارندIDEجهت كار با اين آموزشي چندين مقاله 

  
http://members.a1.net/ranmasaotome/main.html  

  
  
  

http://radasm.visualassembler.com
http://members.a1.net/ranmasaotome/main.html

  )Event Object(شيء رخداد  -۱۴مثال 
  
  

دو روش ديگر شامل اسـتفاده از متغيرهـاي عمـومي و    . پيغامها بررسي شد    در مثال قبل تبادل اطلاعات بين تردها توسط        
 دو حالـت خـاموش    وشيء رخداد شبيه به يك سوئيچ است. شيء رخداد باقي ماندند كه در اين مثال بررسي خواهند شد      

 شـيء رخـداد   .ودش ـ  نيز گفتـه مـي  signalledزمانيكه اين سوئيچ روشن باشد به آن حالت       .تواند داشته باشد    و روشن را مي   
 اگـر شـيء رخـداد در حالـت     .گيـرد   حالـت آن مـورد بررسـي قـرار مـي     ۲گرفتـه و  در كد ترد مربوطه قـرار    ،ايجاد شده   

nonsignalled    ، وقتـي تردهـا در حالـت انتظـار     .باشـد  تردي كه منتظر وضعيت روشن آن است در حالت خواب مـي            باشد
  .هند داد را به خود اختصاص خواCPUزمان كمي از ، باشند 

  .:دشو  مطابق تعريف زير ، ايجاد ميCreateEvent   يك شيء رخداد با استفاده از تابع 
  

CreateEvent proto lpEventAttributes:DWORD,\  
                              bManualReset:DWORD,\  
                              bInitialState:DWORD,\  
                              lpName:DWORD  

  
lpEventAttribute:     اگر مساويNULL   هاي امنيتي پيش فـرض مـورد اسـتفاده قـرار        قرار گيرد ويژگي

  .دنگير مي
bManualReset :         اگر مايل هستيد پس از فراخواني تابعWaitForSingleObject   ،     شيء ترد بـه حالـت

nonsignalled   ،ر را مساوي بايد اين پارامت برگردانده شودFALSE  در غيراينـصورت بايـد از   . قرار دهيد
  . ، جهت به حالت اول بازگرداندن وضعيت شيء رخداد استفاده كنيدResetEventتابع 

bInitialState:     با مـساويTRUE     در حالـت ايـن پـارامتر ، شـيء رخـداد توليـدي        قـرار دادن signalled  
  .ايجاد خواهد شد

lpName :  ه اشاره گري به رشتASCIIZ) ايـن نـام در هنگـام    .باشـد  رخداد مينام شيء كه  )مختوم به نال
  .شود  بكار برده ميOpenEventفراخواني 

  
، در غيراينـصورت   گردانـد  دستگيره اي را به شيء رخداد ايجاد شده بـاز مـي   ،      اگر فراخواني موفقيت آميز باشد ، تابع        

  ResetEvent و) signalledبـه حالـت   ( SetEvent فراخـواني دو تـابع  حالـت شـيء رخـداد را بـا     . خروجي نال خواهد بود  
  .  تغيير دادتوان مي
 را در يـك تـرد جهـت بررسـي وضـعيت آن قـرار       WaitForSingleObjectبايد تـابع   هنگاميكه شيء رخداد ايجاد شد ،          

  :گردد اين تابع به شكل زير تعريف مي.داد
WaitForSingleObject proto hObject:DWORD, dwTimeout:DWORD  

hObject :  آيد شيء رخداد يك شيء همزماني بشمار مي. اشياء همزمانيدستگيره اي به يكي از.  
dwTimeout :    تا رسيدن به وضـعيت      كه تابع   ) برحسب ميلي ثانيه  (زماني استsignalled    صـبر خواهـد 

اگـر  . بع اجـرا خواهـد شـد   تـا بـود ،   nonsignalledاگر اين زمان سپري شد و هنوز وضعيت شـيء    . كرد
 قـرار  INFINITEخواهيد مدت زمان اين صبركردن بي نهايت شود بايد پارامتر ذكر شـده را مـساوي           مي

  .دهيد
  
  

  :كد برنامه
  

اگـر كـاربر گزينـه    .    اين مثال پنجره اي را نمايش خواهد داد كه منتظر انتخاب گزينه اي در منو توسط كـاربر مـي باشـد     
run threadپس از پايان ترد پيغامي به كـاربر جهـت   . ، ترد محاسبات بسيار زمانبري را شروع خواهد كرداب نمود  را انتخ

  . را خواهد داشتstop threadهنگاميكه ترد در حال اجرا است كاربر امكان گزينه . شود اطلاع او نمايش داده مي
  

.386  

.model flat,stdcall  
option casemap:none  
WinMain proto :DWORD,:DWORD,:DWORD,:DWORD  
include \masm32\include\windows.inc  
include \masm32\include\user32.inc  
include \masm32\include\kernel32.inc  
includelib \masm32\lib\user32.lib  
includelib \masm32\lib\kernel32.lib  

.const  
IDM_START_THREAD equ 1  
IDM_STOP_THREAD equ 2  
IDM_EXIT equ 3  
WM_FINISH equ WM_USER+100h  

.data  
ClassName db "Win32ASMEventClass",0  
AppName  db "Win32 ASM Event Example",0  
MenuName db "FirstMenu",0  
SuccessString db "The calculation is completed!",0  
StopString db "The thread is stopped",0  
EventStop BOOL FALSE  

.data?  
hInstance HINSTANCE ?  
CommandLine LPSTR ?  
hwnd HANDLE ?  
hMenu HANDLE ?  
ThreadID DWORD ?  
ExitCode DWORD ?  
hEventStart HANDLE ?  

.code  
start:  
    invoke GetModuleHandle, NULL  
    mov    hInstance,eax  
    invoke GetCommandLine  
    mov CommandLine,eax  
    invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT  
    invoke ExitProcess,eax  

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:DWORD  
    LOCAL wc:WNDCLASSEX  
    LOCAL msg:MSG  
    mov   wc.cbSize,SIZEOF WNDCLASSEX  
    mov   wc.style, CS_HREDRAW or CS_VREDRAW  
    mov   wc.lpfnWndProc, OFFSET WndProc  
    mov   wc.cbClsExtra,NULL  
    mov   wc.cbWndExtra,NULL  
    push  hInst  
    pop   wc.hInstance  
    mov   wc.hbrBackground,COLOR_WINDOW+1  
    mov   wc.lpszMenuName,OFFSET MenuName  
    mov   wc.lpszClassName,OFFSET ClassName  
    invoke LoadIcon,NULL,IDI_APPLICATION  
    mov   wc.hIcon,eax  
    mov   wc.hIconSm,eax  
    invoke LoadCursor,NULL,IDC_ARROW  
    mov   wc.hCursor,eax  
    invoke RegisterClassEx, addr wc  
    invoke CreateWindowEx,WS_EX_CLIENTEDGE,ADDR ClassName,\  
            ADDR  AppName,\  
           WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\  
           CW_USEDEFAULT,300,200,NULL,NULL,\  
           hInst,NULL  
    mov   hwnd,eax  
    invoke ShowWindow, hwnd,SW_SHOWNORMAL  
    invoke UpdateWindow, hwnd  
    invoke GetMenu,hwnd  
    mov  hMenu,eax  
    .WHILE TRUE  
            invoke GetMessage, ADDR msg,NULL,0,0  
            .BREAK .IF (!eax)  
            invoke TranslateMessage, ADDR msg  
            invoke DispatchMessage, ADDR msg  
    .ENDW  
    mov     eax,msg.wParam  
    ret  
WinMain endp  

WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM  
    .IF uMsg==WM_CREATE  
        invoke CreateEvent,NULL,FALSE,FALSE,NULL  
        mov  hEventStart,eax  
        mov  eax,OFFSET ThreadProc  
        invoke CreateThread,NULL,NULL,eax,\  
                             NULL,0,\  
                             ADDR ThreadID  
        invoke CloseHandle,eax  
    .ELSEIF uMsg==WM_DESTROY  
        invoke PostQuitMessage,NULL  
    .ELSEIF uMsg==WM_COMMAND  

        mov eax,wParam  
        .if lParam==0  
            .if ax==IDM_START_THREAD  
                invoke SetEvent,hEventStart  
                invoke EnableMenuItem,hMenu,IDM_START_THREAD,MF_GRAYED  
                invoke EnableMenuItem,hMenu,IDM_STOP_THREAD,MF_ENABLED  
            .elseif ax==IDM_STOP_THREAD  
                mov  EventStop,TRUE  
                invoke EnableMenuItem,hMenu,IDM_START_THREAD,MF_ENABLED  
                invoke EnableMenuItem,hMenu,IDM_STOP_THREAD,MF_GRAYED  
            .else  
                invoke DestroyWindow,hWnd  
            .endif  
        .endif  
    .ELSEIF uMsg==WM_FINISH  
        invoke MessageBox,NULL,ADDR SuccessString,ADDR AppName,MB_OK  
    .ELSE  
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam  
        ret  
.ENDIF  
    xor    eax,eax  
    ret  
WndProc endp  

ThreadProc PROC USES ecx Param:DWORD  
        invoke WaitForSingleObject,hEventStart,INFINITE  
        mov  ecx,600000000  
        .WHILE ecx!=0  
                .if EventStop!=TRUE  
                        add  eax,eax  
                        dec  ecx  
                .else  
                        invoke MessageBox,hwnd,ADDR StopString,ADDR AppName,MB_OK  
                        mov  EventStop,FALSE  
                        jmp ThreadProc  
                .endif  
        .ENDW  
        invoke PostMessage,hwnd,WM_FINISH,NULL,NULL  
        invoke EnableMenuItem,hMenu,IDM_START_THREAD,MF_ENABLED  
        invoke EnableMenuItem,hMenu,IDM_STOP_THREAD,MF_GRAYED  
        jmp   ThreadProc  
        ret  
ThreadProc ENDP  
end start 

  
  :بررسي كد فوق

  
  .  بررسي قرار گرفت   در اين مثال روش ديگري در مورد پياده سازي ترد ها مورد

.IF uMsg==WM_CREATE  
        invoke CreateEvent,NULL,FALSE,FALSE,NULL  
        mov  hEventStart,eax  
        mov  eax,OFFSET ThreadProc  
        invoke CreateThread,NULL,NULL,eax,\  
                             NULL,0,\  

http://mohandesyar.com


  قسمت دوم- آشنايي با برنامه نويسي اسمبلي ويندوز

 ١١٠

                             ADDR ThreadID  
        invoke CloseHandle,eax  

  . شيء رخداد و ترد ايجاد شدند ، WM_CREATE   در خلال پردازش پيغام 
پس از ايجاد شـيء رخـداد ، تـرد    .  ايجاد گشت، با به حالت اول درآمدن خودكارnonsignalledدر حالت    شيء رخداد   

    مانـد تـا مطـابق كـد زيـر بـه حالـت        منتظـر شـيء رخـداد مـي    بلافاصـله اجـرا نگـشته و        ،  ترد پـس از ايجـاد       . گرديدايجاد  
signalledدرآيد :  

ThreadProc PROC USES ecx Param:DWORD  
        invoke WaitForSingleObject,hEventStart,INFINITE  
        mov  ecx,600000000  

  
 به مـدت  رخداد شيء signalledدر اينجا براي حالت .   است  WaitForSingleObject  اولين خط رويه ترد، فراخواني تابع       

  !يعني پس از ايجاد ترد ، آنرا اجبار به خواب خواهيم فرستاد. صبر خواهد شدابع ، ، پيش از پايان تنامحدودي 
 در signalledبـه حالـت   بـه صـورت زيـر     شـيء رخـداد را  را انتخـاب كنـد،    run thread   هنگاميكه كاربر از منو گزينـه  

   :خواهيم آورد

            .if ax==IDM_START_THREAD  
                invoke  SetEvent,hEventStart  

  .    در اين حالت ترد شروع به كار خواهد كرد
 تنظـيم  TRUE را بـه  EventStopمقـدار متغيـر عمـومي     انتخـاب كنـد ،   stop thread   اگر كاربر از منوي برنامـه گزينـه را   

  . خواهيم كرد

                .if EventStop==FALSE  
                        add  eax,eax  
                        dec  ecx  
                .else  
                        invoke MessageBox,hwnd,ADDR StopString,ADDR AppName,MB_OK  
                        mov  EventStop,FALSE  
                        jmp ThreadProc  
                .endif  

ازي بـه  لازم به ذكر است كه ني.   پرش خواهد كرد   WaitForSingleObject  اينكار ترد را متوقف ساخته و برنامه مجدد به           
 تنظـيم  FALSE را به CreateEvent  تابع bManualResetزيرا پارامتر  به صورت صريح نيست ،   ResetEventفراخواني تابع   

  . يمكرده ا

  : خوانندهاز درخواستي
ــايي     ــتن فــوق خط ــر در م ــد  را اگ ــي  (مــشاهده فرمودي ــا غيرفن ــرا لطفــا ، ) فنــي ي ــسنده بــه آدرس   آن بــا نوي

vahid_nasiri@yahoo.com مطرح بفرماييد.  
  
  

  
  :مآخذ

1- Iczelion's Win32 Assembly tutorials. Web: http://win32assembly.online.fr/tutorials.html 
2- Win32Asm Tutorial by Thomas Bleeker. Web: http://www.madwizard.org  
3- Masm32's help files.  
4- An Introduction to Assembly Language I/II/III by Darwen.   
    Web: http://www.codeguru.com/Cpp/Cpp/cpp_mfc/tutorials/article.php/c9411 
5- Win32 Assembly parts 1-6 by Chris Hobbs. 
    Web: http://www.gamedev.net/reference/list.asp?categoryid=20#101  
6- Win32 Assembler Coding Tutorial. Web: http://www.deinmeister.de/w32asm1e.htm 

