
  .به هر نحوي غيرمجاز است نويسنده  طرف ز ابدون مجوز كتبينشر غير الكترونيكي اين مطالب چاپ و يا 
  . آن به صورت مستقيم و يا غير مستقيم در نشريات الكترونيكي بلا مانع استانتشار اين مطالب بر روي اينترنت و يا استفاده از

  قسمت اول- آشنايي با برنامه نويسي اسمبلي ويندوز

 ٢

  
 : مقدمه 

   
  
دليـل خـود را هـم سـختي درك و كـاربرد آن ذكـر           . گريـزان هـستند   ) و يا زبان اسـمبلي    (اكثر برنامه نويسان از اسمبلر         

هـاي برنامـه نويـسان واقـع      م و احتـرام گـروه  در مقابل كساني هم كه به اسمبلي تسلط دارند، به شدت مورد تكري. كنند  مي
  .هدف از انتشار اين مطالب طرد اين طرز فكر منفي در مورد اسمبلر است. شوند مي

   
تـوان گفـت    در ابتدا شايد اين سوال مطرح شود كه اسمبلر چيست؟ اگر بخواهيم آنـرا بـسيار سـاده تعريـف كنـيم ، مـي           

هـا و   البتـه بجـز مقـادير بـايتي دسـتورالعمل     (تـوان تـصور كـرد       تري از آنرا نمي    اسمبلر زبان پردازشگر است و سطح پايين      
هر دستور آن بوسيله برنامه اسمبلر مـستقيما بـه اعـدادي كـه توسـط پردازشـگر قابـل اجـرا اسـت ترجمـه               ). ها  اينستراكشن

  .شود مي
   

حتي بـا بهينـه   . محض، خام و تخفيف نيافتهسرعتي . مزيت استفاده از اسمبلر بر ساير زبانهاي برنامه نويسي، سرعت است        
دهند، كد اسمبلري كه با دسـت بهينـه سـازي شـده      هايي كه كامپايلرهاي مدرن ساير زبانهاي برنامه نويسي انجام مي   سازي

  .شود است بسيار بسيار سريعتر اجرا مي
   

  
توان كل يك برنامه را با اسمبلر نوشـت امـا    ميهر چند  . شود  اي استفاده نمي    امروزه از اسمبلر براي نوشتن هر نوع برنامه          

و بايـد در نظـر داشـت    ! نظر آيـد   شايد انجام اينكار نوعي مازوخيسم به  و غيره،++Cبا وجود زبانهاي سطوح بالاتري مانند   
  . و يا حتي زبانهاي دات نت قابل قبول هستند++Cكه براي بسياري از كاربردها سرعت 

   
بـراي  . بلر حائز اهميت خواهد شد كه سرعت انجام عمليات جـزو مـوارد بحرانـي محـسوب گـردد     زماني استفاده از اسم    

  .ها مپ مثال در برنامه هاي گرافيكي و اعمال مختلف با بيت

  قسمت اول- آشنايي با برنامه نويسي اسمبلي ويندوز

 ٣

   :++VCنوشتن برنامه اسمبلي در 
   

  :براي مثال.  اسمبلي برنامه نوشتتوان به زبان شوند مي مشخص مي  asm__ با استفاده از قطعات كدي كه با ++Cدر    
   
   

 DWORD Function(DWORD dwValue) 
{ 
   __asm 
   { 
      mov eax, dwValue 
      add eax, 100 
      mov dwValue, eax 
   } 
 
   return dwValue; 
}  

   

 نگــران معنــاي .كنيــد مــي ملاحظــه asm__در اينجــا شــما تعــدادي اينستراكــشن اســمبلر را قــرار گرفتــه در يــك قطعــه    
هـا    بـراي افـزودن داده  addو يـا  اسـت  هـا    انتقـال داده mov نفقط بدانيد كه كار اينستراكش. هاي مختلف آن نباشيد     قسمت

  .گرداند  را به آرگومان ورودي افزوده و حاصل را بر مي۱۰۰كد فوق تنها عدد . گيرد مورد استفاده قرار مي
 if..elseاما مثال زير را در مورد بكـار گيـري   .  مشكل نيستasm__اي   از كد داخل قطعه   تا اينجا نوشتن چند خط كوتاه   

  :در نظر بگيريد
   

DWORD Function2(DWORD dwValue1, DWORD dwValue2) 
{ 
   DWORD dwValue3 = 0; 
 
#if 0 
   // this is the Assembler 
   if (dwValue1 == dwValue2) 
      dwValue3 = 1; 
   else 
      dwValue3 = 2; 
#endif 
 
   __asm 
   { 
      mov eax, dwValue1 
      ; this is the test of the values, i.e. if dwValue1 == dwValue2 
      cmp eax, dwValue2 
      ; jump to 'Else' if they are not equal 
      jne Else 
      mov eax, 1 
      jmp EndIf 
 
   Else: 
      mov eax, 2 
 
   EndIf: 
      mov dwValue3, eax 
   } 
 
   return dwValue3; 
}   

  قسمت اول- آشنايي با برنامه نويسي اسمبلي ويندوز

 ٤

 if..elseامـا واقعـا بايـد بـراي پيـاده سـازي هـر         . بازهم در مورد دستورالعمل هاي اسمبلر بكار گرفته شده نگران نباشيد              
   ساده اين مقدار كد نوشت؟

 و دسـتورات خـط فرمـان بـراي برنامـه      Notepadهماننـد اسـتفاده از   اسـمبلر   براي كد نويـسي     asm__   استفاده از قطعات    
 Microsoft Macro Assembler براي برنامه نويـسي اسـمبلر نيـز    .باشد  مي++VC قدرتمند IDEاده از  بجاي استف++Cنويسي 

 قابـل اسـتفاده در   obj. همچنين قابليت توليـد فايلهـاي   .باشد   ارائه شده و مجاني مي     ۱۹۸۰اين اسمبلر از سال     . موجود است 
VC++امروزه نام . را نيز دارا است MASM32زير قابل دريافت است را يافته و در آدرس :  

http://www.masm32.com/  

  
  .كنيد و نصب ريافت   قويا توصيه مي شود قبل از ادامه مطالعه مطالب ، آنرا د

  
  

  
  

 MASM32 نمايي از -۱شكل 
  
  

  اين اسمبلر داراي چه مزايايي است؟! خوب  
 .دباش  ميmsasm32\help\قرار گرفته در مسير ، داراي راهنمايي بسيار جامع  §

  . هاي فراواني جهت سهولت برنامه نويسي اسمبلي استmacroثانيا حاوي  §
  

  : بنويسيم نتيجه به شكل زير خواهد بودMASM   براي مثال اگر كد قبلي را بخواهيم در 
  

Function proc dwValue1:DWORD, dwValue2:DWORD 
   mov eax, dwValue1 
   .if eax == dwValue2 
      mov eax, 1 
   .else 
      mov eax, 2 
   .endif 
   ret 
Function endp 

  

  .باشد به طور قطع موافقيد كه كد فوق بسيار خواناتر مي

http://www.masm32.com/

 ٥

   اسمبليآشنايي با زبان
  
  

 سـالها  .تواند آنرا درك كند، بوجود آمده است بايناري خامي كه يك پردازشگر مي        زبان اسمبلي جهت جايگزيني كد    
 زبـان اسـمبلي   .شـدند   بـه زبـان اسـمبلي نوشـته مـي     هـا   برنامـه ،اي برنامه نويسي سطح بالا وجود نداشتند    زمانيكه زبانه ،  قبل  

  : براي مثال .تواند اجرا كند مستقيما بيانگر دستوراتي است كه يك پروسسور مي
add eax, edx  

  

ادير بـوده و درون پردازشـگر    حـاوي مق ـ  هـستند،  رجيـستر EDX و EAX. كند   اينستراكشن فوق دو مقدار را با هم جمع مي   
پردازشـگر آنـرا خوانـده و دسـتورات     ). در مبنـاي شـانزده  ( ترجمه مي شـود  C2 03 66  كد فوق نهايتا به.ذخيره مي شوند

 ابتدا كد خود را به اسمبلي تبديل و نهايتا آنـرا بـه كـدهاي باينـاري     C زبانهاي سطح بالاتري مانند .مربوطه را اجرا مي كند  
  : كرد خواهندترجمه

  
C code Assembly  Raw output (hex) 
a = a + b; >> Compiler > > add eax, edx >>Assembler>> 66 03 C2 

  

  .تواند درك كند  ، اينستراكشني است كه يك پردازشگر ميOpcode .شود  هم گفته ميOpcodesها     به اينستراكشن
  
  
  

  :بين اسمبلي داس و ويندوز مهمتفاوت 
  

 بـراي كـار بـا فايلهـا و      int 13 بـراي مثـال   . استفاده مي كننـد "تابع" بعنوان )interrupts (ها امه هاي تحت داس از وقفه   برن
اين اينترفيس حاوي توابعي اسـت كـه   . )Application Programming Interface ( سر و كار داريمAPI در ويندوز با .غيره

  .به مرور در ادامه ذكر خواهند شدنيز ها   ساير تفاوت. استفاده نماييدهاي خود از آنها شما مي توانيد در برنامه
  
  

  :متغيرها
  

هـاي حافظـه و رجيـسترها در اختيـار      آدرس  ، در اسـمبلر  . وجود خارجي ندارند   ++C در اسمبلر چيزي همانند متغيرها در       
 همچنـين در   نـدارد nMyIntegerام  بـا نـوع صـحيح بـه ن ـ    يبراي مثال پروسسور اطلاعي در مورد وجود متغيـر   .  هستند شما

دانـد در مـورد رجيـسترها و يـا       پروسسور تنها چيزي را كه مـي      .باخبر نيست  نيز   CMyClassمورد بكارگيري كلاسي به نام      
  .دسترسي به مقداري از حافظه در آدرسي مشخص است

 ٦

  رجيستر چيست و چگونه مي توان به حافظه دسترسي پيدا كرد؟! خوب
  
  

  :رجيسترها
  
ي مشخـصي از رجيـسترها در     تنها مجموعه.رود رجيستر همانند يك متغير است اما تنها براي استفاده پردازشگر بكار مي    

يـك رجيـستر متغيـري    : توان يك رجيستر را به ايـن صـورت نيـز تعريـف كـرد         مي .ي يك پردازشگر وجود دارند     تراشه
  .ر تراشه آن موجود استبه صورت فيزيكي داست از پيش تعريف شده در پردازشگر و 

  
توانـد پـردازش كنـد ،     ي تعداد بيتي كه يـك پردازشـگر در هـر لحظـه مـي      توانند اعدادي را هم اندازه   اين رجيسترها مي 

 نيـز  DWORD بـه آنهـا   ++C در . بيـت خواهـد بـود   ۳۲ بيتي ، اندازه اين اعـداد  ۳۲براي مثال در پردازشگري  . نمايش دهند 
   .شود گفته مي
  :در حالت هگز اينتل و يا سازگار با آن  بيتي۳۲ سه اندازه داده بومي پردازشگرهاي   نمايش

BYTE        00 
WORD     00 00 
DWORD   00 00 00 00 

  
در اينجا اگر به اعداد منفي نياز شـد ، منفـي يـك    ).  و همواره مثبت بدون علامت ( هستند   unsigned ،   همچنين اين اعداد  

در مورد اعداد درآينده بيشتر صـحبت خواهـد    ( و الي آخربوده 0xFFFFFFFE دو معادل  و منفيت اس 0xFFFFFFFFمعادل  
  .)شد

د در  عـدد از آنهـا را باي ـ  ۶   در پروسسورهاي جديد اينتل رجيسترهاي چندي از پيش تعريف شـده هـستند امـا شـما تنهـا        
  :هاي خود بكار بگيريد برنامه

  
eax - Accumulator Register 
ebx - Base Register 
ecx - Counter Register 
edx - Data Register 
esi - Source (for memory operations) register 
edi - Destination (for memory operations) register  

  

  : داريمEAXبراي مثال در مورد .    هر رجيستر در اينجا قابل تجزيه به بايتهاي سازنده آن است
al : ( در كلمه )پاييني(يت اولين باword ( پاييني ثبات)رجيستر (EAX  =۸بيت   
ah  :( در كلمه )بالايي( بايت دومينword ( پاييني ثبات)رجيستر (EAX = ۸بيت   
ax :  پاييني در رجيستر )  بايت۲(كلمهEAX = ۱۶بيت   

EAX :  بيت۳۲=  بايت ۴كل رجيستر معادل   

 ٧

  
  

00  00 0000 
ECX 

CX   
CL  CH     

  
00  00 0000 

EAX 
AX   

AL AH   
  
  

  
00  00 0000 

EDX 
DX   

DL  DH      

  
00  00 0000 

EBX 
BX   

BL BH    
  
  

  
  

 و ESIو البتـه نـه بـراي     ( نيـز وجـود دارد  EDX و ECX ، EBXهمانطور كه ملاحظه مي نماييد چنين تقسيم بندي در مـورد      
EDI(.  

  
 بيتـي تنهـا   ۱۶پروسـسورهاي  .  اسـت Extended در اينجا به معناي      'E' . است ها از نوع پروسسور مشتق شده        اين نامگذاري 

 بيـت  ۱۶ هنگاميكـه ايـن نـوع پروسـسورها توسـعه داده شـدند ،       . و غيـره بودنـد   ax, bx, cx, dxسترهايي مانند يداراي رج
 ۱۶اي دسترسـي بـه    هـيچ روش مـستقيمي بـر   .) مهيـا هـستند  +۳۸۶ يو تنها در پردازشگرها ( مشخص كردند Eاضافي را با  

  .بيت بالايي رجيسترها موجود نيست
  

  :اجزاي كوچكتر آن حاوي مقادير مشخص شده خواهند بود باشد ، EA7823BBh حاوي  EAX   براي مثال اگر 
  
  

EAX EA 78 23 BB 
AX EA 78 23 BB 
AH EA 78 23 BB 
AL EA 78 23 BB 

  
 
eax = EA7823BB (32-bit) 
ax = 23BB (16-bit) 
ah = 23 (8-bit) 
al = BB (8-bit) 

  
  

  :در حالت كلي انواع مختلفي از رجيسترها وجود دارند  

 ٨

  :رجيسترهاي عمومي كه براي هركاري مي توانند مورد استفاده قرار گيرند) الف
  

eax (ax/ah/al) Accumulator 
ebx (bx/bh/bl) Base 
ecx (cx/ch/cl) Counter 
edx (dx/dh/dl) Data 

  

  :كنند رفته شده را مشخص ميگت حافظه بكار ن كه سگمجيسترهاي سگمنتر) ب
  

CS code segment 
DS data segment 
SS stack segment 
ES extra segment 
FS (only 286+) general purpose segment 
GS (only 386+) general purpose segment 

  

 تحـت داس حافظـه بـه    .باشـد   مـي flat memory systemزيرا ويندوز يـك  . ها در ويندوز كاربردي ندارند اين سگمنت   
نياز به ذكر سگمنت و آفـست  ، ظه بنابراين براي مشخص كردن يك آدرس حاف     . تقسيم شده بود  ايي   64Kbهايي   سگمنت

  : براي مثال .بود 
0172:0500 (segment:offset) 

  

هـا همـواره رجيـسترهايي      سگمنت.ازي در بكارگيري آنها نيست بنابراين ني  . گيگ حجم دارند   ۴ها   در ويندوز سگمنت     
  .)بيشتر توضيح داده خواهد شددر قسمت معماري حافظه ها در ادامه  در مورد سگمنت(. بيتي هستند۱۶
  
  

  :گر رجيسترهاي اشاره) ج
  

esi (si) Source index 
edi (di) Destination index 
eip (ip) Instruction pointer 

  
نيـز مـي تـوان اسـتفاده كـرد      تا زمانيكه مقادير اصلي آنها حفظ گردند  نوع رجيسترها بعنوان رجيسترهاي عمومي      ز اين   ا 
 حـاوي  EIP .دليل نامگذاري آنها نيز استفاده از آنهـا بـه جهـت ذخيـره سـازي آدرس هـاي حافظـه اسـت        ) . eipالبته بجز  (

  .بنابراين از آن بعنوان رجيستري عمومي نمي توان استفاده كرد.داشاره گري است به اينستراكشني كه قرار است اجرا شو
  
  

  )stack(رجيسترهاي پشته ) د
esp (sp) Stack pointer 
ebp (bp) Base pointer 

  

   ESP موقعيت فعلي پشته را نگه مي دارد.EBPدر توابع بعنوان اشاره گر به متغيرهاي محلي استفاده مي شود .  

  :حافظه در ويندوز و داسنگاهي بر معماري 
   

     WIN 3.xx و DOS) الف
تقـسيم شـده   ) سگمنت هـايي ( ، حافظه به قطعاتي ) بيتي۱۶در برنامه هاي  (   در داس و  براي مثال ويندوز سه و يك دهم       

 و گرهـاي سـگمنت   در اين حالت براي دسترسي به حافظـه نيـاز بـه اشـاره         . ايي دارند  64Kbاي   ها اندازه  اين سگمنت  .است
  .آفست مي باشد

  
  :تقسيم بندي حافظه به قطعات   

MEMORY 
SEGMENT 1 

(64kb) 
SEGMENT 2 

(64kb) 
SEGMENT 3 

(64kb) 
SEGMENT 

4(64kb) and so on 

  
  :)آفست مكاني است درون يك سگمنت ( در يك سگمنتها نمايش آفست   

SEGMENT 1(64kb) 
Offset 1 Offset 2 Offset 3 Offset 4 Offset 5 and so on 

  
  . وجود داشته باشدتواند  آفست مي۶۵۵۳۶ون يك سگمنت حداكثر ر سگمنت و همچنين د۶۵۵۳۶در اينجا حداكثر    

  :   در اين حالت نمايش يك آدرس حافظه به صورت زير خواهد بود
SEGMENT:OFFSET 

  :   براي مثال 
0030:4012 (all hex numbers) 

   .ت ذكر شدهن در سگم۴۰۱۲ آفست و ۳۰   يعني سگمنت 
    

  : در ادامه توضيحات بيشتري در مورد آنها را خواهيد يافت.   در قسمت قبل با رجيسترهاي سگمنت آشنا شديد
  

• CS و يا  Code segment :  استاجرا درحال شماره قطعه اي كه كد جاري حاوي. 

• DS و يا Data segment : ا از آن تامين مي كندسگمنت داده اي كه قطعه جاري داده هاي خود ر.  
  

همچنين رجيسترهاي عمومي نيـز مـي تواننـد بـراي ايـن         (گر در اغلب اوقات حاوي آفست ها هستند           رجيسترهاي اشاره 
 نگهدارنـده ي  SP. كه هم اكنون در حال اجرا اسـت      )CSدر   (بيانگر آفست اينستراكشني است    IP .)منظور بكار برده شوند   

  .)SSدر  ( است)stack(آفست موقعيت فعلي پشته

  قسمت اول- آشنايي با برنامه نويسي اسمبلي ويندوز

 ١٠

   بيتي۳۲ويندوز ) ب
ر رجيـسترهاي  يي ـو البته با تغ  گيگابايت است۴ي آنها   و يا بالاتر نيز سگمنت ها وجود دارند اما اندازه ۹۵   در ويندوزهاي   

 flat memoryبـه ايـن حالـت    .افتـد   و از كـار مـي  ويندوز احتمالا دچار اختلال شديد در كاركرد خواهـد شـد  ، سگمنت 

model  تواننـد    مـي ۴۲۹۴۹۶۷۲۹۵ بيتي هـستند يعنـي دربـازه صـفر تـا      ۳۲ها وجود داشته و اكنون       تنها آفست .شود گفته مي
هـاي   ايـن مـورد يكـي از مهمتـرين مزيـت     .هر موقعيت در حافظه تنها با يك آفست مشخص مـي شـود          بنابراين  .قرارگيرند

توان رجيسترهاي سگمنت را فراموش كرد و بـر روي   بنابراين اكنون مي . بيتي است  ۱۶ بيتي نسبت به سيستمي      ۳۲مي  تسيس
اين مورد سبب ساده تر شدن برنامه نويسي اسمبلي و همچنين بـالا رفـتن سـرعت اجـراي كـد        .ساير رجيسترها متمركز شد   

  .گردد مي
  . يا بالاتر اجرا شوندو  ۳۸۶هاي  ها لزاما بايد بر روي پردازنده اينگونه برنامه   
ني كه تحت داس برنامه نويسي اسمبلي انجام داده اند فرمت فايلهاي اجرايـي وينـدوز از جهـاتي شـبيه            براي برنامه نويسا     

هـا كـار    و هر دو مـستقيما بـا آفـست   بوده كه شامل كد و داده است    داراي يك سگمنت     ،اس مي باشند     د comبه فايلهاي   
  .گيرند آفست براي آدرس دهي كمك نمي/از روش سگمنتنيز كنند و  مي

  . گيگابايت است۴ در بازه NEARاز نوع آدرس دهي ، ) flat(هاي مدل تخت   حالت پيش فرض آدرس دهي در برنامه  
  
  

  ) :Protected Mode Memory(ي محافظت شده  آشنايي با حالت حافظه
  
مت هـايي  قـس ) overwrite(، امكان جاي نويـسي   مانند سيستم عامل داس real modeدر سيستم عاملهايي با خصوصيت    

اين امر در صورت وقوع سـبب از كـار افتـادن    .اي كه صحيح نوشته نشده باشد ميسر است از كد سيستم عامل توسط برنامه 
  .باشد سيستم عامل خواهد گرديد و نهايتا نياز به بوت مجدد مي

شـروع مـي شـوند     B800 كـه از  CGA   براي مثال اگر تحت داس مشغول برنامه نويسي مـستقيم نـواحي حافظـه ويـديويي                
تواند كل ناحيه بالايي آنرا تا رسيدن به حداكثر بـازه آدرس دهـي در داس       باشيد ، برنامه در اثر اشتباهي در يك حلقه مي         

  .گردد مي در حافظه BIOS اين مورد سبب جاي نويسي همه چيز منجمله تنظيمات.جاي نويسي كند
ي  مـديريت حافظـه  .ن نـوع رخـدادها ايجـاد شـده اسـت     اي ـوقـوع  ي حافظـه جهـت جلـوگيري از             حالت محافظت شـده   

ي آدرسي را كه يك برنامه مي تواند در آن بنويسد را كنترل كرده و در صورت بروز تخلفي از ايـن            بازه  ،محافظت شده   
  .دهد برنامه را خاتمه مي، ي تعريف شده  بازه

 امـا بـدليل روش بكـار گرفتـه شـده در شـبيه سـازي        ، بيتي نيز مهيا بـود ۱۶در ويندوزهاي  ،     اين مدل محافظت از حافظه      
multitasking هـا بودنـد ،    هايي از حافظه كه تحت مالكيت سـاير برنامـه     در اين نوع ويندوزها ، امكان جاي نويسي قسمت

  . نيز ميسر بود

  قسمت اول- آشنايي با برنامه نويسي اسمبلي ويندوز

 ١١

   بيتـي ، ۳۲اي  بـر مبنـاي سـخت افـزار در وينـدوزه     multitaskingو پيـاده سـازي      تغييرات حاصل شده در سخت افزارها     
اساس برنامه نويسي در سيستم عـاملي بـا حالـت محافظـت     .ي محافظت شده را قابل اطمينان تر كرده است      مديريت حافظه 

  .اي است كه برنامه به آن دسترسي دارد خواندن و يا نوشتن در حافظهتنها ي حافظه ،  شده
  
آنـرا  دهد شـما بايـد بادقـت بيـشتري      ي آدرسي را مي در هر بازهي خواندن و نوشتن  با توجه به اينكه اسمبلر تقريبا اجازه      

 20k را بـراي خوانـدن تخـصيص داده باشـيد و سـعي در خوانـدن       10kبراي مثال اگر شما بافري بـه انـدازه ي         .بكار گيريد 
 بـه  exception از طرف سيستم عامـل بعنـوان يـك    خطاهااين نوع .مواجه خواهيد شد  page read faultنماييد ، با خطاي 

  .يابد  و اگر برنامه راهي را براي مواجه با آن پيش بيني نكرده باشد ، از طرف سيستم عامل خاتمه ميشدهبرنامه گوشزد 
  
  
 
  

   :Opcodesآشنايي با 
  

    Opcodes بـه  .هگـز هـستند  خام  كدهاي "متني قابل خواندن"در حقيقت اينها نگارش .هاي پردازشگر هستند   اينستراكشن
هـر دسـتور آن مـستقيما بـه كـدهاي هگـز        (سطح پايين ترين زبان برنامه نويسي است كه وجود دارد   ،  ل اسمبلي   همين دلي 

 و گـردد   خـود بخـود حـذف مـي     و تبديل زباني سطح بالا به زباني سـطح پـايين   به بياني ديگر فاز كامپايل.)شود تبديل مي 
  .هاي خام هگز است ه دادهگيرد تبديل كدهاي اسمبلر ب تنها كاري كه در اينجا صورت مي

درون مدرات مربوطـه  ) ها اينستراكشن(ستورالعمل ها د  ،در سطح سخت افزاري پردازشگرهاي اينتل و يا سازگار با آن             
  .شود  گفته ميOpcodesو به آنها  اند قرار گرفته

  
 اسـتفاده  A1مـساوي    Opcodeاز  ، EAXبه درون رجيستر   A7 00 FE 56  بيتي مساوي۳۲   براي مثال براي كپي مقداري 

  :بنابراين نتيجه نهايي مساوي است با . )Mov EAX  (شود مي
A1 FE 00 A7 56 

  
اسـتفاده  در  تـا  ه مشابه نامهايي اعطـاء گرديـد  Opcodesبه گروهي از تر كردن برنامه نويسي اسمبلي    جهت ساده  در اينجا    

  .شود  گفته ميmnemonics نامها به اين.از آنها نيازي به فراگيري كدهاي هگز نباشد
 كـه كـاري مـشابه را در پردازشـگر انجـام      Opcodesاي از     نامي است اختصاص داده شده بـه خـانواده          mnemonic   يك  

  .دهند مي
  
  

http://mohandesyar.com


  قسمت اول- آشنايي با برنامه نويسي اسمبلي ويندوز

 ١٢

  : محاسباتيOpcodesتعدادي معرفي 
  
  

MOV 

  
   
هـا درون پروسـسور     ، انتقال دادهآن كاراين اينستراكشن براي كپي مقداري از يك مكان به مكاني ديگر بكار مي رود و               

  :براي مثال. باشد مي
mov eax, 100 

  

بـه اينگونـه اعـداد،    . دانـست eax=100خط فـوق را مـي تـوان معـادل     .  انتقال داده مي شود   EAX به رجيستر    ۱۰۰در اينجا     
Immediate numbersشود  نيز گفته مي.  

  
  : و مقصد بايد هم اندازه باشند  حالت كلي اين دستورالعمل به شكل زير است و مبدا

  
mov (destination), (source)  

  
  :)اندازه مبدا و مقصد يكي نيست (براي مثال دستور زير نادرست است

mov al, ecx ; NOT VALID 

  
  
  

  : مثالهايي بيشتر در اين زمينه 
         
     
mov al, bl ;move the lower byte of ebx into the lower byte of eax 
mov al, 0ffh        ;move 0xFF into the lower byte of eax 
mov ah, 0ffh ;move 0xFF into the high byte of the low word (2-bytes) of 

eax 
mov ax, 0ffffh ;move 0xFFFF into the low word of eax 
mov eax, 0ffffh ;move 0xFFFF into eax 
mov eax, 12345678h ;mov loads a value into a register (note: 12345678h is a  

;hex value because of the 'h' suffix) 
mov cl, ah ;move the high byte of ax (67h) into cl 
sub cl, 10 ;substract 10 (dec.) from the value in cl 
mov al, cl ;and store it in the lowest byte of eax. 

  قسمت اول- آشنايي با برنامه نويسي اسمبلي ويندوز

 ١٣

  :دريافت آدرس و محتواي آدرسي در حافظه
  

بـه يـك رجيـستر و يـا بـرعكس      ) شـود  نيز گفتـه مـي   "محتواي"كه به آن  ([]توان محتواي حافظه را نيز با استفاده از     مي
  :منتقل نمود 

mov al, [esi] ;move the byte contained in the memory address 
;in register esi into the lower byte of eax 

mov [edi], bl ;move the byte value in the lowest byte of ebx 
;into the memory address in register edi 

mov cx, [esi] ;move the word (2-byte) value contained in the 
;memory address of register esi into the lower 
;word of ecx 

mov [edi], edx ;move the dword (4-byte) value contained in edx 
;into the memory address contained in register edi 

                    
  
  

  :توان در اينجا آفست را نيز در نظر گرفت    همچنين مي
  

mov al, [esi + 3] ;move the byte contained in the memory address 
;in register esi + 3 into the lower byte of eax 

mov [edi + 2], dx ;move the lower word (2-bytes) contained in 
;edx into the memory address contained in the 
;register edi + 2 

                      
  

  :مثال جدول زير را در نظر بگيريددرك موارد فوق ،    براي 
offset 34 35 36 37 38 39 3A 3B 3C 3D 3E 3F 40 41 42 

data 0D 0A 50 32 44 57 25 7A 5E 72 EF 7D FF AD C7 

  
البتـه بـديهي اسـت كـه     .شـوند  هاي حافظـه بـا آفـست مـشخص مـي       ويندوز موقعيت   همانطور كه پيش تر نيز ذكر شد در 

براي دريافت محتواي آن مي شـود از دسـتور   . است0000003Ah معادل با 3Aبراي مثال آفست . بيتي هستند۳۲مقادير فوق   
  :زير استفاده كرد

mov  eax, dword ptr [0000003Ah] 

  
بعـد از اجـراي دسـتور فـوق مقـدار      . قرار بـده eax را در رجيستر 3Aيتي آفست  ب۳۲معناي خط فوق اين است كه مقدار        

eax 725 مساويE7A25hهـا در حافظـه بـه فرمـت      همانطور كه ملاحظه نموديد ترتيب قرار گيـري داده . خواهد بودlittle 

endian مثال زير اين مطلب را روشن تر خواهد نمود.). شود ترتيب بايتها معكوس مي( است:  
 4050 بيتـي  ۱۶ ذخيره مي شود و همچنين مقـدار  10 ,20 ,30 ,40 در حافظه به صورت  hex 10203040  بيتي۳۲ قدار   م

hex شود ذخيره مي  40 ,50  در حافظه به صورت.  

  قسمت اول- آشنايي با برنامه نويسي اسمبلي ويندوز

 ١٤

  

  
  .)جايي كه در حافظه واقع شده است (آدرس يك متغير در حافظه -۲ شكل

  
  

  
  .در حافظهذكر شده   مقدار آدرس-۳ شكل

  
  
  :گويند  نيز ميdereferencingبه عمليات زير    

mov eax, lpvar ; copy address into eax 
mov eax, [eax] ; dereference it 
mov nuvar, eax ; copy eax into new variable 

  
  :هاي مختلف نيز ميسر است    همچنين دريافت مقادير حافظه در اندازه

mov  cl, byte ptr [34h] ; cl will get the value 0Dh (see table above) 
mov  dx, word ptr [3Eh] ; dx will get the value 7DEFh (see table above, remember the reversed byte order) 

  
  : براي مثال.)شود  بيت در نظر گرفته مي۳۲به صورت پيش فرض  (   ذكر اندازه مقادير دريافتي الزامي نيست

mov  eax, [00403045h] 

  
  :)به سايز نهايي منتقل شده دقت بفرماييد (   مثالي ديگر

mov  eax, 403045h ;make eax have the value 403045 hex. 
mov  cx, [eax] ;put the word size value at the memory location EAX (403045) into register CX. 

  
  

 كـه كـاري شـبيه بـه     Opcodesاي از   است اختصاص داده شده براي خانودهنامي،  movتوان نتيجه گرفت كه    بنابراين مي 
  .دهند هم را انجام مي

  قسمت اول- آشنايي با برنامه نويسي اسمبلي ويندوز

 ١٥

  :اندازه متغيرهايي كه مي توان انتقال داد
  

بـراي مثـال   .متفـاوت اسـت قـرار داد   اصـلي  اي را در رجيستري كه اندازه آن با اندازه داده      توان داده     در حالت كلي نمي   
  :تعبارت زير نامعتبر اس

mov eax, cl ; this fails as eax is 32 bit, cl is 8 bit 

  
  :هاي مختلفي تبديل كرد  بيتي قرار داد بايد ابتدا آنرا با استفاده از روش۳۲ را در يك رجيستر cl   براي اينكه بتوان 

movzx  eax, cl ; zero extend unsigned integer 
movsx  eax, cl ; sign extend signed integer 

  
  :و يا    

Xor  eax, eax ; clear eax 
mov  al, cl ; copy cl into al 

  
  :و يا    

mov  al, cl ; copy cl into al 
cbw ; convert BYTE in AL to WORD in AX 
cwde ; convert WORD in AX to DWORD in EAX 

  
  

  :توان در رجيسترها قرار داد اي كه مي هاي داده نوع
  
  . ، حافظه و يا رجيستري ديگرimmediate: توان در رجيسترها قرار داد  اي پايه را مي مقدار دادهسه نوع    

كه توسط اسـمبلر بـه معـادل اسـكي خـود         (تواند يك رشته نيز باشد       ميمعمولا يك عدد است اما        Immediate    عملوند
  : براي مثال.)شود تبديل مي

mov  al, "a" ; string literal 
mov  edx, 0 ; numeric immediate 

  
تنها بايد دقت داشت كه انتقال يـك عملونـد حافظـه بـه ديگـري      . در مورد حالتهاي ديگر پيش تر به تفصيل بحث شد و     

بنـابراين عبـارت زيـر نـامعتبر     .  ايي تعريف شده براي آن در پردازشگر وجـود خـارجي نـدارد           opecodeزيرا  . ميسر نيست 
  :است

! mov mVar, lpMem ; this fails, no opcode to do it. 
  
  :اين عمليات را به صورت زير مي توان انجام داد   

! mov eax, lpMem ; copy memory value into register. 
! mov mVar, eax ; copy register into memory value. 

  قسمت اول- آشنايي با برنامه نويسي اسمبلي ويندوز

 ١٦

  : مهيا است نيز   همچنين روش كندتري براي اين منظور
! push lpMem ; push memory value onto the stack. 
! pop mVar ; pop it back off as another memory value. 

  
  

  :گرها اشاره
  
  :در اسمبلي هنگاميكه از دستورالعملي مانند خط زير استفاده مي شود   

Lea  eax, MyVar 

  
  :شود   به متغير ديگري كپي ميeaxگيرد و سپس هنگاميكه   قرار ميeax   آدرس متغير در 

Mov  lpMyVar, eax 

  .به آدرس ذكر شدهبود گري خواهد  اشاره   اين متغير 
  
  :)by value (گر بوسيله مقدار آن  انتقال يك اشاره   

mov eax, lpMyVar ; copy the VALUE into eax 
push eax ; push it as a parameter 
call MyProcedure ; call the procedure 

  
   :)by reference (گر بوسيله ارجاعي از آن ل يك اشارهو يا انتقا   

lea eax, lpMyVar ; load the ADDRESS into eax 
push eax ; push it as a parameter 
call MyProcedure ; call the procedure 

  
  :  در اين حالت پس از خاتمه روال بايد از روش زير براي بدست آوردن آدرس استفاده كرد 

mov eax, lpMyVar 
mov eax, [eax] 

  
  

ADD, SUB, MUL, DIV 

  
. عملكرد بسياري از آنها از نامشان قابـل حـدس زدن اسـت   . شوند  در محاسبات بكارگرفته مي   Opcodes   تعداد زيادي از    

Add، جهت جمع  sub ، جهت تفريق Mul جهت ضرب و div بكار مي روند تقسيمبه منظور .  
  

   :Add   نحوه استفاده از 
add  destination, source 

 ١٧

  :در اين حالت خواهيم داشت   
destination = destination + source 

  
  :براي مثال 

  
Destination Source Example 

Register Register add ecx, edx 
Register Memory add ecx, dword ptr [104h] / add ecx, [edx] 
Register Immediate value add eax, 102 
Memory Immediate value add dword ptr [401231h], 80 
Memory Register add dword ptr [401231h], edx 

  
  :  و براي تفريق و ضرب خواهيم داشت 

Sub  destination, source ;(destination = destination - source) 
mul  destination, source ;(destination = destiantion * source) 

  
  :)گيرد قرار مي  edx و باقيمانده در eaxخارج قسمت در  ، sourceبسته به اندازه  (تقسيم كمي متفاوت استعمليات    

div source ; (eax = eax / source, edx = remainer)  

  
Source size Division Quotient stored in ...  Remainder Stored in...  

BYTE (8-bits) ax / source AL AH 

WORD (16-bits) dx:ax* / source AX DX 

DWORD (32-bits) edx:eax* / source EAX EDX 

  
  :تواند باشد   مبدا در اين حالت يكي از موارد زير مي

• an 8-bit register (al, ah, cl,...)  
• a 16-bit register (ax, dx, ...)  
• a 32-bit register (eax, edx, ecx...)  
• an 8-bit memory value (byte ptr [xxxx])  
• a 16-bit memory value (word ptr [xxxx])  
• a 32-bit memory value (dword ptr [xxxx])  

  :   تذكر 
  .تواند تعيين كند  باشد زيرا پردازشگر اندازه آنرا نمينبايدمبدا در اين حالت يك عدد معمولي 

 ١٨

BITWISE OPERATIONS 

  
روش عمليـات نيـز بـه صـورت     ) . Notبجـز  (بر روي مقادير مبدا و مقـصد عمـل مـي كننـد        هاي بيتي     مام اينستراكشن   ت   

  :خلاصه مطابق جدول زير است
  

Instruction AND OR XOR NOT 
Source Bit 0 0 1 1 0 0 1 1 0 0 1 1 0 1 
Destination 
Bit 0 1 0 1 0 1 0 1 0 1 0 1 X X 

Output Bit 0 0 0 1 0 1 1 1 0 1 1 0 1 0 
  

  :مثال يك 
mov    ax, 3406 
mov    dx, 13EAh 
xor      ax, dx 
ax = 3406 (decimal), which is 0000110101001110 in binary. 
dx = 13EA (hex), which is 0001001111101010 in binary. 

  :مثال دو
mov ecx, FFFF0000h 
not ecx 
FFFF0000 is in binary 11111111111111110000000000000000 (16 1's, 16 0's) 
If you take the inverse of every bit, you get: 
00000000000000001111111111111111 (16 0's, 16 1's), which is 0000FFFF in hex. 
So ecx is after the NOT operation 0000FFFFh. 

  
  

 

IN/DECREMENTS 

  
  . كنند يا زياد ميو ادير مكاني از حافظه و يا يك رجيستر را يك واحد كم ها مق    اين اينستراكشن

  
inc   reg  -> reg = reg + 1 
dec  reg -> reg = reg - 1 
inc   dword ptr [103405]  -> value at [103405] will increase by one. 
dec  dword ptr [103405] -> value at [103405] will decrease by one. 

 ١٩

NOP 

  
يـك كـد   ) patching(و يا وصله كـردن  صرفا جهت پركردن فضا !    اين اينستراكشن مطلقا كار خاصي را انجام نمي دهد 

  .گيرد مورد استفاده قرار مي
  
  

Bit Rotation and Shifting 

  
  .دكنن صرفا جهت روشن شدن مطلب استفاده مي،  بيتي ۸ ياكثر مثالهاي زير از اعداد:    تذكر 

  
  :توابع شيفت

Shifting functions 
SHL   destination, count 
SHR   destination, count 

  
   SHL و SHR به تعداد countدهند هاي يك مكان از حافظه و يا يك رجيستر را به چپ و راست انتقال مي  ، بيت.  

  : مثال 
; al = 01011011 (binary) here 
shr   al, 3 

  

  
 SHRانجام ي   نحوه– ۴ شكل

  
 خواهـد  ۰۰۰۰۱۰۱۱ بنـابراين نتيجـه معـادل    . را سـه مكـان بـه سـمت راسـت انتقـال بـده            al  بدين معنا كه تمام بيـت هـاي         

هـايي    بيـت . حـذف خواهنـد شـد    هاي سمت راست، هايي كه در سمت چپ قرار مي گيرند با صفر پر شده و بيت     بيت.بود
 ايـن  . پردازشـگر Flag ، بيتـي اسـت در رجيـسترهاي    carry-bit .د ذخيـره خواهنـد ش ـ  carry-flag  درگردنـد  كه حذف مي

ار ك ـ مخـصوص اين Opcodesهـر چنـد    ( و غيره به صورت مستقيم قابل دسترسي نيستند  eaxرجيسترها همانند رجيسترهاي    
ه حـذف شـده   اين بيت معادل است با آخرين بيتي ك. ها  اما محتواي آنها وابسته است به نتايج اينستراكشن     .)نيز وجود دارد  

  .است

 ٢٠

   shl همانند shrها را به سمت چپ انتقال مي دهد  است اما با اين تفاوت كه بيت.  
  :مثال 

; bl = 11100101 (binary) here 
shl   bl, 2 
 

  
  SHLي انجام   نحوه–۵شكل 

  
ر گرفتنـد بـا صـفر    هايي كه در سمت راست قرا بيت. خواهد بود  ۱۰۰۱۰۱۰۰ مساوي   bl   پس از اجراي دستور فوق مقدار       

  .باشد  مساوي يك ميcarry-bitپرشده و 
  

 بـه   بيـت n بـه ميـزان  شـيفت يـك مقـدار    . براي انجام ضرب و تقسيم هاي سريع نيز مي توان استفاده كرد        SHR و   SHL   از  
  :براي مثال  . 2n مساوي است با آن مقدار ضربدر چپ

Mov   dl,5 
=20 Shl    dl,2    ; dl = 5 * 22  

  
  :براي مثال . 2n بيت به راست مساوي است با تقسيم آن مقدار بر nنين شيفت يك مقدار به ميزان    همچ

mov   dl,80 
shr     dl,1 ; DL = 40 
shr     dl,2 ; DL = 10 

  
  

  : ديگر نيز در اين زمينه وجود داردOpcodes   دو 
SAL destination, count ;(Shift Arithmetic Left) 
SAR destination, count ;(Shift Arithmetic Right) 

  

  
 SARي انجام   نحوه–۶شكل 

 ٢١

   SAL  بـا SHLامــا . يكــي مــي باشــد SAR دقيقــا بــا SHRيكــي نيــست .SARهــاي ســمت چــپ را بــا صــفر پــر   هميــشه بيــت
 Most significant bit ويـا  MSBدر حقيقـت بـا    (اگر عدد منفي باشد ، بيت هاي سمت چپ با يك پر خواهد شد.كند نمي

  .)شود پر مي
اما اگر بجـاي  .  خواهد شد00111110b بيت به سمت راست شيفت دهيم حاصل ۲ را  ۵ بيتي منفي    ۸   براي مثال اگر عدد     

SHR از SAR شود  مي11111110 استفاده كنيم حاصل نهايي.  
  
  

  :توابع چرخش
  

Rotation functions 
rol destination, count ; rotate left 
ror destination, count ; rotate right 
rcl destination, count  ; rotate through carry left 
rcr destination, count  ; rotate through carry right 

  
انـد در جهتـي ديگـر بـه درون انتقـال       هـايي كـه بـه بيـرون رانـده شـده       بيـت    توابع چرخش همانند توابع شيفت هستند اما     

  .يابند مي
  

ROR:   
  

  
 RORي انجام   نحوه–۷شكل 

  
   : RORمثالي در مورد 

    Bit 7 Bit 6 Bit 5 Bit 4 Bit 3 Bit 2 Bit 1  Bit 0   
Before   1 0 0 1 1 0 1 1   
Rotate, count= 3         1 0 0 1 1 0 1 1 (Shift out) 
Result   1 1 0 1 0 0 1 1   

  
هر بيتـي كـه بـه بيـرون شـيفت داده شـده از سـمتي        . دوران يافته استبيت ها   همانطور كه در جدول فوق مشخص است        

  carry-bitكـه بـه بيـرون انتقـال داده شـده در       آخـرين بيتـي     هماننـد اينستراكـشن هـاي شـيفت ،    .ديگر وارد گرديده است  
  .شود ذخيره مي

 ٢٢

  :مثالي ديگر
mov al,11110000b 
ror al,1 ; AL = 01111000b 
 
mov dl,3Fh 
ror dl,4 ; Dl=F3h 

  
  

ROL:   
  

  
 ROLي انجام   نحوه–۸شكل 

  
  :مثال 

mov al,11110000b 
rol al,1 ; AL = 11100001b 
 
mov dl,3Fh 
rol dl,4 ; DL = F3h 

  
   RCR و RCLد شد شبيه به ن كه در ادامه توضيح داده خواهROR و ROL هستند اما از carry bit     بـراي تعيـين آخـرين بيـت 

  .ندخارج شده استفاده مي كن
  

RCL:   
  

  
  RCLي انجام   نحوه–۹شكل 

  :مثال
clc  ; CF = 0 
mov bl,88h ; CF,BL = 0 10001000b 
rcl bl,1  ; CF,BL = 1 00010000b 
rcl bl,1  ; CF,BL = 0 00100001b 

  
  

RCR:   
  

  
 RCRي انجام   نحوه–۱۰شكل 

  
  :مثال 

stc ; CF = 1 
mov ah,10h ; CF,AH = 00010000 1 
rcr ah,1 ; CF,AH = 10001000 0 

  
  

  :مثالي ديگر
  
  ):۷بيت هاي صفر تا (براي خواندن اولين بايت رجيستر به صورت زير عمل مي كنيم .  را در نظر بگيريدEAXرجيستر    

mov byteval, al ; move 1st byte size value in to variable. 

  
  :)۱۵ تا ۸بيت هاي  (   براي خواندن دومين بايت رجيستر داريم

mov byteval, ah ; move 2nd byte size value into a variable. 

  
  :)۱۵بيت هاي صفر تا  ( آن word   براي خواندن اولين 

mov wordval, ax ; move 1st word size value into a variable. 

  
  

تـا بتـوان آنهـا را بـا اسـتفاده از      هاي موجـود در آن چرخانـده شـوند     بايد بيت،  رجيستر ۳۱ تا   ۱۶هاي      براي خواندن بيت  
  :براي مثال.استخراج كردهاي موجود  اينستراكشن

 ٢٤

rol  eax, 16 ; rotate EAX by 16 bits 
or 
ror  eax, 16 ; rotate EAX by 16 bits 

  
  .گيرند  بيت پايين قرار مي۱۶ بيت بالا در ۱۶ بالا قرار گرفته و متقابلا ۱۶ بيت پايين در ۱۶در اين حالت    
  
  

Exchange 

  
  .رود    جهت جابجايي دو رجيستر و يا يك رجيستر و مكاني در حافظه بكار مي

eax = 237h 
ecx = 978h 
xchg  eax, ecx 
eax = 978h  
ecx = 237h  

 ٢٥

  : اسمبلي)source (هاي منبع ساختار فايل
  
  :ندشو تقسيم ميزير  به قطعات )asm.* (فايلهاي منبع اسمبلي   

.code ; code section starts here 

.data ; data section starts here 

.data? ; un initialized data starts here 

.const ; constants section starts here 

  .ها و غيره هستند كه در آينده در مورد آنها بيشتر توضيح داده خواهد شد    ساير قطعات در مورد ريسورس
  

هـاي برنامـه بـوده و داراي دسترسـي خوانـدن و       حـاوي داده  ،   Dataقـسمت   . حاوي كدهاي برنامـه اسـت     ،   code  قسمت  
   .گيرد  نهايي قرار ميexe در فايل dataكل قسمت .باشد نوشتن مي

ي در نظـر    لحـاظ نـشده و جزئـي از حافظـه    exeدر فايـل   .هايي بدون مقدار دهي اوليـه هـستند         حاوي داده  ?dataقسمت     
  .باشند فته شده توسط ويندوز براي برنامه ميگر
در كـل بهتـر اسـت ثوابـت در فايلهـاي ضـميمه       .هايي ثابت با دسترسي فقـط خوانـدني هـستند        حاوي داده  constقسمت     

  .كرد توان استفاده   ميimmediate از آنها بعنوان مقادير. تعريف شده و سپس به برنامه ملحق شوند
  

قطعـات  . هـستند PE و يـا  portable executable و غيـره داراي قالـب   exe ، dllيندوز مانند فايلهـاي     فايلهاي اجرايي در و
  :شوند چند ويژگي خاص مشخص مي با PE-headerفوق در

  
Section name 
RVA (relative virtual address) 
Offset 
Raw size 
Virtual size and flags. 

 
  
   RVA   نسبي در اينجا به معناي نسبت به .شود ميجايي كه قطعه مورد نظر بارگذاري ، مكاني است نسبي در حافظه base 

addressشود بارگذاري ميدر آن آدرسي در حافظه كه برنامه  ( است(.  
  
  exeآفـست ، آفـستي خـام در فايـل       . قابـل تغييـر اسـت       PE-loaderقرار دارد اما توسـط        PE-header اين آدرس نيز در        

اي اسـت كـه نهايتـا در حافظـه خواهـد       انـدازه مجـازي ، انـدازه   .هـايي آغـازين در آن قـرار دارنـد       ه داده جـايي ك ـ  ،  است  
  .هايي هستند براي حالتهاي فقط خواندني ، اجرايي و غيره  ، پرچمflags.داشت

 ٢٦

  :يك مثال 

.data 
Number1  dd 12033h 
Number2  dw 100h,200h,300h,400h 
Number3  db "blabla",0 
 
.data? 
Value  dd ? 
 
.code  
mov  eax, Number1 
mov  ecx, offset Number2 
add   ax, word ptr [ecx+4] 
mov  Value, eax 

  
  exe هر كدي كه قرار گيرد بـه فايـل   ،برنامهداده در قطعات .   برنامه فوق به خوبي اسمبل نخواهد شد اما اهميتي هم ندارد  

در قـسمت  .گيرنـد   در مكانهاي مشخصي از حافظه قرار مـي شود  در حافظه بارگذاري مي   نيز منتقل شده و هنگاميكه برنامه     
data      برنامه فوق سه برچسب )label (Number1  ، Number2 و Number3خـود اين برچسب ها آفست موقعيت . قرار دارند 

  .برنامه استفاده كرددر ها  برچسبتوان جهت تعيين مكان  بنابراين از آنها مي.كنند را در برنامه نگهداري مي
  :در اينجا    

DD  = dword 
DW = word 
DB  = byte 

  .باشند ها مي توان بكار گرفت زيرا آنها ليستي از بايت   رشته ها را نيز مي
  :ها در حافظه به شكل زير خواهد بود قسمت داده،    در كد فوق 

33,20,01,00,00,01,00,02,00,03,00,04,62,6c,61,62,6c,61,00 (all hex numbers) 
(every value is a byte) 

  
 Number3 و ۰۰ بـه موقعيـت   Number2 ، قرار دارد اشاره مي كنـد ۳۳ به موقعيتي در حافظه كه بايت  Number1   در اينجا   
  . اشاره دارند) با حروف درشت مشخص شده اند (۶۲به موقعيت 

  
  :   اگر در برنامه خود از دستور زير استفاده كنيد 

mov  ecx, Number1 

  :يعني 
mov  ecx, dword ptr [location where the dword 12033h is in memory] 

  
  .شود  دريافت ميNumber1  مقدار موجود در موقعيت حافظه 

  اما 
mov ecx, offset Number1 

  :به معناي دستور زير است
mov ecx, location where dword 12033h is in memory 

  .شود  در حافظه دريافت ميNumebr1  موقعيت 
  

  :   بنابراين  نتيجه نهايي دو دستور زير يكي است
(1) 
mov ecx, Number1 
 
(2) 
mov ecx, offset Number1 
mov ecx, dword ptr [ecx] ( or mov ecx, [ecx]) 

  
  
  

  :براي تعريف داده ها به شكل زير نيز مي توان عمل كرد
.data? 
ManyBytes1  db  5000  dup (?) 
 
.data 
ManyBytes2  db  5000  dup (0) 

  
  

  :  در اينجا 
5000 dup = 5000 duplicates 

   . Value db 7 dup (4) معادل است با Value db 4,4,4,4,4,4,4براي مثال    
  
  

ر حافظه بـراي آن   بايت د۵۰۰۰قرار نگرفته و صرفا اجرايي نهايي    ، در فايل     ManyBytes1  با توجه به قسمت قرار گرفتن       
  .كند  بايت بيشتر مي۵۰۰۰ در فايل اجرايي قرار گرفته و حجم آنرا ManyBytes2شود اما  در نظر گرفته مي

 ٢٨

  :هاي شرطي پرش
  

  :هايي به شكل زير نيز مي توان استفاده كرد  از برچسبcode   در قسمت 

.code 

mov  eax, edx 
sub   eax, ecx 
cmp  eax, 2 
jz  loc1 
xor   eax, eax 
jmp  loc2 
loc1: 
xor   eax, eax 
inc   eax 
loc2: 

  :نكته 
xor eax, eax به معناي  eax=0است .  

  
  :   آناليز كد فوق به شكل زير است

Mov  eax, edx ; put edx in eax 
sub   eax, ecx ; substract ecx from eax 
cmp  eax, 2 

  
 اگـر ايـن دو مقـدار    .خـوريم  رود بر مـي  بكار ميدو مقدار  كه جهت مقايسه cmpام    در اينجا به اينستراكشن جديدي به ن     

در  . داخلـي flagبيتي است در رجيـستر  ،  carry همانند Zeroflag .تنظيم خواهد شد )Z-flag) zeroflag ،با هم مساوي بودند
  :ادامه داريم

jz  loc1 

  
  jz  يعنـي  .  يك پرش شـرطي اسـتJz = jump if zero ،   يـا بعبـارتي پـرش كـن اگـر      وzeroflag   باشـد  تنظـيم شـده.loc1 

  :ادامه كد بصورت زير است .برچسبي است براي پرش به اينستراكشني كه در پرش شرطي ذكر شده است
  

cmp  eax, 2 ; set zero flag if eax=2 
jz  loc1 ; jump if zero flag is set 

  
  . پرش كنloc1 بود به ۲ مساوي eaxاگر :    بصورت خلاصه دو خط فوق بدين معنا هستند 

  
  :  در كد فوق دستور زير به معناي پرش غير شرطي است و همواره اجرا مي شود 

jmp   loc2  

 ٢٩

  : ترجمه كنيم حاصل به شكل زير خواهد بودC   اگر كد فوق را بخواهيم به زبان 

if ((edx-ecx)==2) 
{ 
     eax = 1; 
} 
else 
{ 
     eax = 0; 
} 

  
  
  

   :flagيسترهاي رج
  

      : به شرح زير هستند flag   مهمترين و پركاربردترين رجيسترهاي 
  

  معنا رجيستر
ZF (Zero flag) اين پرچم زماني تنظيم مي شود كه نتيجه محاسبات صفر باشد.  
SF (Sign flag)  اگر تنظيم شده بود بدين معنا است كه نتيجه محاسبات منفي است.  
CF (Carry flag)  گيرد حاوي آخرين بيتي است كه پس از محاسبات در سمت چپ قرار مي.  

OF (Overflow flag)  بدين معنا كه نتيجه محاسبه در مقصد جاي نگيرد(.بيانگر سرريز محاسبات است(.  

 ٣٠

  :ها جدول انواع پرش
  
  :اند ورده شده آهاي شرطي بر اساس وضعيت پرچم مربوطه در جدول زير ليستي از انواع پرش   
  

Opcode Meaning Condition 

JA Jump if above CF=0 & ZF=0 

JAE Jump if above or equal CF=0 
JB Jump if below CF=1 
JBE Jump if below or equal CF=1 or ZF=1 
JC Jump if carry CF=1 
JCXZ Jump if CX=0 register CX=0 
JE (is the same as JZ) Jump if equal ZF=1 
JG Jump if greater (signed) ZF=0 & SF=OF 
JGE Jump if greater or equal (signed) SF=OF 
JL Jump if less (signed) SF != OF 
JLE Jump if less or equal (signed) ZF=1 or SF!=OF 
JMP Unconditional Jump - 
JNA Jump if not above CF=1 or ZF=1 
JNAE Jump if not above or equal CF=1 
JNB Jump if not below CF=0 
JNBE Jump if not below or equal CF=1 & ZF=0 
JNC Jump if not carry CF=0 
JNE Jump if not equal ZF=0 
JNG Jump if not greater (signed) ZF=1 or SF!=OF 
JNGE Jump if not greater or equal (signed) SF!=OF 
JNL Jump if not less (signed) SF=OF 
JNLE Jump if not less or equal (signed) ZF=0 & SF=OF 
JNO Jump if not overflow (signed) OF=0 
JNP Jump if no parity PF=0 
JNS Jump if not signed (signed) SF=0 
JNZ Jump if not zero ZF=0 
JO  Jump if overflow (signed) OF=1 
JP Jump if parity PF=1 
JPE Jump if parity even PF=1 
JPO Jump if paity odd PF=0 
JS Jump if signed (signed) SF=1 
JZ Jump if zero ZF=1 

  
  :براي مثال .   البته در اكثر مواقع نيازي به دانستن وضعيت پرچم مربوطه نيست و معناي اينستراكشن بسيار واضح است

"Jump if above" means: 
cmp  x, y 
jump if x is above y 

 ٣١

  :توضيحاتي در مورد اعداد
  
 در اسـمبلر  .نويسي تنها به تعريف متغيـر وابـسته اسـت   استفاده از اعداد صحيح و يا اعشاري در بسياري از زبانهاي برنامه              

در مـورد   . انجـام مـي شـود   FPU co-processorمحاسبات اعشاري توسـط رجيـسترهايي در   .اين مورد كاملا متفاوت است
  .اين رجيسترها در آينده بيشتر صحبت خواهد شد

 همـانطور  unsigned اعـداد . هـستند unsigned  وsigned اعداد صحيح ،  سيدر زبان   : اعداد صحيح    •
 براي درك اين موضوع به جدول زير دقـت بفرماييـد       . مثبت هستند  يتر ذكر گرديد اعداد    كه پيش 

  :)اندازه ها در مثال زير بايت هستند(
  

Value 00 01 02 03 ... 7F 80 ... FC FD FE FF 
Unsigned meaning 00 01 02 03 ... 7F 80 ... FC FD FE FF 
Signed meaning 00 01 02 03 ... 7F -80 ... -04 -03 -02 -01 

  
اعداد منفي قرار دارنـد و  ،  FF تا ۸۰ مربوط به اعداد مثبت و از 7Fبازه ي صفر تا ،  در اعداد علامت دار     

ــت   ــال در حال ــراي مث ــا  DWORDب ــفر ت ــازه ص ــت و از  7FFFFFFF ، ب ــت اعــداد مثب ــا ۸۰۰۰۰۰۰۰ جه  ت
FFFFFFFFh    انطور كه ملاحظـه مـي نماييـد در اعـداد منفـي      هم.مربوط به اعداد منفي استMSB  تنظـيم 

  .شود  به اين بيت ، بيت علامت نيز گفته مي.شده است
  

بـراي  . ، اعمال جمع و تفريق بدون هيچ نكته خاصي قابـل انجـام اسـت        )اعداد علامت دار   (   در مورد اعداد منفي و مثبت     
  :مثال 

Calculate: -4 + 9 
FFFFFFFC + 00000009 = 00000005. (which is correct) 
 
Calculate 5 - (-9) 
00000005 - FFFFFFF7 = 0000000E (which is correct, too ( 5 - -9 = 14) 

  
  :خاص زير وجود دارندهاي  تقسيم و مقايسه اعداد علامت دار اينستراكشن   اما در مورد ضرب ،

imul   and   idiv 

  
  :است  و حالت هاي مختلف آنها به صورت زير

imul src 
imul src, immed 
imul dest,src, 8-bit immed 
imul dest,src 
idiv src 

http://mohandesyar.com


  قسمت اول- آشنايي با برنامه نويسي اسمبلي ويندوز

 ٣٢

 ja براي مثـال  .   همچنين پرش هاي علامت دار و بدون علامت نيز وجود دارند كه ليست آنها در جدول مربوطه ارائه شد            
  :يك پرش بدون علامت است 

cmp ax, bx 
ja somewhere 

  
  :مت مي باشد پرشي با علاjg   اما 

cmp ax, bx 
jg somewhere 

  
  
  

   :Opcodesآشنايي با تعدادي ديگر از 
  

TEST 

  
   Test        بر روي دو عملگر مبدا و مقصد عمل and  داده و رجيستر انجام راflag نتيجـه بـه   . كنـد   را بر مبناي نتيجه تنظيم مـي

  .گردد تنهايي ذخيره نمي
  

  :    مثال
test  eax, 100b ; (b-suffix stands for binary) 
jnz   bitset 

  
  . تنظيم شده باشدeax  در اينجا پرش صورت خواهد گرفت اگر سومين بيت از سمت راست در 

   بررسي اين مورد است كه آيا يك رجيستر صفر است يا نه؟ test  معمولترين استفاده از 
Test  ecx, ecx 
jz  somewhere 

  . صفر باشدecxت اگر   در اين مثال پرش صورت خواهد گرف

  

 

STACK OPCODES 

  
مكـاني اسـت در حافظـه كـه     پـشته  .هاي مربوطه بايد در مورد پشته بيشتر توضـيح داده شـود   آشنايي با اينستراكشناز    قبل  

 pop و pushدو اينستراكـشن  .شـود   از آن براي نگهداري مقادير موقتي استفاده مـي     شده و  به آن اشاره     espتوسط رجيستر   

 ٣٣

 در ابتـدا خـارج    گيـرد ،  مقداري كه آخر از همه در پشته قرار مـي .روند جهت قرار دادن و دريافت مقادير از پشته بكار مي  
  : به مثال زير دقت بفرماييد.خواهد شد

(1) mov  ecx, 100 
(2) mov  eax, 200 
(3) push ecx ; save ecx 
(4) push eax 
(5) xor   ecx, eax 
(6) add  ecx, 400 
(7) mov edx, ecx 
(8) pop  ebx 
(9) pop  ecx 

  :   توضيح
 .گيرد قرار مي  ecx در ۱۰۰ .۱

 .گيرد قرار مي  eax در ۲۰۰ .۲

۳. ecxاولين قرارگيري(شود   بر روي پشته قرار داده مي.( 

۴. eaxگيرد  بر روي پشته قرار مي. 

 .كند عملياتي صورت گرفته و مقدار آن تغيير مي  ecx بر روي ۷ و ۶ و ۵در مراحل  .۵

آخـرين مقـدار قـرار داده شـده ،     . (شـود   مي۲۰۰شود و اكنون مقدار آن   دوباره از پشته دريافت مي ebxدر اينجا    ۸
  )اولين مقدار دريافت شده

 آخـرين  مقدار قرار داده شده ، اولين. (شود  مي۱۰۰  دوباره  از پشته دريافت شده و اكنون مقدار آن        eaxدر اينجا    ۹
  )مقدار دريافت شده

  
  :به جداول زير دقت بفرماييد، اي توضيح بيشتر در مورد اتفاقات رخ داده شده در حافظه  بر

Offset 1203 1204 1205 1206 1207 1208 1209 120A 120B 
Value 00 00 00 00 00 00 00 00 00 
          ESP         

(the stack is here initially filled with zeroes, but in reality it is not like this. ESP indicates the offset that ESP 
points to) 

mov ax, 4560h 
push ax 

Offset 1203 1204 1205 1206 1207 1208 1209 120A 120B 
Value 00 00 60 45 00 00 00 00 00 
      ESP            

mov cx, FFFFh 
push cx 

Offset 1203 1204 1205 1206 1207 1208 1209 120A 120B 
Value FF FF 60 45 00 00 00 00 00 
  ESP                

 ٣٤

pop edx 

Offset 1203 1204 1205 1206 1207 1208 1209 120A 120B 
Value FF FF 60 45 00 00 00 00 00 
          ESP      

  
  . خواهد شد4560FFFFh مساوي  edxدر اينجا    
  

هاي در حال اجرا بـه اشـتراك    منابع محدود بوده و بين انواع مختلف پروسه    ،    x86ردازشگر   مهيا در يك پ    ي    رجيسترها
  . بنابراين استفاده صحيح از آنها براي نوشتن كدي قابل اطمينان بسيار اساسي است.شوند گذاشته مي

  : رجيستر عمومي است۸ داراي x86   يك پردازشگر 
EAX EBX ECX EDX ESI EDI ESP EBP 

  
بنـابراين بـه   .گيرنـد  منحصرا براي شروع و خاتمه يك روال مـورد اسـتفاده قـرار مـي        EBP و   ESP اين بين دو رجيستر         در

  . رجيستر عمومي جهت برنامه نويسان مهيا است۶تنها ، صورت مؤثر 
  

 تقـسيم   رجيـستر بـه دو گـره   ۶ ايـن  . رجيستر روشـي خـاص موجـود اسـت    ۶ بيتي در مورد كار با اين   ۳۲  در ويندوزهاي   
  .اما گروه دوم بايد مقاديرشان حفظ شوند. گروه اول را آزادانه مي توان مورد استفاده و تغيير قرار داد . شوند مي

  :  رجيسترهاي زير بايد مقاديرشان حفظ شوند
EBX ESI and EDI 

  
  :  رجيسترهاي زير را آزادانه مي توان مورد استفاده قرار داد

EAX ECX and EDX 

  
  

  :مثال 
TestProc proc var1:DWORD, var2:DWORD 
   push ebx 
   push esi 
   push edi 
   ; ------------------------------------- 
   ; write code that uses EBX ESI and EDI 
   ; ------------------------------------- 
   pop edi 
   pop esi 
   pop ebx 
   ret 
TestProc endp 

 را مـورد اسـتفاده قـرار داده و تغييـر     EBX ESI and EDIرجيـسترهاي  ،  در اين تابع فرض شده اسـت كـه قـسمتي از كـد      
بنابراين نياز است تا با استفاده از اينستراكشن هاي مربوط به پشته ، مقـادير آنهـا را پـيش از تغييـر ذخيـره و سـپس            .دهد  مي

  .پس از تغيير به سيستم عامل بازگشت داد
امـا بـديهي اسـت كـه ايـن      . شـوند   ويندوز ، رجيسترهاي ذكر شده كاملا مراقبت و حفظ مي        APIهنگام استفاده از توابع        

 نيـاز بـه   APIبنابراين اگر پس از فراخـواني يـك تـابع    . توانند آزادانه ساير رجيسترهاي را استفاده و تغيير دهند  مينيز  توابع  
مقـادير رجيـسترها را   ، پـشته  هـاي   اين مورد دقت كرده و با استفاده از اينستراكـشن به استفاده از ساير رجيسترها است بايد   

  .حفظ نمود
 EBX ESI and EDIكند بهتـر اسـت از رجيـسترهاي      درون يك حلقه استفاده ميAPI   براي بهينه سازي كدي كه از توابع 

 حفـظ شـده و سـربار تحميلـي ناشـي از      APIع زيرا اين رجيسترها در فراخواني هاي تواب  . جهت ساخت حلقه استفاده كنيد    
  .حفظ ساير رجيسترها را در پشته ندارند

  
  

  :هايي جهت ساده سازي حفظ رجيسترها دستورالعمل
  

اي از كد را كـه مخلـوطي از دسـتورات     توان تمام رجيسترهاي قطعه مي  popad و pushadبا استفاده از دو اينستراكشن         
  : براي مثال.است ، حفظ و بازيابي نمود ويندوز APIاسمبلي و توابع 

  
; assembler code 
pushad 
; Write the code you wish to use to display the data you require 
popad 
; more assembler code 

  
 و PUSHFD، مـي تـوان از    هاي مربوط بـه آنهـا مـورد نيـاز اسـت      flagشود و حفظ  هاي شرطي استفاده مي    و اگر از پرش   

POPFDكنند هاي پردازشگر را حفظ مي ها حالت پرچم اين دستورالعمل. استفاده كرد.  
  

  :تر به پشته نگاهي دقيق
  

درون يـك روال و   (تواند مورد اسـتفاده قـرار گيـرد    ها مي اي از حافظه است كه براي ذخيره سازي موقتي داده      پشته بازه 
قابـل دسـتيابي     pop و  pushدر كد بوسـيله دسـتورالعمل هـاي     پشته . )رواليا روشي متداول براي ارسال پارامترها به يك     

  .گردد  در ابتدا از آن بازيابي مي گيرد ، مقداري كه در آخر بر روي پشته قرار مي.است

 ٣٦

ن شـد   popدهـد و هنگـام    را كاهش مي)  ESP(گر پشته  گيرد ، پردازشگر اشاره     اي بر روي پشته قرار مي          هنگاميكه داده 
  .يابد گر افزايش مي اين اشارهاي  داده

  
در سمت چپ هر عكـس قـرار     ،بالاي پشته . هر مربع بيانگر يك بايت است.پشته در زير ارائه شده استيك      شماتيك  
  . كه بر روي پشته قرار گرفته اندهاي مختلفي است  رنگ هاي مختلف در جهت مشخص كردن داده.گرفته است

  
Existing stack layout 

 
  

Pushes 
Push a 32 bit value = push eax 

  
 

Push a 16 bit value = push cx 

  
 

Push another 16 bit value = push dx 

  
  

 
Pops 

Pop a 16 bit value = pop dx 

  
 

Pop second 16 bit value = pop cx 

  
 

Pop the 32 bit value = pop eax 

  
  

  :   براي مثال كدي در اين باره
mov  edx, 100 
mov  ecx, 500 
; --------------------------------- 
; push the 2 values onto the stack 
; --------------------------------- 
push  edx ; edx has the value 100 
push  ecx ; ecx has the value 500 
; ------------------------------- 
; pop the 2 values off the stack 
; ------------------------------- 
pop  eax ; eax has the value 500 
pop  ecx ; ecx has the value 100 

 EAX,ECX,EDX بيتي بدين شـرح اسـت كـه رجيـسترهاي     ۳۲روال متداول ويندوزهاي    همانطور كه پيشتر نيز گفته شد ،            
در اغلـب  .  درصـورت تغييـر درون كـد بايـد حفـظ شـوند      EBX,ESI,EDIرجيـسترهاي  اما مقادير . آزادانه قابل تغيير هستند 

زيرا به صـورت خودكـار در ابتـدا و انتهـاي روالهـا      (گيرند  ستقيما مورد استفاده قرار نمي  م ESP و   EBPحالتها رجيسترهاي   
  .)شوند استفاده مي

  
  

  :متوازن كردن پشته
  
در غيـر  . شده وجـود داشـته باشـد   pop شده و تعداد بايت  push لحاظ تعداد بايت     هنگام استفاده از پشته بايد تقارني از            

اينصورت هنگام پايان كد يك روال ، ادامه كد از آدرسي نادرسـت آغـاز شـده و تقريبـا انحـصارا سـبب از كـار افتـادن                     
  .شود برنامه مي

  
  

  :    نكاتي در مورد استفاده از پشته
  
 در اينجـا چنـد نكتـه    . توان در آن ذخيـره و يـا بازيـابي كـرد بـسيار انعطـاف پـذير اسـت        هايي كه مي پشته در مورد داده     

 بيتـي را بـر روي پـشته    ۳۲ براي مثال مي تـوان يـك مقـدار    :هنگام كار با پشته بسيار مفيد هستند    د كه   نكوچك وجود دار  
  . بيتي را بازيابي نمود۱۶قرار داد و سپس دو مقدار 

; ----------------------------------- 
; push a 32 bit value onto the stack 
; ----------------------------------- 
push edx 
; ------------------------------------ 
; pop two 16 bit values off the stack 
; ------------------------------------ 
pop ax 
pop cx 

  
  . بايت بازيابي شده در آخر پشته متوازن شده است۴ روي استك قرار گرفته و  بايت بر۴   با توجه به اينكه 

  
براي مثال مي توان مقداري را بر روي پشته قـرار داد و سـپس زمـاني كـه      .شود     از پشته براي كارهاي بسياري استفاده مي      

   .حافظه براي انجام اينكار نيستبه اين صورت نيازي به تخصيص دادن يك متغير در .به آن نياز است آنرا بازيابي كرد

  :براي مثال 
  :كپي داده ها بين يك يك رجيستر و يك عملوند حافظه   

push ecx 
pop memVar 
 
or 
 
push memVar 
pop edx 
 

  :   و يا بين دو متغير حافظه
 
push memVar1 
pop memVar2 
 

  :   بجاي استفاده از رجيستر براي انجام اعمال فوق
mov edx, memvar1 
mov memvar2, edx 

  
  .اي كاهش داد    به اين صورت تعداد رجيسترهاي مورد استفاده را مي توان به صورت بهينه

  
  
  

CALL & RET 

  
   call    به قطعه اي از كد پرش كرده و سپس به محض رسيدن به اينستراكشن ret و روند عادي كـار ادامـه   گرديده متوقف 

  : براي مثال.دل توابع و رويه ها در ساير زبانهاي برنامه نويسي در نظر گرفتتوان معا  آنها را مي.يابد مي

..code.. 
call   0455659 
..more code.. 

Code at 455659: 
add   eax, 500 
mul   eax, edx 
ret 

 .يابـد  امـه مـي   ادret پرش كرده و تـا  0455659رسد به كد قرار گرفته در آدرس   ميcall   هنگاميكه كد به سطر مربوط به       
  .شود  اجرا ميcallسپس روال عادي كار ادامه يافته و اينستراكشن پس از 

بـر روي پـشته قـرار داده    ) گـري بـه اينستراكـشن بعـدي كـه بايـد اجـرا شـود         اشاره (EIP ، رجيستر   call  در ابتداي اجراي    
   .گردد  شدن اين رجيستر ميpop سبب retاينستراكشن در آخر  .شود مي

  : قابل دريافت هستندcall هاي قبل از pushانهاي يك تابع نيز با استفاده از   آرگوم
Push  something 
Push  something2 
Call  procedure 

  .شوند  ، آرگومانها از پشته خوانده شده و استفاده ميcall   درون يك 
  

  : نكته
البته بديهي است كـه  ( قرار مي گيرد  eaxر خروجي يك تابع د  ،در ويندوز و توابع ويندوز به صورت استاندارد      

از هر رجيستر مناسبي براي اين امر مي توان استفاده كـرد امـا ايـن يـك اسـتاندارد پذيرفتـه شـده تحـت وينـدوز              
  .)است

  
  
  

   : masm با استفاده از asm32برنامه نويسي 
  

 برنامه نويـسي اسـمبلي   masm استفاده از ر شد، همانگونه كه پيش تر نيز ذك    اما،   است   مطالعه اين قسمت كاملا اختياري      
  .كند تر مي ويندوز را به شدت ساده

  
  :ساختارهاي تصميم گيري و حلقه ها

  
   Masmها و يا ساختارهاي تصميم گيري است  حاوي ماكروهايي جهت سهولت تعريف حلقه:  

.IF, .ELSE, .ELSEIF, .ENDIF 

.REPEAT, .UNTIL 

.WHILE, .ENDW, .BREAK 

.CONTINUE 
.If  
، استفاده از سـاختارهاي تـصميم گيـري بـه شـدت سـاده تـر        ) نقطه قبل از آنرا فراموش نكنيد (If.با استفاده از ماكروي        

  .باشد ها نمي شده  و ديگر نيازي به درگير شدن با جزئيات انواع پرش
  :مثال يك   

.IF eax==1 
;eax is one 
.ELSEIF eax=3 
; eax is three 
.ELSE 
; eax is not one or three 
.ENDIF 

  ): تو در تو نيز مجاز استif(   مثال دو 
.IF eax==1 
     .IF ecx!=2 
          ; eax= 1 and ecx is not 2 
     .ENDIF 
.ENDIF 

  :توان عملي نمود   البته مثال فوق را به شكل ساده تر زير نيز مي
.IF (eax==1 && ecx!=2) 
; eax = 1 and ecx is not 2 
.ENDIF 

  :باشند    عملگرهاي زير در اينجا مجاز مي
== is equal to 
!= is not equal to 
> is greater than 
< is less than 
>= is greater than or equal to 
<= is less than to equal to 
& bit-test 
! logical NOT 
&& logical AND 
|| logical OR 
CARRY? carry bit set 
OVERFLOW? overflow bit set 
PARITY parity bit set 
SIGN? sign bit set 
ZERO? zero bit set 

  
Repeat  

  :مثال. كند اي از كد را تا زمانيكه شرط آن معتبر باشد اجرا و تكرار مي    اين عبارت قطعه
.REPEAT 
; code here 
.UNTIL eax==1 

  .كند  را اجرا و تكرار ميuntil و repeat مساوي يك باشد ، قطعه كد قرار گرفته بين eax  كد فوق ، تا زمانيكه 
  

While  
     :مثال. كند اين عبارت قطعه كد را تا هنگاميكه شرط آن معتبر باشد اجرا و تكرار مي   

.WHILE eax==1 
; code here 
.ENDW 

  : مثال .ده كرد استفاbreak.   براي خارج شدن از حلقه مي توان از عبارت 
.WHILE edx==1 
    inc eax 
    .IF eax==7 
        .BREAK 
    .ENDIF 
.ENDW 

  . شود حلقه خاتمه خواهد يافت۷ مساوي eax   در اينجا اگر 
  

اسـتفاده   نيز درون حلقه براي صرفنظر كردن از اجراي كـدهاي بعـد از آن مـي تـوان     continue از عبارت C   همانند زبان  
  .كرد

  
  :به صورت خلاصه    
  

IF - ELSE IF - ELSE 

The C version:  
 
if ( var1 == var2 ) 
{ 
    // Code goes here 
} 
else 
if ( var1 == var3 ) 
{ 
    // Code goes here 
} 
else 
{ 
    // Code goes here 
} 

The MASM version:  
 
.if ( var1 == var2 ) 
    ; Code goes here 
 
 
.elseif ( var1 == var3 ) 
    ; Code goes here 
 
.else 
    ; Code goes here 
 
.endif 

 

DO - WHILE 

The C version:  
 
do 
{ 
    // Code goes here 
} 
while ( var1 == var2 ); 

The MASM version:  
 
.repeat 
    ; Code goes here 
 
.until ( var1 != var2 ) 

 

WHILE 

The C version:  
 
while ( var1 == var2 ) 
{ 
    // Code goes here 
} 

The MASM version:  
 
.while ( var1 == var2 ) 
    ; Code goes here 
 
.endw 

Invoke 

 جهـت اسـتفاده   invoke بكـارگيري  ، nasm و tasmهاي مـشابه آن ماننـد     نسبت به نمونهmasmيكي از مهمترين مزاياي      
  .و توابع مي باشدها  از رويهساده تر 

  :   حالت معمول استفاده از توابع به شكل زير است
push  parameter3 
push  parameter2 
push  parameter1 
call  procedure 

  : داريمinvoke   با استفاده از 
invoke   procedure, parameter1, parameter2, parameter3 

  
كـد را  ،  invokeبـا حالـت معمـول هـيچ تفـاوتي نـدارد امـا اسـتفاده از         فوق ده  ش   لازم به ذكر است كه كد نهايي اسمبل         

  .نمايد تر مي خواناتر و ساده
     

  : براي مثال. تعريف كردprototypeها بايد   در مورد رويه invoke   براي استفاده از
PROTO STDCALL testproc: DWORD, :DWORD, :DWORD 

  
  :اكنون .كند  تعريف ميDWORD با سه آرگومان با اندازه testproc اي را به نام    عبارت فوق رويه

Invoke  testproc, 1, 2, 3, 4 

  
.  خطايي را در مورد بكارگيري يـك آرگومـان اضـافي بـه شـما گوشـزد خواهـد كـرد        masmپس از اجراي كد فوق ،      

  .وندش همچنين نوع و اندازه پارامترهاي ورودي نيز در اين حالت دقيقا بررسي مي
  

  . استفاده كردaddr از offsetمي توان بجاي  invoke   در عبارت 
  :شوند ها به شكل زير تعريف مي    رويه

  
testproc PROTO STDCALL :DWORD, :DWORD, :DWORD 
.code 
testproc proc param1:DWORD, param2:DWORD, param3:DWORD 
ret 
testproc endp 

  
  

جهـت اسـتفاده در عبـارت     ذكـر شـده   prototype .شـود   با سه آرگومان ورودي تعريف مي   testproc   در اين مثال  رويه      
invokeتعريف شده است .  

testproc  PROTO STDCALL :DWORD, :DWORD, :DWORD 
.code 
testproc proc param1:DWORD, param2:DWORD, param3:DWORD 
   mov ecx, param1 
   mov edx, param2 
   mov eax, param3 
   add edx, eax 
   mul eax, ecx 
 
   ret 
testproc endp 

  :   عملكرد مثال فوق به صورت زير است
testproc(param1, param2, param3) = 

param1 * (param2 + param3) 
  

  .قرار گرفته است  EAX  و نتيجه مطابق استاندارد ذكر شده در 
  

  :ن به صورت زير عمل كردتوا    براي تعريف متغيرهاي محلي مي
  

testproc proc param1:DWORD, param2:DWORD, param3:DWORD 
   LOCAL var1:DWORD 
   LOCAL var2:BYTE 
   mov ecx, param1 
   mov var2, cl 
   mov edx, param2 
   mov eax, param3 
   mov var1, eax 
   add edx, eax 
   mul eax, ecx 
   mov ebx, var1 
   .IF bl==var2 
        xor eax, eax 
   .ENDIF 
 
   ret 
testproc endp 

  
  

توان استفاده كرد زيرا در ابتـداي كـار بـر روي پـشته قـرار گرفتـه و در آخـر            از اين متغيرهاي محلي خارج از رويه نمي     
  .شوند  از پشته حذف ميretهنگام 

  اسمبلي مقدماتي ويندوز
  

  .توان آموخت اكنون طرز استفاده از اسمبلي در ويندوز را مي، ت    پس از ذكر مقدما
  

API  
و ) Application programming interface ( اسـتوار شـده اسـت   windows API  اساس برنامه نويـسي وينـدوز بـر مبنـاي       

ين همـانطور كـه پيـشتر نيـز     بنابرا (دهد اي از توابع مهيا شده توسط سيستم عامل را در اختيار برنامه نويس قرار مي      مجموعه
هـاي تحـت وينـدوز از ايـن توابـع اسـتفاده        تمام برنامه.) ويندوز جايگزين شده اندAPIهاي داس با    ذكر شد، در اينجا وقفه    

دو ، در حالت كلي براي تعدادي از توابـع  .اند  و غيره قرار گرفته   kernel   ، gdi هاي سيستمي مانند     dllاين توابع در    .كنند مي
 ، هـر  Ansiدر حالـت  .شـود  هـا مربـوط مـي    ي رفتار و ذخيره سازي رشـته   و به نحوهگرديده و يونيكد تعريف    Ansiنگارش  

درجهـت مـشخص كـردن انتهـاي رشـته      byte-0 از (بوده و مختوم به نال مـي باشـد   ) كد اسكي(بايت بيانگر يك سمبول 
در ايـن  .گيـرد  كند، كمك مـي  استفاده مي بايت ۲از كه به ازاي هر سيمبول     widecharيونيكد از قالب    . .)استفاده مي شود  

  حالـت در ايـن .گـردد   فـراهم مـي  ،حالت امكان تعريف زبانهايي كه نياز به كاراكترهاي بيشتري مانند زبـان چينـي هـستند           
وينـدوز بـراي مـشخص نمـودن ايـن مـوارد در توابـع بـه شـكل زيـر عمـل            .شـود  انتهاي رشته با دو بايت صفر مشخص مي      

  :كند مي
MessageBoxA (A-suffix for ansi) 
MessageBoxW (W-suffix for widechar (unicode)) 

  
   ويندوزAPI ها و توابع dllبكار گيري 

  
فايلهـاي   (import librariesايـن كـار توسـط    . ها در برنامه اسـت dll ويندوز نياز به فراخواني APIجهت بكارگيري توابع    
lib ( جهت بارگذاري پويـاي فايلهـاي       ها در    اين كتابخانه .ميسر استdll     ضـروري هـستند )      بـراي مثـال بارگـذاري پويـا در

، فايلهـاي    قابـل دريافـت اسـت   win32asm.cjb.net كـه در آدرس      win32asm در بـسته     .)آدرس پايه اي پويـا در حافظـه       
  .اند  هاي سيستمي ارائه شدهdllكتابخانه اي اكثر 

  :براي مثال.  استفاده كردincludelibتوان از عبارت  مياي     براي بارگذاري يك فايل كتابخانه
includelib  C:\masm32\lib\kernel32.lib 

  
  :شود  استفاده مي)مسيردهي نسبي (   در اغلب مثالها از روش زير

includelib  \masm32\lib\kernel32.lib 

  
ايـن  .باشـد   نيـز مـي  include file و يا incنياز به فايلهاي   واي به برنامه به تنهايي كافي نيست البته الحاق يك فايل كتابخانه   

  .قابل توليد هستنداي  از فايلهاي كتابخانه به صورت خودكار l2incنوع فايلها با استفاده از برنامه كمكي 

  :شود  به برنامه به صورت زير عمل ميinc   براي الحاق كردن فايلهاي 
Include  \masm32\include\kernel32.inc 

  
موجـود    invokeجهت بكار گيري آنها توسط عبارت  توابع تعريف شده در يك كتابخانه    inc   ، prototypes   در فايلهاي   

  :براي مثال .هستند
kernel32.inc: 
... 
MessageBoxA proto stdcall :DWORD, :DWORD, :DWORD, :DWORD 
MessageBox textequ <MessageBoxA> 
... 

  
  . ي اصلي يكي هستند اند و با تعريف نمونه  در انتها مشخص نشدهA با حرف Ansiا توابعي از نوع   در اين فايله   
  :براي مثال. توان از توابع موجود در كتابخانه استفاده كرد  ميinc   پس از الحاق فايلهاي كتابخانه اي و فايلهاي  

Invoke  MessageBox, NULL, ADDR MsgText, ADDR MsgTitle, NULL 

  
  
  

   :windows.incمعرفي 
  

 وينـدوز  API وجـود دارد كـه حـاوي تمـام سـاختارها و ثوابـت تعريـف شـده در           windows.inc ويژه اي به نـام        inc   فايل  
  : به صورت زير در آن تعريف شده اندMessageBoxبراي مثال ثوابت مربوط به حالتهاي مختلف يك .است

  
MB_OK equ 0 
MB_OKCANCEL equ ... 
MB_YESNO equ ... 

  :توان استفاده كرد ان ثوابت مي  بنابراين از اين اسامي بعنو
invoke MessageBox, NULL, ADDR MsgText, ADDR MsgTitle, MB_YESNO 

  
  

  :  براي استفاده از آن
include \masm32\include\windows.inc 

   در ويندوز)asm.* (ساختار كلي يك فايل منبع اسمبلي
  

.486 

.model flat, stdcall 
 
option casemap:none 
 
includelib \masm32\lib\kernel32.lib 
includelib \masm32\lib\user32.lib 
includelib \masm32\lib\gdi32.lib 

include \masm32\include\kernel32.inc 
include \masm32\include\user32.inc 
include \masm32\include\gdi32.inc 
include \masm32\include\windows.inc 

.data 

blahblah 

.code 

start: 

blahblah 

end start  

  :   معاني قسمت هاي مختلف ذكر شده به صورت زير است
  

 386.از.توليـد كنـد  ) يـا بـالاتر   (٤٨٦ جهت پردازشگرهاي opcodesگويد كه بايد  به اسمبلر مي   486
  .حالات پاسخگو است در اغب 486.د اما  كردهنيز مي توان استفا

.model flat, stdcall  و تعريـف   )كـه پيـشتر در مـورد آن صـحبت شـده بـود       (استفاده از مدل حافظه تخـتstdcall 

calling convention  بدين معنا كه پارامترهاي تابع از راست به چپ ،pushآخـرين   (شـوند   مي
يبا حالتي استاندارد براي تمامي توابـع  اين مورد تقر. )گيرد پارامتر در ابتدا بر روي پشته قرار مي    

  .ويندوز است
option casemap:none        براي اينكه فايل .كند كنترل مي، نگاشت كاراكترها را به حروف بزرگwindows.inc  صـحيح 

  . تنظيم گرددnoneبكار گرفته شود بايد به 
includelib در بالا توضيح داده شد.  
include در بالا توضيح داده شد.  
.data پيشتر توضيح داده شد. هاي برنامه  شروع قسمت داده.  

.code  پيشتر توضيح داده شد. شروع قسمت كد برنامه. 
start: 

 
end start  

البته برچسب شروع هر نام دلخواهي مي تواند باشـد بـراي     . برچسبي براي شروع و خاتمه برنامه     
  :مثال

startofprog: 

end   startofprog  
  
  

  :ها  يي بيشتر با كتابخانهآشنا
  

هدف از ايجاد آنها كپسوله كـردن تعـدادي از ماژولهـاي     .ها بسيار شبيه به كد نويسي متداول است        روش ايجاد كتابخانه      
در قـسمت هـاي   بدون نياز به كپي كردن كل كد آنهـا   برنامه  هاي مختلف    برنامه است تا بتوان از آنها به دفعات در قسمت         

  .اده كردتكراري ، استف
 اسمبلر از فايل كتابخانـه  linkerكنيد ، تمام كد مورد نياز توسط        هنگاميكه شما از كتابخانه اي در برنامه خود استفاده مي         

  .باشد  ها نمي  نهايي نيازي به ارائه كتابخانهexeبنابراين هنگام توزيع فايل . شود دريافت و به برنامه ضميمه مي
هايي از كتابخانه را كه در برنامه مورد استفاده قرار گرفته است به فايـل اجرايـي     تنها قسمتlinker   لازم به ذكر است كه     

  .بسيار بهينه خواهد بوداجرايي  به اين صورت حجم نهايي فايل .كند  نهايي ضميمه مي
  

  :   ساختار يك فايل كتابخانه در مرحله كد نويسي به صورت زير است
.386 ; set processor type 
.model flat, stdcall ; memory model & calling convention 
option casemap :none ; case sensitive 
; -------------------------------- 
; Include any needed INCLUDE files 
; or procedure prototypes. 
; -------------------------------- 
; include \masm32\include\windows.inc 
; include \masm32\include\gdi32.inc 
; include \masm32\include\user32.inc 
; --------------------------------------------------------- 
; Write the prototype for the procedure name below ensuring 
; that the parameter count & size match and put it in your 
; library include file. 
; --------------------------------------------------------- 
; YourModule PROTO :DWORD,:DWORD etc..... 
.code 
; ----------------------------------------------------- 
YourModule proc par1:DWORD,par2:DWORD etc..... 
ret 
YourModule endp 
; ----------------------------------------------------- 

پـس از اطمينـان حاصـل كـردن از     . كنيد كدنويسي آن با برنامه نويـسي متـداول تفـاوتي نـدارد        همانطور كه ملاحظه مي  
  .توان از آن يك فايل كتابخانه اي ساخت صحت عملكرد يك رويه به راحتي مي

  :از كد فوق مي توان به صورت زير عمل كرد lib  براي ساخت فايل 
@echo off 
\masm32\bin\ml /c /coff *.asm 
\masm32\bin\lib *.obj /out:yourlib.lib 
: The following line works as well 
: \masm32\bin\link -lib "*.obj" "/out:yourlib.lib" 

  
قـرار داد    incهاي موجود در كتابخانه را در يك فايل  عاريف رويهت،  ، بايد جهت استفاده از آن     lib   پس از ساخت فايل     

  .تا سايرين بتوانند از كتابخانه استفاده نمايند
  .هاي قبل توضيح داده شدند روش فراخواني اين فايلهاي نهايي در قسمت

  
  
  
  

  !اولين برنامه
  

  . استMessageBoxمعروف به كمك يك   hello worldاز اين برنامه نمايش    هدف 
  

.486 
 
.model   flat, stdcall 
 
option    casemap:none 
 
includelib   \masm32\lib\kernel32.lib 
includelib   \masm32\lib\user32.lib 
 
include     \masm32\include\kernel32.inc 
include     \masm32\include\user32.inc 
include     \masm32\include\windows.inc 
 
.data 
MsgText   db   "Hello world!",0 
MsgTitle   db   "This is a messagebox",0 
 
.code 
start: 
    invoke MessageBox, NULL, ADDR MsgText, ADDR MsgTitle, MB_OK or  MB_ICONINFORMATION 
    invoke ExitProcess, NULL 
end start 

  : كد فوقري در مورد بيشتاتتوضيح

  :استفاده شده است)  MSDNمطابق ( به شرح زير API اين مثال از دو تابع    در

int MessageBox( 
HWND hWnd, // handle of owner window 
LPCTSTR lpText, // address of text in message box 
LPCTSTR lpCaption, // address of title of message box  
UINT uType // style of message box 
);  

Parameters 

hWnd Identifies the owner window of the message box to be created. If this 
parameter is NULL, the message box has no owner window.  

lpText Points to a null-terminated string containing the message to be 
displayed.  

lpCaption Points to a null-terminated string used for the dialog box title. If this 
parameter is NULL, the default title Error is used.  

uType 
Specifies a set of bit flags that determine the contents and behavior of 
the dialog box. This parameter can be a combination of flags from the 
following groups of flags.  

  
 ، ☺! ي والـد  اي بـه پنجـره   دسـتگيره : پـذيرد  مـي را  آرگومـان  MessageBox  ،۴   همانطور كه ملاحظـه مـي فرماييـد تـابع        

   .MessageBox نوع  در آخر وMessageBoxگري به رشته عنوان  اشارهگري به رشته پيغامي كه بايد نمايش يابد ،  اشاره
  
. گـري اسـت بـه رشـته مـا      اشارهlpText . قرار دادNull را مساوي hWndاي ندارد مي توان  با توجه به اينكه برنامه ما پنجره  

كه متن مورد نظر در آن قرار گرفتـه اسـت را   ، آفست مكاني از حافظه ، ذكر شده اين مورد بدين معنا است كه آرگومان    
  .نياز دارد

  
  :ها به صورت زير است  حوه معرفي رشته   اگر به كد دقت بفرماييد ن

.data 
MsgText   db  "Hello world!",0 
MsgTitle   db  "This is a messagebox",0 

   .data كه با (بايت ها است از  يك رشته مجموعه اي .ها است  به معناي شروع قسمت دادهdb مختـوم بـه   )   مشخص شـده
  .) استگرديدهبايت صفر كه مشخص (نال 

  
  . خواهد كردcrash آن كليك شود برنامه Ok روي دكمه  MessageBoxاگر از تابع زير استفاده نشود و پس از نمايش    

VOID ExitProcess( 

UINT uExitCode // exit code for all threads  
); 

  . براي خاتمه كار خود استفاده مي كنندExitProcess   برنامه ها در ويندوز از 
  

  :)شكل زير( و اجر است build قابل masm32بسادگي در    اين مثال 
  

  
  

   . asm32 و اجراي اولين برنامه build –۱۱شكل 
  
  ) و نيم كيلوبايت۲اي به حجم كمتر از  برنامه (!و پس از اجراي برنامه خواهيم داشت   
  

  
   نمايي از اجراي اولين برنامه- ۱۲ شكل

  
 را exe را دانلـود كـرده و سـپس فايـل     OllyDbg را مشاهده كنيد، ابتدا برنامـه  exe  فايلdisassembly   اگر علاقمند باشيد   

و يـا  (قابل دريافت اسـت    /http://home.t-online.de/home/Ollydbg  خانگي آن در    اين برنامه از آدرس    .در آن باز كنيد   
  ).اولين لينك پس از جستجو در گوگل

  

http://home.t-online.de/home/Ollydbg

 بـر روي پـشته و   MessageBox، نحـوه قرارگيـري پارامترهـاي تـابع          شـده ه  ي ترجمه نهايي كد نوشـت         در شكل زير نحوه   
  .توان مشاهده كرد و غيره را ميدستورات  معادل Opcodesاطلاعاتي ديگر مانند 

  :شود ذكر گرديد بيشتر مشخص مينيز در اينجا مفهوم جمله زير كه در ابتداي كار     
  ".شود ا به اعدادي كه توسط پردازشگر قابل اجرا است ترجمه مي بوسيله برنامه اسمبلر مستقيماسمبليهر دستور "

  

  
   . OllyDbg نمايش ديس اسمبلي اولين برنامه نوشته شده در –۱۳شكل 

  
  
  
  

  :ها در اسمبلي ويندوز كار با رشته
  
شـته هـا در   البتـه انـواع ديگـري از ر    . هاي رشته اي تقريبا بـه صـورت انحـصاري مختـوم بـه نـال هـستند                  در ويندوز داده     

 zero(هـاي مختـوم بـه نـال       بـا رشـته  APIاما اغلب توابـع  . هاي يونيكد  مانند رشته،  بيتي نيز موجود هستند ۳۲ويندوزهاي 

terminated strings (ها   لازم به ذكر است كه كار با رشته ها بر روي توالي بايت.كنند كار مي)  بجاي كار بـا كاراكترهـا (
  .نيز قابل انجام است

 لحـاظ مـصرف حافظـه بـسيار     از  .امـا اسـاس بـسيار سـاده اي دارد      هاي مختوم به نال بسيار قديمي است         كنولوژي رشته    ت
  .مقرون به صرفه بوده و پردازش آنها نيز بسيار سريع است

  :)به شكل آرايه اي از بايت هاي مختوم به نال ( به صورت زير استdata.   روش تعريف يك رشته در قسمت 
MyString  db  "This is a string",0 

 و ESI ، EDI از رجيـسترهاي   )REP/REPE/REPNE بـه همـراه   (هـا  رشتههاي كار با      اينستراكشن   بيتي ،  ۳۲   در ويندوزهاي   
ECX     اين دستورالعمل ها كار خـود را تازمانيكـه شـرط مـورد نظـر بـرآورده شـود         .كنند   بعنوان شمارشگر حلقه استفاده مي

  .دنماين تكرار مي
  

  :    مثال
 بـراي بـافر مقـصد    ( destهمچنين فرض شده است كه انـدازه بـافر    . آدرس بافر منبع جهت كپي است        ، source   در اينجا   

  . تعداد بايتي است كه بايد كپي شودln.باشد ميبزرگ به اندازه كافي ) كپي كردن بايت هاي منبع
 
cld ; set direction flag forward 
mov esi, source ; put address into the source index 
mov edi, dest ; put address into the destination index 
mov ecx, ln ; put the number of bytes to copy in ecx 
; -------------------------------------------------- 
; repeat copying bytes from ESI to EDI until ecx = 0 
; -------------------------------------------------- 
rep movsb 

  
 rep زمـاني كـار   .كاهد مي   ECX كپي كرده و يك واحد از EDI به  ESI ، هر بايت را از       movsb   در اين مثال اينستراكشن     

  . مساوي صفر شودecxخاتمه مي يابد كه 
 .بررسـي كـرد  را ) صـفر (كد اسـكي مربوطـه   ،  جهت اتمام كار توان شود مي    هنگاميكه يك رشته مختوم به نال كپي مي     

  :براي مثال
cld ; clear direction flag to read forward 
mov esi, source ; put address into the source index 
mov edi, dest ; put address into the destination index 
label: 
   lodsb ; load byte from source into AL and inc ESI 
   stosb ; write AL to dest and inc EDI 
   cmp al, 0 ; see if its an ascii zero 
   jne label ; read the next byte if its not 

  
بـر روي پنتيـوم و   .  بـه مقـصد  AL انتقـال داد و سـپس از   ALبهتـر اسـت هـر بايـت را از منبـع بـه            براي بـالا بـردن سـرعت        

 و عمـل كـردن   dereferencingبـدليل اسـتفاده از    ( سـريعتر هـستند  LODSB/STOSBاز   MOV/INCهاي جديدتر ،     پردازنده
  .) هاي حافظه رجيسترهاي مورد استفاده بعنوان آدرس

 
  mov esi, source ; put address into the source index 
  mov edi, dest ; put address into the destination index 
label: 
  mov al, [esi] ; copy byte at address in esi to al 
  inc esi ; increment address in esi 
  mov [edi], al ; copy byte in al to address in edi 
  inc edi ; increment address in edi 
  cmp al, 0 ; see if its an ascii zero 
  jne label ; jump back and read next byte if not 

، به صـورت  )شود  هم گفته ميpairingكه به آن ( dual pipelinesتر است اما بر روي پردازشگرهايي با     كد فوق طولاني
  .) صادق نيستstosbتري مانند  هاي قديمي اين مورد براي اينستراكشن (شوند  برابر سريعتر اجرا مي۲اسمي 

  
  

   :Anonymous labelsمعرفي 
  

  :)تعريف برچسب بدون تعريف نامي براي آن ( معرفي شده و به صورت زير استMASMگذاري در  ن نشاناي    
@@: 

   : در اين حالت   
 jmp @F در ادامه كد تعريف شده است:@@ يعني پرش به جايي كه اولين برچسب .  

jmp @B قبل از پرش ذكر شده وجود دارد :@@ يعني پرش به جايي كه اولين برچسب .  
  

  :ها    مثالي ديگر در مورد كار با رشته
  

  .      فيلتر كردن يك رشته به صورتي كه فقط اعداد آن استخراج شوند
  .است ۵۷ مساوي "9" و عدد اسكي معادل ۴۸ مساوي "0"عدد اسكي معادل      

  
NumOnly   proc   source :DWORD,   dest:DWORD 
  mov ecx, source ; put address into ecx 
  mov edx, dest ; put address into edx 
@@: 
  mov al, [ecx] ; copy byte at address in ecx to al 
  inc ecx ; increment address in ecx 
  ; ----------------------------------------------------- 
  ; perform byte modification, replacement or omissions 
  ; ----------------------------------------------------- 
  cmp al, 0 ; test for zero first 
  je @F ; exit loop on ASCII zero 
  cmp al, "0" ; string literal 
  jb @B ; if below 48 "0", jump back 
  cmp al, "9" ; string literal 
  ja @B ; if above 57 "9", jump back 
  ; ------------------------------------------------------ 
  mov [edx], al ; copy byte in al to address in edx 
  inc edx ; increment address in edx 
  jmp @B 
@@: 
  mov [edx], al ; copy ASCII zero to address in EDI 
 
  ret 
NumOnly endp 

  :كار با ماكروها
  

 روش ايجاد يك ماكرو بـه شـكل   .در يك كد مي توان كمك گرفتماكروها جهت خلاصه كردن اعمال تكراري  از      
  :زير است

Do_This_Operation MACRO 
; write your code here 
ENDM 

  :  سپس به سادگي با نوشتن نام آن در كد مي توان از آن استفاده كرد
Do_This_Operation 

  
بـرخلاف يـك    (نمايـد  بـسط داده و در كـد جـايگزين مـي    عينـا  آنـرا   رسد، ام ماكرو در كد شما مياسمبلر به ن   هنگاميكه  

  .باشد به همين جهت حجم نهايي كد در اين حالت بيشتر مي.)رويه
     

  :براي مثال. تواند بپذيرد  همچنين يك ماكرو پارامتر نيز مي
RGB MACRO red, green, blue 
   xor eax, eax 
   mov al, blue ; blue 
   rol eax, 8 
   mov al, green ; green 
   rol eax, 8 
   mov al, red ; red 
ENDM 

  :   روش استفاده از آن به شكل زير است
RGB 125, 175, 225 
or 
RGB Byte1, Byte2, Byte3 
or 
RGB cl, ch, dl 

  
  . در آن حذف شده استسربار كار با پشتهها بيشتر است زيرا     سرعت اجراي ماكروها از رويه

  
  : محلي در ماكروهايپارامترها

  
 نيز تعريـف شـده باشـد و ايـن مـاكرو بـيش از يكبـار        برچسب، دهد كه درون تعريف يك ماكرو            مشكل زماني رخ مي   

ر نمايـد ايـن ام ـ   با توجه به اينكه اسمبلر كد ماكرو را صرفا بدون هيچ تغييري در مكان فراخواني كپـي مـي   . فراخواني شود 
  .گردد هاي تكراري مي سبب بروز خطاي برچسب

 بـراي تعريـف برچـسب در ماكروهـا بـه      DIRECTIVE (LOCAL( با بكار گيري رهنمود  MASM   خوشبختانه اين مشكل در     
   .گردد صورت خودكار حل مي

  :براي مثال    
MyMacro MACRO Parameter1, Parameter2 etc ... 
   LOCAL MyLabel 
   ; asm code 
   jmp MyLabel 
   ; asm code 
   MyLabel: 
   ; other asm code 
ENDM 

  .ها متفاوت است  در اينجا با بحث تعريف آن در رويهLOCAL  بايد دقت داشت كه رهنمود  
   MASM دهـد  در كد قرار مـي ... و  00002?، 00001?شوند در انتها به صورت  هايي را كه بدين شكل تعريف مي      برچسب 

  .ز تكراري نبودن آنها اطمينان حاصل شودتا ا
     
  
  

  :كار با ساختارها
  

 . تا بتوان با آنها همانند يك واحد رفتار كـرد )با اندازه اي مشخص  (است  در كنار هم    گروهي از داده ها     ،  يك ساختار       
  :براي مثال 

RECT STRUCT 
   left         DWORD ? 
   top         DWORD ? 
   right       DWORD ? 
   bottom   DWORD ? 
RECT ENDS 

  . در اينجا به معناي عدم مقدار دهي اوليه عضو تعريف شده است?   علامت 
  .گيرند  به صورت متوالي در حافظه قرار ميDWORD مقدار ۴   اعضاي اين ساختار به صورت 

  :شود درون رويه ها به شكل زير عمل مي اما  مي توان مقدار اوليه نيز براي اعضاي آن در نظر گرفتdata.   در قسمت 
  

LOCAL Rct :RECT 
; code 
mov Rct.left,       1 
mov Rct.top,       2 
mov Rct.right,     3 
mov Rct.bottom, 4 

  
  

ي ديگـر بـه آن    بنابراين انتقال يك عملونـد حافظـه  .باشند كه اعضاي ساختار ، يك عملوند حافظه مي         لازم به ذكر است   
  . در اين حالت يا بايد از پشته كمك گرفت و يا از رجيسترها.به صورت مستقيم ميسر نيست

  :   براي استفاده از ساختارها در توابع ويندوز مي توان از روش زير استفاده كرد
  .   ابتدا مي توان اعضاي ساختار را با مقادير لازم مقدار دهي كرد و سپس آدرس آنرا به تابع انتقال داد

Invoke   APIcall,parameter1,parameter2,ADDR Rct 

  
  :   و اگر رويه اي نوشته ايد كه نياز به دريافت مقادير ساختار بجاي آدرس آن دارد مي توان به شكل زير عمل كرد

MyProc proc par1:DWORD,par2:DWORD,MyRect:RECT 
 
mov  eax, MyRect.left ; copy first member into EAX 

  
  :ارهاي تو در توساخت

  
 ويندوز مرسوم است استفاده از ساختارهاي تـو در تـو مـي باشـد كـه روش       APIيكي از مواردي كه در تعدادي از توابع      

  : به شكل زير استMASMتعريف آنها در 
MyNestedStruct STRUCT 
   item1 RECT <> 
   item2 POINT <> 
MyNestedStruct ENDS 

  
  :كند  كه پيشتر تعريف شد و ساختار زير استفاده ميRECTساختار اين ساختار از دو    

POINT STRUCT 
   x DWORD ? 
   y DWORD ? 
POINT ENDS 

  :   براي استفاده از آن
LOCAL mns:MyNestedStruct 
 
mns.item1.left 
mns.item1.top 
mns.item1.right 
mns.item1.bottom 
mns.item2.x 
mns.item2.y 

  
  :هاشرفته ساختاربكارگيري پي

  
 بـراي  MASM. ويندوز مرسوم است كه ساختارها بعنوان آدرس به آنها انتقال داده شـوند      API   در طراحي تعدادي از توابع      

  .راه حل ساده تري را ارائه داده است، سهولت انجام اينگونه امور 
  :اد ، روش متداول آن به صورت زير است را به يك رويه انتقال دRECT   براي مثال اگر نياز باشد تا آدرس ساختار 

Invoke  MyFunction,ADDR Rct 

  :   تعريف اين تابع نيز به صورت زير است
MyFunction  proc  lpRect:DWORD 

  :   روش آدرس دهي پارامترهاي مختلف ساختار به شكل زير است
  

mov eax, lpRct 
mov [eax],       DWORD PTR 10 
mov [eax+4],   DWORD PTR 12 
mov [eax+8],   DWORD PTR 14 
mov [eax+12], DWORD PTR 16 

  
هـاي پيچيـده تـر احتمـال بـروز          براي ساختاري ساده همانند مثال ذكر شده اين روش شايد زياد مشكل نباشد اما با نمونـه   

  .يابد خطا به شدت افزايش مي
  :ها تعريف شده است  دهي جهت سهولت اينگونه مقدارASSUME رهنمودي به نام MASM   در 

ASSUME  eax:PTR RECT 
 mov  eax, lpRct 
 mov  [eax].left,       10 
 mov  [eax].top,       12 
 mov  [eax].right,     14 
 mov  [eax].bottom, 16 
ASSUME  eax:nothing 

  
  . رفتار كنRECT همانند ساختار eaxشود كه با رجيستر     در اين حالت به اسمبلر گفته مي

   ASSUME eax:nothingبه اين معنا است كه كار با رجيستر به اين شكل خاص به پايان رسيده است .  
  : )type casting (   روش ديگري نيز وجود دارد

mov   eax, lpRct 
mov  (RECT PTR [eax]).left,       10 
mov  (RECT PTR [eax]).top,       12 
mov  (RECT PTR [eax]).right,     14 
mov  (RECT PTR [eax]).bottom,  16 

  
  
  

   :++VCاضافه كردن فايلهاي اسمبلر به 
     

 بـراي  . سـازگار هـستند  ++VC مربـوط بـه   linker بـا  MASM سـاخته شـده در    obj   همانطور كه پيشتر نيز گفته شد فايلهاي        
   :++VCاستفاده از آنها در 

  : براي مثال .)خواهيم استفاده كنيم ميآن  link.exeاز  (را مشخص نماييد masm32ر  فولدbin دايركتوري مسير   ابتداي 
C:\MASM32\bin  

  
  :قسمت زير مراجعه نماييد،  Toolsبه منوي  ++VC مربوط به IDEسپس در    

Tools/Options/Projects/Visual C++ Directories  

 ٥٨

  
 IDE تنظيمات - ۱۴ شكل

  
  .)مهم( و آنرا به پايين ليست انتقال دهيد    سپس دايركتوري ذكر شده را به اين مسير اضافه نماييد

  . ايجاد كنيدconsole  در ادامه يك برنامه 
  . را به پروژه اضافه كنيدtest.asmيدي به نام    سپس مطابق شكل زير فايل جد

  

  
   اضافه كردن فايلي جديد به پروژه- ۱۵شكل 

  :)در مورد ساختار استاندارد آن پيشتر صحبت گرديد ( اضافه نماييدtest.asm   كد ساده زير را به فايل 
  

.486 

.model flat, stdcall 
option casemap :none 
 
.code 
 
TestProc proc dwValue:DWORD 
 
   mov eax, dwValue 
   add eax, 100 
   ret 
 
TestProc endp 
 
end 

  .گرداند  را به ورودي اضافه كرده و سپس حاصل را بر مي۱۰۰ عدد كد   اين 
گزينـه   روي فايل كليك راست كرده و سپس از منـوي ظـاهر شـده    .تنظيم كرد اين پروژه را  build  در ادامه بايد خواص     

  .)شكل زير ( انتخاب كنيدخواص را
  

  
   . Build تنظيم كردن خواص - ۱۶شكل 

  : را انتخاب كرده و سپس خط فرمان زير را در آن وارد كنيدcustom build step/general  قسمت 
ml /c /coff /Zi /Fo"$(OutDir)\$(InputName).asm.obj" "$(InputFileName)"  

  
  : آن بنويسيدoutputs   همچنين در قسمت 

$(OutDir)\$(InputName).asm.obj 
  

  .  آماده شدdebug build   به اين صورت حالت 
  . را حذف نماييدZi/مي توانيد پارامتر   release buildبراي حالت    
  
  

   :++Cفراخواني كد اسمبلر توسط 
  

  :كنيم  به صورت زير عمل مي++C    براي تعريف تابع نوشته شده در اسمبلر در 
extern "C" unsigned int __stdcall TestProc(unsigned int dwValue);  

  
  :   و نهايتا داريم

int main(int argc, _TCHAR* argv[]) 
{ 
   unsigned int dwValue = 100; 
   unsigned int dwReturn = TestProc(dwValue); 
 
   printf("%d\n", dwReturn); 
   getchar(); 
 
   return 0; 
} 

  
 بـر روي تـابع   breakpointتـوان بـا تنظـيم كـردن      مـي ،  اينكه خروجي صـحيحي حاصـل خواهـد شـد     علاوه بر در اينجا      

TestProc درون  ، كد اسمبلي راVC++ IDE  نيز ديباگ كرد.  

 ٦١

  : خوانندهاز درخواستي
ــايي     ــتن فــوق خط ــر در م ــد  را اگ ــي  (مــشاهده فرمودي ــا غيرفن ــرا لطفــا ، ) فنــي ي ــسنده بــه آدرس   آن بــا نوي

vahid_nasiri@yahoo.com مطرح بفرماييد.  
  
  

  
  :مآخذ

1- Win32Asm Tutorial by Thomas Bleeker. Web: http://www.madwizard.org  
2- Masm32's help files.  
3- Iczelion's Win32 Assembly tutorials. Web: http://win32assembly.online.fr/tutorials.html 
4- An Introduction to Assembly Language I/II/III by Darwen.   
    Web: http://www.codeguru.com/Cpp/Cpp/cpp_mfc/tutorials/article.php/c9411 
5- Win32 Assembly parts 1-6 by Chris Hobbs. 
    Web: http://www.gamedev.net/reference/list.asp?categoryid=20#101  
6- Win32 Assembler Coding Tutorial. Web: http://www.deinmeister.de/w32asm1e.htm 

